The inheritanceof local bifurcations in mass action networks

Murad Banaji[1], Balázs Boros[2], Josef Hofbauer[2]

[1] Mathematical Institute, University of Oxford, United Kingdom

[2] Department of Mathematics, University of Vienna, Austria

We compute the first and the second focal values of the homogeneous, rank-three mass action network that appears in Section 4.3 in the paper titled "The inheritance of local bifurcations in mass action networks."

The smallest bimolecular mass-action system with a vertical Andronov-Hopf bifurcation

The network

We have shown in

M. Banaji, B. Boros, J. Hofbauer

The smallest bimolecular mass-action system with a vertical Andronov-Hopf bifurcation

Applied Mathematics Letters, 143:108671, 2023

that the mass-action system

$\mathsf{Z} + \mathsf{X} \xrightarrow{\kappa_1} 2\,\mathsf{X}$

$\mathsf{X} + \mathsf{Y} \xrightarrow{\kappa_2} 2\,\mathsf{Y}$

$\mathsf{Y} + \mathsf{Z} \xrightarrow{\kappa_3} 0 \xrightarrow{\kappa_4} 2\,\mathsf{Z}$

admits a vertical Andronov-Hopf bifurcation (and that this is the smallest such bimolecular system). The bifurcation occurs at $\kappa_1 = \kappa_2 + \kappa_3$.

Verify the transversality of the (vertical) Andronov-Hopf bifurcation

Below we verify the transversality of the bifurcation. For the theoretical details on transversality, see Section 2 in "The inheritance of local bifurcations in mass action networks."

In[1]:=
```
f = κ1 x z {1, 0, -1} + κ2 x y {-1, 1, 0} + κ3 y z {0, -1, -1} + κ4 {0, 0, 2};
κpositive = κ1 > 0 && κ2 > 0 && κ3 > 0 && κ4 > 0;
equilibrium =
  Normal[Simplify[Solve[f == 0 && x > 0 && y > 0 && z > 0 && κpositive, {x, y, z}]][[1]]];
J = D[f, {{x, y, z}}];
{b0, b1, b2, b3} =
  Simplify[CoefficientList[Collect[-CharacteristicPolynomial[J, λ], λ], λ]];
B2B1minusB0 = Simplify[b1 b2 - b0];
Hopf = {κ1 → κ2 + κ3};
HopfCondition = κ2 > 0 && κ3 > 0 && κ4 > 0;
h = Join[f, {B2B1minusB0}];
Dh =
  Simplify[D[h, {{x, y, z, κ1, κ2, κ3, κ4}}] /. equilibrium /. Hopf, HopfCondition];
Reduce[Simplify[Det[Dh[[{1, 2, 3, 4}, {1, 2, 3, 4}]]]] ≠ 0 && HopfCondition]
```

Out[11]=
```
κ2 > 0 && κ3 > 0 && κ4 > 0
```

Andronov-Hopf bifurcation in the homogenised network

The network

Homogenisation of the above network results in the mass-action system

$Z + X \xrightarrow{\kappa_1} 2\,X$

$X + Y \xrightarrow{\kappa_2} 2\,Y$

$Y + Z \xrightarrow{\kappa_3} 2\,W \xrightarrow{\kappa_4} 2\,Z$

Below we analyse the Andronov-Hopf bifurcation that is inherited from the 3d system.

The ray of positive equilibria

Since the molecularity of every complex is the same (namely, two), the stoichiometric classes are given by $x+y+z+w=c$ for $c>0$. Furthermore, since the r.h.s. is homogeneous (of degree two), the phase portrait is the same (up to scaling) in every stoichiometric class. Hence, the set of positive equilibria is a ray.

In[12]:=
```
f = κ1 x z {1, 0, -1, 0} + κ2 x y {-1, 1, 0, 0} + κ3 y z {0, -1, -1, 2} + κ4 w^2 {0, 0, 2, -2};
equilibrium = Normal[Simplify[Solve[
      f == 0 && w == t > 0 && x > 0 && y > 0 && z > 0 && w > 0 && κpositive, {x, y, z, w}]][[1]]];
Print["r.h.s.: ", MatrixForm[f]];
Print["equilibrium ray: ", equilibrium];
```

r.h.s.: $\begin{pmatrix} x\,z\,\kappa_1 - x\,y\,\kappa_2 \\ x\,y\,\kappa_2 - y\,z\,\kappa_3 \\ -x\,z\,\kappa_1 - y\,z\,\kappa_3 + 2\,w^2\,\kappa_4 \\ 2\,y\,z\,\kappa_3 - 2\,w^2\,\kappa_4 \end{pmatrix}$

equilibrium ray: $\left\{x \to t\sqrt{\frac{\kappa_3\,\kappa_4}{\kappa_1\,\kappa_2}},\ y \to t\sqrt{\frac{\kappa_1\,\kappa_4}{\kappa_2\,\kappa_3}},\ z \to t\sqrt{\frac{\kappa_2\,\kappa_4}{\kappa_1\,\kappa_3}},\ w \to t\right\}$

Fix a stoichiometric class and eliminate w

W.l.o.g. we restrict our attention to the stoichiometric class that has the above equilibrium with $t = \sqrt{\frac{\kappa_1\,\kappa_2\,\kappa_3}{\kappa_4}}$. Further, we eliminate the variable w using the conservation law.

In[16]:=

```
tsubst = {t → Sqrt[κ1 κ2 κ3/κ4]};
csubst = {c → (x + y + z + w /. equilibrium)};
ff = Simplify[f[[{1, 2, 3}]] /. {w → c - x - y - z} /. csubst /. tsubst, κpositive];
equil = Normal[Solve[ff == 0 && κpositive && x > 0 && y > 0 && z > 0 &&
      x + y + z < Simplify[(c /. csubst /. tsubst), κpositive], {x, y, z}][[1]]];
Print["r.h.s. of the reduced system: ", MatrixForm[ff]];
Print["the unique positive equilibrium: ", equil];
```

r.h.s. of the reduced system: $\begin{pmatrix} x\,(z\,\kappa_1 - y\,\kappa_2) \\ y\,(x\,\kappa_2 - z\,\kappa_3) \\ -x\,z\,\kappa_1 - y\,z\,\kappa_3 + 2\left(-x - y - z + \kappa_1 + \kappa_2 + \kappa_3 + \sqrt{\frac{\kappa_1\,\kappa_2\,\kappa_3}{\kappa_4}}\right)^2 \kappa_4 \end{pmatrix}$

the unique positive equilibrium: $\{x \to \kappa_3,\ y \to \kappa_1,\ z \to \kappa_2\}$

The Jacobian matrix and higher order derivatives at the positive equilibrium

In[22]:=
```
A = Simplify[D[ff, {{x, y, z}}] /. equil, κpositive];

Idx[set_, n_] := Module[{seq}, seq = (Table[Count[set, i], {i, n}] /. List → Sequence);
    seq];

derivatives = {};
order = 2;
n = 3;
For[i = 0, i ≤ order, i++,
   For[j = 0, j ≤ order - i, j++, For[k = 0, k ≤ order - i - j, k++,
     deriv = Simplify[D[ff, {x, i}, {y, j}, {z, k}] /. equil];
     derivatives = Join[derivatives,
       Simplify[{F_{i,j,k} → deriv[[1]], G_{i,j,k} → deriv[[2]], H_{i,j,k} → deriv[[3]]}]];
    ]]];
B[x_, y_] :=
   Sum[{F_{Idx[{k,l},n]}, G_{Idx[{k,l},n]}, H_{Idx[{k,l},n]}} x[[k]] × y[[l]] /. derivatives, {k, n}, {l, n}];
CC[x_, y_, z_] := {0, 0, 0};
(* due to the bimolecularity all derivatives of order 3 and higher vanish *)
```

The Routh-Hurwitz criterion

In[30]:=
```
{a0, a1, a2, a3} =
   Simplify[CoefficientList[Collect[-CharacteristicPolynomial[A, λ], λ], λ]];
A2A1minusA0 = Simplify[a1 a2 - a0];
RouthHurwitz =
   Simplify[Solve[A2A1minusA0 == 0 && a0 > 0 && a2 > 0 && κpositive, κ4][[1]]];
Hopf = Normal[RouthHurwitz];
HopfCondition = FullSimplify[RouthHurwitz[[1]][[2]][[2]]];
Print["pair of purely imaginary eigenvalues at ",
   Hopf, " assuming ", HopfCondition];
```

pair of purely imaginary eigenvalues at

$\left\{\kappa_4 \to \frac{\kappa_1 (-\kappa_1 + \kappa_2 + \kappa_3)^2}{16 \kappa_2 \kappa_3}\right\}$ assuming $\kappa_2 > 0$ && $\kappa_3 > 0$ && $\kappa_1 > \kappa_2 + \kappa_3$

Left and right eigenvectors of the Jacobian matrix

We eliminate κ_4 in the Jacobian matrix using the Routh-Hurwitz criterion. Then we find the left and right eigenvectors that are needed when computing the first focal value (see e.g. http://www.scholarpedia.org/article/Andronov-Hopf_bifurcation).

In[36]:=
```
AA = Simplify[A /. Hopf, HopfCondition];
mtx = AA - ω ⅈ IdentityMatrix[n];
q = Simplify[NullSpace[mtx⟦Range[1, n - 1]⟧]⟦1⟧];
mtx = AAᵀ - ω ⅈ IdentityMatrix[n];
pconj = Simplify[NullSpace[mtx⟦Range[1, n - 1]⟧]⟦1⟧];
normalize = FullSimplify[pconj.q];
pconj = pconj / normalize;
qconj = FullSimplify[q*, ω > 0 && κpositive];
```

The first focal value (L_1)

We now compute L_1. It takes about 10-20 seconds.

In[44]:=
```
v1 = CC[q, q, qconj];
v2 = Simplify[B[q, Inverse[-AA].B[q, qconj]]];
v3 = Simplify[B[qconj, Inverse[2 I ω IdentityMatrix[n] - AA].B[q, q]]];
c1 = Simplify[pconj.(1/2 v1 + v2 + 1/2 v3)];
L1κω = Simplify[ComplexExpand[Re[c1]], ω > 0 && κpositive];
ωsubst = {ω → Simplify[Sqrt[Det[AA]/Tr[AA]]]};
L1κ = FullSimplify[L1κω /. Hopf /. ωsubst];
Print["L1 equals   ", L1κ];
```

L_1 equals $-\Big(\big(\kappa_1(\kappa_1-\kappa_2-\kappa_3)(\kappa_1+\kappa_2-\kappa_3)^2\kappa_3^2\,\big(5\kappa_1^3+2\kappa_1^2(-3\kappa_2+\kappa_3)+4\kappa_2\kappa_3(-\kappa_2+\kappa_3)+\kappa_1(-7\kappa_2^2+6\kappa_2\kappa_3-3\kappa_3^2)\big)\big)\Big/\Big(8\kappa_2(\kappa_1-\kappa_3)^2\big(\kappa_1(\kappa_1-\kappa_2)^2+(2\kappa_1^2-\kappa_1\kappa_2+\kappa_2^2)\kappa_3+(\kappa_1-\kappa_2)\kappa_3^2\big)\big(\kappa_1^3+4\kappa_2(\kappa_2-\kappa_3)\kappa_3+2\kappa_1^2(-\kappa_2+\kappa_3)+\kappa_1(\kappa_2+\kappa_3)^2\big)\Big)\Big)$

Analyse the sign of L_1

We start with omitting the denominator (which is positive for $\kappa_2 > 0$, $\kappa_3 > 0$, $\kappa_2 + \kappa_3 < \kappa_1$) and a positive factor in the enumerator.

In[52]:=
```
L1 = Numerator[L1κ / (κ1 (κ1 - κ2 - κ3) (κ1 + κ2 - κ3)^2 κ3^2)];
Print["the part of L1 that determines its sign: ", L1];
```

the part of L_1 that determines its sign:
$-5\kappa_1^3-2\kappa_1^2(-3\kappa_2+\kappa_3)-4\kappa_2\kappa_3(-\kappa_2+\kappa_3)-\kappa_1(-7\kappa_2^2+6\kappa_2\kappa_3-3\kappa_3^2)$

W.l.o.g. we may set one of κ_1, κ_2, κ_3 equal to 1. Since L_1 is cubic in κ_1 and only quadratic in κ_2 and κ_3, we set $\kappa_1 = 1$. Then we analyse the expression in the open triangle $\kappa_2 > 0$, $\kappa_3 > 0$, $\kappa_2 + \kappa_3 < 1$.

In[54]:=
```
L1neg = Reduce[(L1 /. {κ1 → 1}) < 0 && κ2 > 0 && κ3 > 0 && κ2 + κ3 < 1];
L1zer = Reduce[(L1 /. {κ1 → 1}) == 0 && κ2 > 0 && κ3 > 0 && κ2 + κ3 < 1, κ3];
L1pos = Reduce[(L1 /. {κ1 → 1}) > 0 && κ2 > 0 && κ3 > 0 && κ2 + κ3 < 1];
Print["L1 negative: ", L1neg];
Print["L1 vanishes: ", L1zer];
Print["L1 positive: ", L1pos];
```

L_1 negative: $\left(0 < \kappa_2 \le \frac{1}{7}\left(-3 + 2\sqrt{11}\right) \&\& \; 0 < \kappa_3 < 1 - \kappa_2\right) \;||\; \left(\frac{1}{7}\left(-3 + 2\sqrt{11}\right) < \kappa_2 < \boxed{0.626\ldots} \;\&\&\; \frac{-1 - 3\kappa_2 + 2\kappa_2^2}{-3 + 4\kappa_2} - 2\sqrt{\frac{4 - 8\kappa_2 + 2\kappa_2^2 + 4\kappa_2^3 + \kappa_2^4}{(-3 + 4\kappa_2)^2}} < \kappa_3 < 1 - \kappa_2\right)$

L_1 vanishes: $\frac{1}{7}\left(-3 + 2\sqrt{11}\right) < \kappa_2 < \boxed{0.626\ldots} \;\&\&\; \kappa_3 == \frac{-1 - 3\kappa_2 + 2\kappa_2^2}{-3 + 4\kappa_2} - 2\sqrt{\frac{4 - 8\kappa_2 + 2\kappa_2^2 + 4\kappa_2^3 + \kappa_2^4}{(-3 + 4\kappa_2)^2}}$

L_1 positive: $\left(\frac{1}{7}\left(-3 + 2\sqrt{11}\right) < \kappa_2 \le \boxed{0.626\ldots} \;\&\&\; 0 < \kappa_3 < \frac{-1 - 3\kappa_2 + 2\kappa_2^2}{-3 + 4\kappa_2} - 2\sqrt{\frac{4 - 8\kappa_2 + 2\kappa_2^2 + 4\kappa_2^3 + \kappa_2^4}{(-3 + 4\kappa_2)^2}}\right) \;||\; \left(\boxed{0.626\ldots} < \kappa_2 < 1 \;\&\&\; 0 < \kappa_3 < 1 - \kappa_2\right)$

Hence, L_1 can have any sign. Below we visualise this.

In[60]:=
```
rneg = RegionPlot[L1neg, {κ2, 0, 1}, {κ3, 0, 1},
    PlotStyle → {Darker[Green], Opacity[0.1]}, GridLines → Automatic,
    MaxRecursion → 6, BoundaryStyle → None, Frame → None,
    AxesLabel → {Style[κ2, Bold, 16], Style[κ3, Bold, 16]}, Axes → True];
rpos = RegionPlot[L1pos, {κ2, 0, 1}, {κ3, 0, 1},
    PlotStyle → {Darker[Red], Opacity[0.1]}, MaxRecursion → 6, BoundaryStyle → None];

L1zersolve = Solve[(L1 /. {κ1 → 1}) == 0 && κ2 > 0 && κ3 > 0 && κ2 + κ3 < 1, κ3][[1]];
curve = Normal[L1zersolve[[1]][[2]]];
min = L1zersolve[[1]][[2]][[2]][[1]];
max = L1zersolve[[1]][[2]][[2]][[5]];
rzer = Plot[curve, {κ2, min, max}, PlotStyle → Blue];

txt = Graphics[{Text[Style["L1<0", Darker[Green], Bold, 16], {0.3, 0.3}],
     Text[Style["L1=0", Blue, Bold, 16], {0.65, 0.42}],
     Text[Style["L1>0", Darker[Red], Bold, 16], {0.7, 0.1}]}];

Show[rneg, rpos, rzer, txt]
```

Out[68]=



The second focal value (L_2)

Below we compute the second focal value along the curve in parameter space, where the first focal value vanishes (note: the process takes about 1-2 minutes). We find that L_2 changes sign. Where $L_2 \neq 0$, a nondegenerate Bautin bifurcation occurs. To understand the behaviour at and near the single point in parameter space with $L_1 = L_2 = 0$, one would need to compute the third focal value, which we leave as an open question.

In[69]:=
```
Id = IdentityMatrix[n];
κ1subst = {κ1 → 1};
```

```
omega = (ω /. ωsubst /. κ1subst);
ωSQsubst = {ω^2 → omega^2, ω^3 → omega^2 ω, ω^4 → omega^4, ω^5 → omega^4 ω,
   ω^6 → omega^6, ω^7 → omega^6 ω, ω^8 → omega^8, ω^9 → omega^8 ω, ω^10 → omega^10};
invA = Simplify[Inverse[AA /. κ1subst]];
inv2 = Simplify[Expand[Inverse[2 ω I Id - AA]] /. ωSQsubst /. κ1subst];
inv3 = Simplify[Expand[Inverse[3 ω I Id - AA]] /. ωSQsubst /. κ1subst];

q = Simplify[q /. ωSQsubst /. κ1subst];
qconj = Simplify[qconj /. ωSQsubst /. κ1subst];
pconj = Simplify[pconj /. ωSQsubst /. κ1subst];

h_{2,0} = Simplify[Expand[Simplify[inv2.B[q, q] /. Hopf /. κ1subst]] /. ωSQsubst];
h_{1,1} = Simplify[Expand[-invA.B[q, qconj]] /. Hopf /. κ1subst /. ωSQsubst];
prec =
  Simplify[ComplexExpand[2 B[q, h_{1,1}] + B[qconj, h_{2,0}]] /. Hopf /. κ1subst /. ωSQsubst];
c_1 = I Simplify[ExpandAll[ComplexExpand[Im[1 / 2 (pconj.prec)]]] /. ωSQsubst];

invbig = Simplify[Expand[Simplify[
      Simplify[Inverse[Join[Join[ω I Id - AA, {q}^T, 2], {Join[pconj, {0}]}]]] /.
        ωSQsubst /. κ1subst]] /. ωSQsubst];
precminus2c1q = Simplify[Expand[Simplify[ExpandAll[prec - 2 c_1 q] /. ωSQsubst]]];
h21 = Simplify[Expand[
     Simplify[Simplify[invbig.Join[precminus2c1q, {0}]] /. ωSQsubst]] /. ωSQsubst];
h_{2,1} = Join[h21[[1 ;; 3]], {0}]; (* turns out the last entry vanishes when L_1=0 *)
h_{3,0} = Simplify[
    Expand[Simplify[inv3.(CC[q, q, q] + 3 B[q, h_{2,0}]) /. Hopf /. κ1subst]] /. ωSQsubst];

v_1 = Simplify[3 B[h_{2,0}, h_{1,1}] /. Hopf /. κ1subst];
v_2 = Simplify[Expand[Simplify[B[qconj, h_{3,0}] /. Hopf /. κ1subst]] /. ωSQsubst];
v_3 = Simplify[Expand[Simplify[3 B[q, h_{2,1}] /. Hopf /. κ1subst]] /. ωSQsubst];
v_4 = Simplify[Expand[-6 c_1 h_{2,0}] /. ωSQsubst];
h_{3,1} =
  Simplify[Expand[Simplify[Expand[inv2.(v_1 + v_2 + v_3 + v_4)] /. ωSQsubst]] /. ωSQsubst];

v_1 = Simplify[2 B[h_{1,1}, h_{1,1}] /. Hopf /. κ1subst];
v_2 = Simplify[
    Expand[Simplify[2 B[q, ComplexExpand[h_{2,1}*]] /. Hopf /. κ1subst]] /. ωSQsubst];
v_3 = Simplify[Expand[Simplify[2 B[qconj, h_{2,1}] /. Hopf /. κ1subst]] /. ωSQsubst];
v_4 = Simplify[
    Expand[Simplify[B[ComplexExpand[h_{2,0}*], h_{2,0}] /. Hopf /. κ1subst /. ωSQsubst]] /.
     ωSQsubst];
R = {0, 0, Simplify[Simplify[v_1[[3]] + v_2[[3]] + v_3[[3]] + v_4[[3]]] /. ωSQsubst]};
(* turns out the first and second entries vanish when L_1=0 *)
h_{2,2} = Simplify[-invA.R];

v_1 = Simplify[
    ComplexExpand[Re[Simplify[Simplify[pconj.(2 B[qconj, h_{3,1}]) /. Hopf /. κ1subst] /.
          ωSQsubst]]] /. ωSQsubst];
v_2 = Simplify[ComplexExpand[Re[Simplify[
```

```
        Simplify[pconj.(3 B[q, h2,2]) /. Hopf /. ϰ1subst] /. ωSQsubst]]] /. ωSQsubst];
v3 = Simplify[ComplexExpand[Re[Simplify[
        Simplify[pconj.B[h2,0*, h3,0] /. Hopf /. ϰ1subst] /. ωSQsubst]]] /. ωSQsubst];
v4 = Simplify[
    ComplexExpand[Re[Simplify[Simplify[pconj.(3 B[h2,1*, h2,0]) /. Hopf /. ϰ1subst] /.
         ωSQsubst]]] /. ωSQsubst];
v5 = Simplify[
    ComplexExpand[Re[Simplify[Simplify[pconj.(6 B[h1,1, h2,1]) /. Hopf /. ϰ1subst] /.
         ωSQsubst]]] /. ωSQsubst];

L2 = Simplify[v1 + v2 + v3 + v4 + v5 /. ωSQsubst];
Print["L2 equals ", L2];
Plot[L2 /. {ϰ3 → curve}, {ϰ2, min, max}, GridLines → Automatic,
 PlotLabel → "L2 (along the curve L1 = 0)", AxesLabel → {Style[ϰ2, Bold, 16], None}]
```

L_2 equals

$$
\begin{aligned}
&-((((1+\kappa_2-\kappa_3)^2 \kappa_3 (-(-1+\kappa_3)^3 \kappa_3 (1+\kappa_3)^{10} (131 + 1475\kappa_3 - 2035\kappa_3^2 + 717\kappa_3^3) + \kappa_2 (-1+\kappa_3)^2 \\
&(1+\kappa_3)^8 (-280 - 2292\kappa_3 - 10791\kappa_3^2 + 34047\kappa_3^3 - 20298\kappa_3^4 + 6046\kappa_3^5 - 11063\kappa_3^6 + 7319\kappa_3^7) + \\
&\kappa_2^{16} (392 + 6347\kappa_3 + 49881\kappa_3^2 + 218545\kappa_3^3 + 527947\kappa_3^4 + 701760\kappa_3^5 + 519824\kappa_3^6 + \\
&203648\kappa_3^7 + 32256\kappa_3^8) - \kappa_2^{15} (3976 + 54068\kappa_3 + 354533\kappa_3^2 + 1318213\kappa_3^3 + \\
&2954387\kappa_3^4 + 4200039\kappa_3^5 + 3949968\kappa_3^6 + 2628528\kappa_3^7 + 1163776\kappa_3^8 + 225792\kappa_3^9) + \\
&\kappa_2^{14} (17192 + 201016\kappa_3 + 1150789\kappa_3^2 + 3858577\kappa_3^3 + 8357361\kappa_3^4 + 12882975\kappa_3^5 + \\
&14456810\kappa_3^6 + 10575152\kappa_3^7 + 5668208\kappa_3^8 + 2754432\kappa_3^9 + 645120\kappa_3^{10}) - \\
&\kappa_2^2 (1+\kappa_3)^6 (-3416 - 7256\kappa_3 - 243\kappa_3^2 + 170387\kappa_3^3 - 270516\kappa_3^4 + 101160\kappa_3^5 + \\
&153558\kappa_3^6 - 450418\kappa_3^7 + 557300\kappa_3^8 - 294952\kappa_3^9 + 17973\kappa_3^{10} + 26423\kappa_3^{11}) - \\
&\kappa_2^{13} (39144 + 424268\kappa_3 + 2293753\kappa_3^2 + 7252657\kappa_3^3 + 15681533\kappa_3^4 + 26478445\kappa_3^5 + \\
&31104078\kappa_3^6 + 26650218\kappa_3^7 + 16109424\kappa_3^8 + 7019984\kappa_3^9 + 3572736\kappa_3^{10} + 903168\kappa_3^{11}) + \kappa_2^{12} \\
&(40936 + 538892\kappa_3 + 3031245\kappa_3^2 + 9436909\kappa_3^3 + 22821095\kappa_3^4 + 39101234\kappa_3^5 + 49310603\kappa_3^6 + \\
&47604437\kappa_3^7 + 28421257\kappa_3^8 + 12083664\kappa_3^9 + 5770912\kappa_3^{10} + 2959104\kappa_3^{11} + 451584\kappa_3^{12}) + \\
&\kappa_2^3 (1+\kappa_3)^4 (-18424 - 5692\kappa_3 + 115813\kappa_3^2 + 307593\kappa_3^3 + 62521\kappa_3^4 - 515315\kappa_3^5 + \\
&947318\kappa_3^6 - 2711362\kappa_3^7 + 3200622\kappa_3^8 - 1871342\kappa_3^9 + 997045\kappa_3^{10} - 691927\kappa_3^{11} + \\
&184769\kappa_3^{12} + 57773\kappa_3^{13}) + \kappa_2^{11} (19096 - 328436\kappa_3 - 2264289\kappa_3^2 - 8776945\kappa_3^3 - \\
&25936551\kappa_3^4 - 44173299\kappa_3^5 - 65994811\kappa_3^6 - 60341187\kappa_3^7 - 44378901\kappa_3^8 - \\
&16820261\kappa_3^9 + 56672\kappa_3^{10} - 3894752\kappa_3^{11} - 1658880\kappa_3^{12} + 451584\kappa_3^{13}) + \\
&\kappa_2^{10} (-125048 - 264648\kappa_3 - 370511\kappa_3^2 + 5339221\kappa_3^3 + 18705413\kappa_3^4 + 45886649\kappa_3^5 + \\
&63311343\kappa_3^6 + 72323471\kappa_3^7 + 51062455\kappa_3^8 + 20089963\kappa_3^9 + \\
&698716\kappa_3^{10} - 6510368\kappa_3^{11} + 3666400\kappa_3^{12} + 13056\kappa_3^{13} - 903168\kappa_3^{14}) - \\
&\kappa_2^4 (1+\kappa_3)^2 (-56952 + 56788\kappa_3 + 605073\kappa_3^2 + 1049723\kappa_3^3 + 2427020\kappa_3^4 - \\
&738749\kappa_3^5 + 401655\kappa_3^6 - 5184794\kappa_3^7 + 1972112\kappa_3^8 + 625458\kappa_3^9 - 830393\kappa_3^{10} + \\
&2269551\kappa_3^{11} - 3151348\kappa_3^{12} + 1329655\kappa_3^{13} - 64127\kappa_3^{14} + 125936\kappa_3^{15}) + \\
&\kappa_2^9 (179256 + 1072468\kappa_3 + 3420683\kappa_3^2 + 1204995\kappa_3^3 - 5509473\kappa_3^4 - 36461521\kappa_3^5 - \\
&53708387\kappa_3^6 - 69971699\kappa_3^7 - 47424571\kappa_3^8 - 27055863\kappa_3^9 - 205476\kappa_3^{10} + \\
&10039620\kappa_3^{11} + 2417184\kappa_3^{12} - 4239008\kappa_3^{13} + 1471488\kappa_3^{14} + 645120\kappa_3^{15}) + \\
&\kappa_2^8 (-109032 - 1679094\kappa_3 - 4949373\kappa_3^2 - 9275525\kappa_3^3 - 3218407\kappa_3^4 + 10074848\kappa_3^5 + \\
&48869222\kappa_3^6 + 44953034\kappa_3^7 + 54276082\kappa_3^8 + 13492878\kappa_3^9 + 4460151\kappa_3^{10} - \\
&1545629\kappa_3^{11} - 10507475\kappa_3^{12} + 3268064\kappa_3^{13} + 2847184\kappa_3^{14} - 1655424\kappa_3^{15} - 225792\kappa_3^{16}) + \\
&\kappa_2^7 (-28952 + 1637956\kappa_3 + 4734561\kappa_3^2 + 12894625\kappa_3^3 + 9678375\kappa_3^4 + 5742811\kappa_3^5 - \\
&23163214\kappa_3^6 - 41875854\kappa_3^7 - 24999986\kappa_3^8 - 23369390\kappa_3^9 + 3058637\kappa_3^{10} + 1221549\kappa_3^{11} - \\
&1161453\kappa_3^{12} + 5495103\kappa_3^{13} - 4174544\kappa_3^{14} - 341872\kappa_3^{15} + 795136\kappa_3^{16} + 32256\kappa_3^{17}) - \\
&\kappa_2^6 (-115192 + 984664\kappa_3 + 3607009\kappa_3^2 + 9966925\kappa_3^3 + 14048957\kappa_3^4 + 6541919\kappa_3^5 - \\
&1684968\kappa_3^6 - 26621470\kappa_3^7 - 22379214\kappa_3^8 - 6308314\kappa_3^9 + 731985\kappa_3^{10} + 2279185\kappa_3^{11} - \\
&3716983\kappa_3^{12} + 174931\kappa_3^{13} + 1935846\kappa_3^{14} - 1855728\kappa_3^{15} + 548432\kappa_3^{16} + 146048\kappa_3^{17}) + \\
&\kappa_2^5 (-107576 + 276860\kappa_3 + 2062277\kappa_3^2 + 5334909\kappa_3^3 + 10951737\kappa_3^4 + 7225513\kappa_3^5 + \\
&379244\kappa_3^6 - 9746912\kappa_3^7 - 14043846\kappa_3^8 - 4968402\kappa_3^9 + 1699725\kappa_3^{10} + 5075181\kappa_3^{11} - \\
&2284891\kappa_3^{12} - 3766035\kappa_3^{13} + 1666834\kappa_3^{14} + 604358\kappa_3^{15} - 87984\kappa_3^{16} + 200048\kappa_3^{17}))) / \\
&(48\kappa_2^3 (-1+\kappa_2-\kappa_3) (-1+\kappa_3)^4 ((1+\kappa_3)^2 + \kappa_2^2 (1+9\kappa_3) + \kappa_2 (-2+7\kappa_3-9\kappa_3^2)) \\
&((1+\kappa_3)^2 + \kappa_2^2 (1+4\kappa_3) + \kappa_2 (-2+2\kappa_3-4\kappa_3^2))^3 \\
&(\kappa_2^2 (1+\kappa_3) + (1+\kappa_3)^2 - \kappa_2 (2+\kappa_3+\kappa_3^2))^3))
\end{aligned}
$$

Out[106]=



Next, we visualize in parameter space the signs of L_1 and L_2. In fact, the drawing is in the (κ_2, κ_3)-plane (recall that κ_4 was eliminated using the Routh-Hurwitz criterion, and we assumed w.l.o.g. that $\kappa_1 = 1$). Since we do compute symbolically the single point in parameter space, where both L_1 and L_2 vanish, the process takes about 40-50 seconds.

In[107]:=

```
κ2TH = Solve[Simplify[L2 /. {κ3 → curve}, min < κ2 < max] == 0 && min < κ2 < max, κ2][[1]];
κ3TH = {κ3 → FullSimplify[curve /. κ2TH]};
TH = {κ2, κ3} /. κ2TH /. κ3TH;
plTH = ListPlot[{TH}, PlotStyle → Blue];
L2neg = Plot[curve, {κ2, min, TH[[1]]}, PlotStyle → RGBColor["Green"]];
L2pos = Plot[curve, {κ2, TH[[1]], max}, PlotStyle → Red];

txtL2 = Graphics[{Text[Style["L1<0", Darker[Green], Bold, 16], {0.2, 0.4}],
    Text[Style["L1=0,L2=0", Blue, Bold, 16], {0.46, TH[[2]]}],
    Text[Style["L1=0,L2>0", Red, Bold, 16], {0.48, 0.35}],
    Text[Style["L1=0,L2<0", RGBColor["Green"], Bold, 16], {0.42, 0.15}],
    Text[Style["L1>0", Darker[Red], Bold, 16], {0.7, 0.1}]}];

Show[rneg, rpos, L2neg, L2pos, plTH, txtL2]
```

Out[114]=



Verify the transversality of the Andronov-Hopf bifurcation

We append the vector field with the Routh-Hurwitz condition and thus get h, see Section 2 in “The inheritance of local bifurcations in mass action networks.” Then we compute the derivative at the equilibrium and the critical parameter value. Thus, we obtain an $(n + c) \times (n + m)$ matrix (here, $n = 3$ is the number of species, $m = 4$ is the number of rate constants, and $c = 1$ is the codimension).

In[115]:=

```
J = D[ff, {{x, y, z}}];
{b0, b1, b2, b3} =
  Simplify[CoefficientList[Collect[-CharacteristicPolynomial[J, λ], λ], λ]];
B2B1minusB0 = Simplify[b1 b2 - b0];
h = Join[ff, {B2B1minusB0}];
Dh = Simplify[D[h, {{x, y, z, κ1, κ2, κ3, κ4}}] /. equil /. Hopf, HopfCondition];
```

We check the nonsingularity of the 4 by 4 matrix formed by the 1st, 2nd, 3rd, and 7th column of $D\,h$ at every κ_1, κ_2, κ_3 with $\kappa_2 + \kappa_3 < \kappa_1$.

In[120]:=

```
Reduce[Simplify[Det[Dh[[{1, 2, 3, 4}, {1, 2, 3, 7}]]]] ≠ 0 && HopfCondition, κ1]
```

Out[120]=

$\kappa_2 > 0 \,\&\&\, \kappa_3 > 0 \,\&\&\, \kappa_1 > \kappa_2 + \kappa_3$

Verify the transversality of the Bautin bifurcation

For the transversality of the Bautin bifurcation, it suffices to check that L_1 (as a function of, say, κ_2) changes sign transversally. I.e., its derivative w.r.t. κ_2 is nonzero when $L_1 = 0$.

In[121]:=

```
der = Simplify[D[L1, {κ2}] /. {κ1 → 1} /. {κ3 → curve}, min < κ2 < max];
Reduce[der ≠ 0 && min < κ2 < max]
```

Out[122]=

$\frac{1}{7}\left(-3 + 2\sqrt{11}\right) < \kappa_2 < 0.626\ldots$