Oscillations in planar deficiency-one mass-action systems

Balázs Boros and Josef Hofbauer

This Mathematica Notebook is a supplementary material to the paper which has the same title as this document. It contains some of the calculations appearing in the paper.

0 Focal values

Compute L_1, L_2, ..., L_m,
the first m focal values. Theoretical background : Chapter 4 in Dumortier, Llibre, Artés : Qualitative Theory of Planar Differential Systems.

For m = 1 or 2 it runs under a second,
for m = 3 in less than a minute. For m = 4 it takes about 20 minutes.

In[1]:=

```
m = 4;
cd = {}; R2cd = {};
For[k=2, k≤2m+1, k++, For[i=0, i≤k, i++,
    {cd=Join[cd,{c_{k,i},d_{k,i}}], R2cd = Join[R2cd,{R_{k,i}→c_{k,i}+d_{k,i}I}]}]];
coeffsxy = CoefficientList[ComplexExpand[Sum[Sum[R_{k,i} z^(k-i) (z*)^i,{i,0,k}], {k,2,2m+1}]
    /.R2cd/.{z→x+y I}], {x,y}];
cond = True;
For[k=2, k≤2m+1, k++, For[i=0, i≤k, i++,
    {cond = cond && (f_{i,k-i}==ComplexExpand[Re[coeffsxy[[i+1,k-i+1]]]]) &&
    (g_{i,k-i}==ComplexExpand[Im[coeffsxy[[i+1,k-i+1]]]])}]];
cd2fg = Solve[cond,cd][[1]];

For[k=2, k≤2m+1, k++, R_k=Sum[R_{k,i} z^(k-i) w^i,{i,0,k}]];

(* F[i,j] computes the polynomial F_i(h_j) *)
F[i_,j_] := Module[{coeffs,M,mtx},
 coeffs = CoefficientList[D[R_i h_j,{z,1}],{z,w}];
 M = Dimensions[coeffs][[1]]-1;
 mtx = (coeffs + Transpose[coeffs*]) Table[If[k+l==M&&k≠l, 1/(k-l),0],{k,0,M},{l,0,M}];
 I z^Range[0,M] .mtx.w^Range[0,M]];

h_0 = 1;
For[k=1, k≤2m-1, k++, h_k = Sum[F[k+1-l,l],{l,0,k-1}]];

(* H[k,j] computes H_k(h_j), note that one of k and j is even, the other is odd
    in all of the interesting cases *)
H[k_,j_] := Module[{},
 coeffs = CoefficientList[h_j,{z,w}];
 Sum[Coefficient[R_k,z^a w^(k-a)]×coeffs[[((k-2a+1)+j)/2+1, (j-(k-2a+1))/2+1]],{a,(k+1-j)/2,(k+1+j)/2}]
];

For[j=1,j≤m,j++,L_j=Simplify[ComplexExpand[2π Re[Sum[H[2j+1-l,l],{l,0,2j-1}]]/.R2cd/.cd2fg]]];
```

The first and the second focal values are as follows. (The third and the fourth ones are very long, we don’t display them on the screen.)

Important note: $f_{i,j}$ and $g_{i,j}$ include the division by $i!\,j!$, so they are not simply the respective partial derivatives, but the coefficients in the Taylor series expansion.

In[14]:=

```
L_1
```

Out[14]=

$$\frac{1}{4}\pi\left(f_{1,2}+f_{1,1}f_{2,0}+3f_{3,0}+f_{0,2}\left(f_{1,1}+2g_{0,2}\right)+3g_{0,3}-g_{0,2}g_{1,1}-2f_{2,0}g_{2,0}-g_{1,1}g_{2,0}+g_{2,1}\right)$$

In[15]:=

```
L2
```

Out[15]=

$$
\begin{aligned}
&\frac{1}{48}\pi \\
&(3 f_{0,3} f_{1,2} - 6 f_{1,1}^2 f_{1,2} + 6 f_{1,4} + 11 f_{0,3} f_{1,1} f_{2,0} - 6 f_{1,1}^3 f_{2,0} + 14 f_{1,3} f_{2,0} - 37 f_{1,2} f_{2,0}^2 - 43 f_{1,1} f_{2,0}^3 + \\
&\quad 3 f_{1,2} f_{2,1} + 7 f_{1,1} f_{2,0} f_{2,1} + 2 f_{1,1} f_{2,2} - 9 f_{0,3} f_{3,0} - 22 f_{1,1}^2 f_{3,0} - 129 f_{2,0}^2 f_{3,0} + 3 f_{2,1} f_{3,0} + \\
&\quad 18 f_{2,0} f_{3,1} + 6 f_{3,2} - 10 f_{1,1} f_{4,0} + 30 f_{5,0} - 41 f_{1,1} f_{1,2} g_{0,2} + 16 f_{0,3} f_{2,0} g_{0,2} - 43 f_{1,1}^2 f_{2,0} g_{0,2} - \\
&\quad 12 f_{2,0}^3 g_{0,2} - 20 f_{2,0} f_{2,1} g_{0,2} - 8 f_{2,2} g_{0,2} - 109 f_{1,1} f_{3,0} g_{0,2} - 44 f_{4,0} g_{0,2} - 55 f_{1,2} g_{0,2}^2 - \\
&\quad 53 f_{1,1} f_{2,0} g_{0,2}^2 - 139 f_{3,0} g_{0,2}^2 + 12 f_{2,0} g_{0,2}^3 + 5 f_{0,2}^3 (f_{1,1} + 2 g_{0,2}) + 6 f_{0,4} (f_{1,1} + 2 g_{0,2}) + \\
&\quad 9 f_{0,3} g_{0,3} - 24 f_{1,1}^2 g_{0,3} - 139 f_{2,0}^2 g_{0,3} - 3 f_{2,1} g_{0,3} - 117 f_{1,1} g_{0,2} g_{0,3} - 129 g_{0,2}^2 g_{0,3} + 44 f_{2,0} g_{0,4} + \\
&\quad 30 g_{0,5} + 4 f_{0,3} f_{1,1} g_{1,1} + 4 f_{1,3} g_{1,1} - 33 f_{1,2} f_{2,0} g_{1,1} - 39 f_{1,1} f_{2,0}^2 g_{1,1} - 117 f_{2,0} f_{3,0} g_{1,1} + \\
&\quad 5 f_{0,3} g_{0,2} g_{1,1} + 8 f_{1,1}^2 g_{0,2} g_{1,1} + 53 f_{2,0}^2 g_{0,2} g_{1,1} + f_{2,1} g_{0,2} g_{1,1} + 39 f_{1,1} g_{0,2}^2 g_{1,1} + 43 g_{0,2}^3 g_{1,1} - \\
&\quad 109 f_{2,0} g_{0,3} g_{1,1} + 10 g_{0,4} g_{1,1} - 8 f_{1,2} g_{1,1}^2 - 8 f_{1,1} f_{2,0} g_{1,1}^2 - 24 f_{3,0} g_{1,1}^2 + 43 f_{2,0} g_{0,2} g_{1,1}^2 - \\
&\quad 22 g_{0,3} g_{1,1}^2 + 6 g_{0,2} g_{1,1}^3 + 3 f_{1,2} g_{1,2} + f_{1,1} f_{2,0} g_{1,2} + 3 f_{3,0} g_{1,2} - 20 f_{2,0} g_{0,2} g_{1,2} - \\
&\quad 3 g_{0,3} g_{1,2} + 7 g_{0,2} g_{1,1} g_{1,2} - 18 g_{0,2} g_{1,3} - 11 f_{1,1} f_{1,2} g_{2,0} + 6 f_{0,3} f_{2,0} g_{2,0} + 3 f_{1,1}^2 f_{2,0} g_{2,0} + \\
&\quad 86 f_{2,0}^3 g_{2,0} - 14 f_{2,0} f_{2,1} g_{2,0} - 4 f_{2,2} g_{2,0} - 19 f_{1,1} f_{3,0} g_{2,0} - 40 f_{4,0} g_{2,0} - 24 f_{1,2} g_{0,2} g_{2,0} + \\
&\quad 54 f_{1,1} f_{2,0} g_{0,2} g_{2,0} - 32 f_{3,0} g_{0,2} g_{2,0} + 106 f_{2,0} g_{0,2}^2 g_{2,0} - 27 f_{1,1} g_{0,3} g_{2,0} - 60 g_{0,2} g_{0,3} g_{2,0} + \\
&\quad 3 f_{0,3} g_{1,1} g_{2,0} + 8 f_{1,1}^2 g_{1,1} g_{2,0} + 133 f_{2,0}^2 g_{1,1} g_{2,0} + 3 f_{2,1} g_{1,1} g_{2,0} + 42 f_{1,1} g_{0,2} g_{1,1} g_{2,0} + \\
&\quad 57 g_{0,2}^2 g_{1,1} g_{2,0} + 57 f_{2,0} g_{1,1}^2 g_{2,0} + 6 g_{1,1}^3 g_{2,0} - 10 f_{2,0} g_{1,2} g_{2,0} + 5 g_{1,1} g_{1,2} g_{2,0} - 6 g_{1,3} g_{2,0} - \\
&\quad 5 f_{1,2} g_{2,0}^2 + 11 f_{1,1} f_{2,0} g_{2,0}^2 + 15 f_{3,0} g_{2,0}^2 + 28 f_{2,0} g_{0,2} g_{2,0}^2 - 15 g_{0,3} g_{2,0}^2 + 3 f_{1,1} g_{1,1} g_{2,0}^2 + \\
&\quad 9 g_{0,2} g_{1,1} g_{2,0}^2 - 10 f_{2,0} g_{2,0}^3 - 5 g_{1,1} g_{2,0}^3 + f_{0,2}^2 (5 f_{1,2} - 15 f_{3,0} - 28 f_{2,0} g_{0,2} + 15 g_{0,3} - \\
&\qquad 11 g_{0,2} g_{1,1} - 3 f_{1,1} (3 f_{2,0} + g_{1,1}) + 10 f_{2,0} g_{2,0} + 5 g_{1,1} g_{2,0} - 5 g_{2,1}) - 3 f_{0,3} g_{2,1} - \\
&\quad 8 f_{1,1}^2 g_{2,1} - 55 f_{2,0}^2 g_{2,1} - 3 f_{2,1} g_{2,1} - 33 f_{1,1} g_{0,2} g_{2,1} - 37 g_{0,2}^2 g_{2,1} - 41 f_{2,0} g_{1,1} g_{2,1} - \\
&\quad 6 g_{1,1}^2 g_{2,1} - 3 g_{1,2} g_{2,1} - 3 f_{1,1} g_{2,0} g_{2,1} - 4 g_{0,2} g_{2,0} g_{2,1} + 5 g_{2,0}^2 g_{2,1} + 8 f_{2,0} g_{2,2} - \\
&\quad 2 g_{1,1} g_{2,2} + 6 g_{2,3} + 3 f_{1,2} g_{3,0} + 5 f_{1,1} f_{2,0} g_{3,0} - 9 f_{3,0} g_{3,0} + 16 f_{2,0} g_{0,2} g_{3,0} + 9 g_{0,3} g_{3,0} + \\
&\quad 4 f_{1,1} g_{1,1} g_{3,0} + 11 g_{0,2} g_{1,1} g_{3,0} + 26 f_{2,0} g_{2,0} g_{3,0} + 13 g_{1,1} g_{2,0} g_{3,0} - 3 g_{2,1} g_{3,0} - \\
&\quad f_{0,2} (6 f_{1,1}^3 - 10 f_{1,3} + 4 f_{1,2} f_{2,0} + 60 f_{2,0} f_{3,0} - 6 f_{3,1} + 106 f_{2,0}^2 g_{0,2} + 10 f_{2,1} g_{0,2} + 86 g_{0,2}^3 - 13 f_{0,3} \\
&\qquad (f_{1,1} + 2 g_{0,2}) + 32 f_{2,0} g_{0,3} - 40 g_{0,4} + 3 f_{1,2} g_{1,1} + 27 f_{3,0} g_{1,1} + 54 f_{2,0} g_{0,2} g_{1,1} + 19 g_{0,3} g_{1,1} + \\
&\qquad 3 g_{0,2} g_{1,1}^2 + 14 g_{0,2} g_{1,2} - 40 f_{2,0}^2 g_{2,0} + 4 f_{2,1} g_{2,0} + 40 g_{0,2}^2 g_{2,0} - 42 f_{2,0} g_{1,1} g_{2,0} - 11 g_{1,1}^2 g_{2,0} + \\
&\qquad 4 g_{1,2} g_{2,0} + 10 g_{0,2} g_{2,0}^2 + f_{1,1}^2 (57 g_{0,2} + 11 g_{2,0}) + 24 f_{2,0} g_{2,1} + 11 g_{1,1} g_{2,1} - 4 g_{2,2} + \\
&\qquad f_{1,1} (57 f_{2,0}^2 - 5 f_{2,1} + 133 g_{0,2}^2 + 42 f_{2,0} g_{1,1} + 8 g_{1,1}^2 - 3 g_{1,2} + 42 g_{0,2} g_{2,0} + 5 g_{2,0}^2 - 3 g_{3,0}) - \\
&\qquad 6 g_{0,2} g_{3,0}) - 4 f_{1,1} g_{3,1} - 14 g_{0,2} g_{3,1} - 10 g_{2,0} g_{3,1} - 12 f_{2,0} g_{4,0} - 6 g_{1,1} g_{4,0} + 6 g_{4,1})
\end{aligned}
$$

Good practice : compute the focal values once, and save them in a file. When needed, load from that file.

In[16]:=

```
L1 = L₁; L2 = L₂; L3 = L₃; L4 = L₄; (* when storing in a file, better avoiding subscripts *)
DumpSave["C://path/focal_values.mx",{L1,L2,L3,L4}];
Remove["Global`*"]; (* clear all variables *)
Get["C://path/focal_values.mx"];
```

3 Quadrangle

3.1 Unstable equilibrium and a stable limit cycle

Find the positive equilibrium.

In[18]:=
```
f = κ1 (a2-a1) x^a1 y^b1 + κ2 (a3-a2) x^a2 y^b2 + κ3 (a4-a3) x^a3 y^b3 + κ4 (a1-a4) x^a4 y^b4;
g = κ1 (b2-b1) x^a1 y^b1 + κ2 (b3-b2) x^a2 y^b2 + κ3 (b4-b3) x^a3 y^b3 + κ4 (b1-b4) x^a4 y^b4;
absubst = {a1→0, b1→1, a2→1, b2→0, a3→1, b3→2, a4→0, b4→3};
Reduce[(f/.absubst)==0 && (g/.absubst)==0 && κ1>0 && κ2>0 && κ3>0 && κ4>0 && x>0 && y>0,{x,y}]
```

Out[21]= $\kappa_1 > 0\ \&\&\ \kappa_2 > 0\ \&\&\ \kappa_3 > 0\ \&\&\ \kappa_4 > 0\ \&\&\ x == \mathrm{Root}\left[\#1^4\,\kappa_2\,\kappa_3^3 - \kappa_1^3\,\kappa_4\ \&,\ 2\right]\ \&\&\ y == \frac{\kappa_1}{x\,\kappa_3}$

Figure out the condition for the trace being positive at the equilibrium.

In[22]:=
```
J = D[{f,g},{{x,y}}];
xysubst = {x→(κ1^3 κ4/(κ3^3 κ2))^(1/4), y→(κ1 κ2/(κ3 κ4))^(1/4)};
trJ = Tr[J/.absubst]/.xysubst;
Reduce[trJ>0 && κ1>0 && κ2>0 && κ3>0 && κ4>0]
```

Out[25]= $\kappa_3 > 0\ \&\&\ \kappa_4 > 0\ \&\&\ \kappa_1 > 0\ \&\&\ 0 < \kappa_2 < \frac{\kappa_1\,\kappa_3\,\kappa_4}{\kappa_3^2 + 12\,\kappa_3\,\kappa_4 + 36\,\kappa_4^2}$

Phase portrait when the equilibrium is unstable.

In[26]:=
```
kappasubst = {κ1→55, κ2→1, κ3→1, κ4→1};
xytime = {x→x[t], y→y[t]};
fspec = f/.absubst/.xytime/.kappasubst;
gspec = g/.absubst/.xytime/.kappasubst;

tmax = 1.5; x0 = (x/.xysubst/.kappasubst)+1; y0 = (y/.xysubst/.kappasubst);
s = NDSolve[{x'[t]==fspec, y'[t]==gspec, x[0]==x0, y[0]==y0}, {x,y}, {t,tmax}];
p1 = ParametricPlot[Evaluate[{x[t],y[t]}/.s], {t,0,tmax}, PlotRange→All, PlotStyle→Red];

tmax = 5; x0 = (x/.xysubst/.kappasubst)+5; y0 = (y/.xysubst/.kappasubst);
s = NDSolve[{x'[t]==fspec, y'[t]==gspec, x[0]==x0, y[0]==y0}, {x,y}, {t,tmax}];
p2 = ParametricPlot[Evaluate[{x[t],y[t]}/.s], {t,0,tmax}, PlotRange→All, PlotStyle→Brown];

strpl = StreamPlot[{f,g}/.absubst/.kappasubst,
 {x,(x/.xysubst/.kappasubst)-5,(x/.xysubst/.kappasubst)+5}, {y,0,(y/.xysubst/.kappasubst)+3}];
pequil = ListPlot[{{x,y}/.xysubst/.kappasubst}, PlotStyle→{Orange,PointSize[Large]}];
Show[strpl,p1,p2,pequil]
```

Out[38]=



3.2 Three limit cycles

Figure out where the trace vanishes.
(Note: we take the scaled ODE, thus in the paper the κ's have overline and the positive equilibrium is at (1,1).)

In[39]:=
```
f = κ1 (a2-a1) x^a1 y^b1 + κ2 (a3-a2) x^a2 y^b2 + κ3 (a4-a3) x^a3 y^b3 + κ4 (a1-a4) x^a4 y^b4;
g = K (κ1 (b2-b1) x^a1 y^b1 + κ2 (b3-b2) x^a2 y^b2 + κ3 (b4-b3) x^a3 y^b3 + κ4 (b1-b4) x^a4 y^b4);
J = D[{f,g},{{x,y}}];
equilibrium = {x→1, y→1};
absubst = {a1→0, b1→1, a2→0, b2→0, a3→1, b3→2, a4→1, b4→5};
kappasubst = {κ1→γ, κ2→1, κ3→(γ+2)/3, κ4→1};
trJ = Simplify[Tr[J/.absubst/.kappasubst]/.equilibrium]
Reduce[trJ==0 && K>0 && γ>0,γ]
```

Out[45]= $-1+K\,(-16+\gamma)$

Out[46]= $K>0\ \&\&\ \gamma == \frac{1+16\,K}{K}$

Store the value of γ, where the trace vanishes.

In[47]:=
```
gammasubst = {γ→16+1/K};
```

Move the equilibrium to (0, 0), bring the equation to canonical form (i.e., the linearization at the origin is $\dot{x}$=-y, $\dot{y} = x$),
and compute the partial derivatives. These partial derivatives will then be substituted into the focal value formulas.

In[48]:=
```
f = κ1 (a2-a1) x^a1 y^b1 + κ2 (a3-a2) x^a2 y^b2 + κ3 (a4-a3) x^a3 y^b3 + κ4 (a1-a4) x^a4 y^b4;
g = K (κ1 (b2-b1) x^a1 y^b1 + κ2 (b3-b2) x^a2 y^b2 + κ3 (b4-b3) x^a3 y^b3 + κ4 (b1-b4) x^a4 y^b4);
J = D[{f,g},{{x,y}}]/.{x→1, y→1};
xyshift = {x→x+1, y→y+1};
f = f/.xyshift;
g = g/.xyshift;
T = {{1,0},{-aa/ω,-bb/ω}};
omegasubst = {ω→Sqrt[Det[J]]};
Tinv = Inverse[T];
Tinvuv = Tinv.{u,v};
FG = (T . {f,g})/ω /.{x→Tinvuv[[1]],y→Tinvuv[[2]]}/.{aa→J[[1,1]],bb→ J[[1,2]]};
F = FG[[1]];
G = FG[[2]];
equil = {u→0,v→0};
Clear[f,g];
m = 3;
derivatives={};
For[i=0, i≤2m+1, i++, For[j=0, j≤2m+1-i, j++, derivatives = Join[derivatives,
{f_{i,j}→(D[F,{u,i},{v,j}]/((i!)*( j!))/.equil), g_{i,j}→(D[G,{u,i},{v,j}]/((i!)*( j!))/.equil)}]]];
```

Compute the first focal value.

In[66]:=
```
derivativessimplified = Simplify[derivatives/.omegasubst/.absubst/.kappasubst/.gammasubst];
L1 = Simplify[L1/.derivativessimplified]
Reduce[L1==0 && K>0]
Plot[L1, {K,0,1/2}, GridLines→Automatic]
```

Out[67]=
$$\frac{\left(-5-29\,K+1250\,K^2+3416\,K^3\right)\pi}{20\,\sqrt{2}\,\left(2+35\,K\right)^{3/2}}$$

Out[68]= $K == 0.0686\ldots$

Out[69]=


Find out the sign of L_2 at $K_0 = 0.0686218$

In[70]:=
```
Ksubst = {K→Root[-5-29 #1+1250 #1^2+3416 #1^3&,3]};
L2 = N[L2/.derivativessimplified/.Ksubst]
```

Out[71]= 0.0129336

Combining with permanence,
this allows us to produce 3 limit cycles. However, with a little more work,
we can find parameter values for which $trJ = L_1 = L_2 = 0$ and $L_3 < 0$. Thus,
3 small limit cycles can be bifurcated from the equilibrium.
We will achieve this by keeping b_4 a parameter, we will call it b.

We start by finding where the trace vanishes.
(Note : we take the scaled ODE, thus in the paper the κ ' s have overline
and the positive equilibrium is at (1, 1). We check this below.)

In[72]:=
```
f = κ₁(a₂-a₁)x^a₁ y^b₁ + κ₂(a₃-a₂)x^a₂ y^b₂ + κ₃(a₄-a₃)x^a₃ y^b₃ + κ₄(a₁-a₄)x^a₄ y^b₄;
g = K(κ₁(b₂-b₁)x^a₁ y^b₁ + κ₂(b₃-b₂)x^a₂ y^b₂ + κ₃(b₄-b₃)x^a₃ y^b₃ + κ₄(b₁-b₄)x^a₄ y^b₄);
J = D[{f,g},{{x,y}}];
equilibrium = {x→1, y→1};

absubst = {a₁→0, b₁→1, a₂→0, b₂→0, a₃→1, b₃→2, a₄→1, b₄→b};
kappasubst = {κ₁→γ, κ₂→1, κ₃→(γ+(b-3))/(b-2), κ₄→1};
f/.absubst/.kappasubst/.equilibrium
g/.absubst/.kappasubst/.equilibrium
trJ = Simplify[Tr[J/.absubst/.kappasubst]/.equilibrium]
Reduce[trJ==0 && K>0 && γ>0 && b>2,γ]
Clear[f,g];
```

Out[78]= 0

Out[79]= 0

Out[80]= $-1+K\left(-6+3\,b-b^2+\gamma\right)$

Out[81]= $b>2\;\&\&\;K>0\;\&\&\;\gamma==\dfrac{1+6\,K-3\,b\,K+b^2\,K}{K}$

Compute L_1, L_2,
L_3 (assuming $trJ=0$) and plot in the $(K,\ b)$ − plane where they vanish.

In[83]:=
```
gammasubst = {γ→(1+6 K-3 b K+b^2 K)/K};
derivativessimplified=Simplify[derivatives/.omegasubst/.absubst/.kappasubst/.gammasubst];
L₁ = Simplify[L1/.derivativessimplified];
L₂ = Simplify[L2/.derivativessimplified];
L₃ = Simplify[L3/.derivativessimplified];
p1 = ContourPlot[{L₁==0, L₂==0, L₃==0}, {K,0.085,.095}, {b,4.7,4.8}, GridLines→Automatic,
PlotLegends→{"L₁=0","L₂=0","L₃=0"}, ImageSize→Medium];
p2 = RegionPlot[L₃<0, {K,0.085,.095}, {b,4.7,4.8}, GridLines→Automatic,
PlotLegends→"L₃<0", ImageSize→Medium];
Row[{p1,p2}]
```

Out[90]=





3.3 Global stability of the equilibrium

We show that the divergence can be made negative for all κ exactly when $a_1 < a_4 < a_2 < a_3$ and $b_1 < b_4 < b_2 < b_3$ are both violated.

For symmetry reasons it suffices to check for the a_i ' s.

In[91]:=
```
A₁ = (α-a₁) (a₁-a₂);
A₂ = (α-a₂) (a₂-a₃);
A₃ = (α-a₃) (a₃-a₄);
A₄ = (α-a₄) (a₄-a₁);
p1 = Reduce[Exists[α, A₁≤0 && A₂≤0 && A₃≤0 && A₄≤0 && A₁+A₂+A₃+A₄<0]]
p2 = Reduce[Exists[α, A₁≤0 && A₂≤0 && A₃≤0 && A₄≤0 && A₁+A₂+A₃+A₄<0] && Not[a₁<a₄<a₂<a₃]]
Reduce[Exists[α, A₁≤0 && A₂≤0 && A₃≤0 && A₄≤0 && A₁+A₂+A₃+A₄<0] && a₁<a₄<a₂<a₃]
Equal[p1,p2]
```

Out[95]=

$(a_1 \mid a_4) \in \mathbb{R}$ && $((a_2 < a_1$ && $((a_3 < a_2$ && $(a_4 \le a_2 \mid\mid a_4 \ge a_1)) \mid\mid a_2 \le a_3 \le a_1 \mid\mid (a_3 > a_1$ && $a_4 \le a_3))) \mid\mid (a_2 == a_1$ && $(a_3 < a_1 \mid\mid (a_3 == a_1$ && $(a_4 < a_1 \mid\mid a_4 > a_1)) \mid\mid a_3 > a_1)) \mid\mid (a_2 > a_1$ && $((a_3 < a_1$ && $a_4 \ge a_3) \mid\mid a_1 \le a_3 \le a_2 \mid\mid (a_3 > a_2$ && $(a_4 \le a_1 \mid\mid a_4 \ge a_2)))))$

Out[96]=

$(a_1 \mid a_4) \in \mathbb{R}$ && $((a_2 < a_1$ && $((a_3 < a_2$ && $(a_4 \le a_2 \mid\mid a_4 \ge a_1)) \mid\mid a_2 \le a_3 \le a_1 \mid\mid (a_3 > a_1$ && $a_4 \le a_3))) \mid\mid (a_2 == a_1$ && $(a_3 < a_1 \mid\mid (a_3 == a_1$ && $(a_4 < a_1 \mid\mid a_4 > a_1)) \mid\mid a_3 > a_1)) \mid\mid (a_2 > a_1$ && $((a_3 < a_1$ && $a_4 \ge a_3) \mid\mid a_1 \le a_3 \le a_2 \mid\mid (a_3 > a_2$ && $(a_4 \le a_1 \mid\mid a_4 \ge a_2)))))$

Out[97]=
```
False
```

Out[98]=
```
True
```

4 Chain of three reactions

Define f, g, h_1, h_2, h_3, h_4.

In[99]:=
```
f = (a₂-a₁) κ₁ x^a₁ y^b₁ + (a₃-a₂) κ₂ x^a₂ y^b₂ + (a₄-a₃) κ₃ x^a₃ y^b₃;
g = (b₂-b₁) κ₁ x^a₁ y^b₁ + (b₃-b₂) κ₂ x^a₂ y^b₂ + (b₄-b₃) κ₃ x^a₃ y^b₃;
Psubst = {P₁→{a₁,b₁}, P₂→{a₂,b₂}, P₃→{a₃,b₃}, P₄→{a₄,b₄}};
hsubst = {h₁→Det[{P₄-P₂,P₃-P₂}/.Psubst],
 h₂→Det[{P₃-P₁,P₄-P₁}/.Psubst],
 h₃→Det[{P₄-P₁,P₂-P₁}/.Psubst],
 h₄→Det[{P₂-P₁,P₃-P₁}/.Psubst]};
```

Verify equation (9).

In[103]:=
```
Simplify[(b₃-b₂) f - (a₃-a₂) g - (-(h₁+h₂+h₃) κ₁ x^a₁ y^b₁ + h₁ κ₃ x^a₃ y^b₃ /.hsubst)]
Simplify[(b₄-b₃) f - (a₄-a₃) g - (+(h₁+h₂) κ₁ x^a₁ y^b₁ - h₁ κ₂ x^a₂ y^b₂ /.hsubst)]
```

Out[103]=
```
0
```

Out[104]=
```
0
```

Verify $h_1 + h_2 + h_3 + h_4 = 0$ and $h_1 + h_2 = \det(P_2 - P_1, P_4 - P_3)$.

In[105]:=
```
h1+h2+h3+h4/.hsubst
Simplify[(h1+h2/.hsubst) - Det[{P2-P1,P4-P3}/.Psubst]]
```

Out[105]= 0

Out[106]= 0

Verify the formula for the determinant of the Jacobian matrix at the equilibrium
(after the linear scaling, so the equilibrium is at (1,1), and the κ's have overline in the paper).

In[107]:=

$$f = (a_2-a_1)\kappa_1 x^{a_1} y^{b_1} + (a_3-a_2)\kappa_2 x^{a_2} y^{b_2} + (a_4-a_3)\kappa_3 x^{a_3} y^{b_3};$$
$$g = K\left((b_2-b_1)\kappa_1 x^{a_1} y^{b_1} + (b_3-b_2)\kappa_2 x^{a_2} y^{b_2} + (b_4-b_3)\kappa_3 x^{a_3} y^{b_3}\right);$$
$$J = D\left[\{f,g\},\{\{x,y\}\}\right]/.\{x\to 1,y\to 1\};$$
$$\text{kappasubst}=\{\kappa_1\to\lambda\, h_1,\ \kappa_2\to\lambda(h_1+h_2),\ \kappa_3\to\lambda(h_1+h_2+h_3)\};$$
$$\text{Simplify}\left[\left(\text{Det}[J]-\frac{h_1+h_2+h_3}{\lambda}K\ \kappa_1\kappa_2\kappa_3\right)/.\text{kappasubst}/.\text{hsubst}\right]$$

Out[111]= 0

4.1 Three limit cycles

Calculate f, g, h_1, h_2, h_3, and the trace.

In[112]:=

$$\text{absubst} = \left\{a_1\to 0,\ b_1\to 0,\ a_2\to 0,\ b_2\to -q,\ a_3\to 1,\ b_3\to\frac{1}{2},\ a_4\to 0,\ b_4\to\frac{1}{2}+r\right\};$$
$$\text{lambdasubst} = \left\{\lambda\to -\frac{1}{q}\right\};$$
$$\{h_1,h_2,h_3\}/.\text{hsubst}/.\text{absubst}$$
$$\{f,g\}/.\text{kappasubst}/.\text{hsubst}/.\text{absubst}/.\text{lambdasubst}$$

```
trJ = Simplify[Tr[J/.kappasubst/.hsubst/.absubst/.lambdasubst]]
Reduce[trJ==0 && K>0 && q>0 && r>0, K]
```

Out[114]= $\left\{-\frac{1}{2}-q-r,\ \frac{1}{2}+r,\ 0\right\}$

Out[115]= $\left\{-x\sqrt{y}+y^{-q},\ K\left(-\frac{1}{2}-q-r+r\,x\sqrt{y}+\left(\frac{1}{2}+q\right)y^{-q}\right)\right\}$

Out[116]= $-1-K\,q\left(\frac{1}{2}+q\right)+\frac{K\,r}{2}$

Out[117]= $q>0\ \&\&\ r>q+2q^2\ \&\&\ K==\frac{2}{-q-2q^2+r}$

Move the equilibrium to (0, 0), bring the equation to canonical form (i.e., the linearization at the origin is $\dot{x} = -y$, $\dot{y} = x$), and compute the partial derivatives. These partial derivatives will then be substituted into the focal value formulas.

In[118]:=
```
f = (a2-a1)κ1 x^a1 y^b1 + (a3-a2)κ2 x^a2 y^b2 + (a4-a3)κ3 x^a3 y^b3;
g = K((b2-b1)κ1 x^a1 y^b1 + (b3-b2)κ2 x^a2 y^b2 + (b4-b3)κ3 x^a3 y^b3);
J = D[{f,g},{{x,y}}]/.{x→1, y→1};
xyshift = {x→x+1, y→y+1};
f = f/.xyshift;
g = g/.xyshift;
T = {{1,0},{-aa/ω,-bb/ω}};
omegasubst = {ω→Sqrt[Det[J]]};
Tinv = Inverse[T];
Tinvuv = Tinv.{u,v};
FG = (T . {f,g})/ω /.{x→Tinvuv[[1]],y→Tinvuv[[2]]}/.{aa→J[[1,1]],bb→ J[[1,2]]};
F = FG[[1]];
G = FG[[2]];
equil = {u→0,v→0};
Clear[f,g];
m = 3;
derivatives = {};
For[i=0, i≤2m+1, i++, For[j=0, j≤2m+1-i, j++, derivatives = Join[derivatives,
{f_{i,j}→(D[F,{u,i},{v,j}]/((i!)*( j!))/.equil), g_{i,j}→(D[G,{u,i},{v,j}]/((i!)*( j!))/.equil)}]]];
```

Compute the first focal value and check where it vanishes.

In[136]:=
```
Ksubst = {K→2/(r-q(1+2 q))};
derivativessimplified =
Simplify[derivatives/.omegasubst/.kappasubst/.hsubst/.absubst/.lambdasubst/.Ksubst];
L1 = Simplify[L1/.derivativessimplified]
Reduce[L1==0 && q>0 && r>q(2q+1)]
```

Out[138]=
$$-\frac{\pi\, r\, \left(16\, q^2 + 4\, q^3 - 3\, r + q\, (7 + 6\, r)\right)}{8\, (1 + 2\, q)\, \left(q + 2\, q^2 - r\right)^2 \sqrt{-\frac{q\, (1+2\, q+2\, r)}{q+2\, q^2-r}}}$$

Out[139]=
$$0 < q < \frac{1}{2} \,\&\&\, r == \frac{-7\, q - 16\, q^2 - 4\, q^3}{-3 + 6\, q}$$

Compute the second focal value and check where it vanishes.

In[140]:=

```
rsubst = {r→(-7 q-16 q^2-4 q^3)/(-3+6 q)};
L₂ = Simplify[L2/.derivativessimplified/.rsubst]
Reduce[L₂==0 && 0<q<1/2]
Plot[L₂, {q,0,1/2}, GridLines→Automatic]
```

Out[141]= $-\frac{\pi \sqrt{3+2q}\,(7+2q)^2\,(3-14q+8q^2)}{1536\,(1+2q)^4}$

Out[142]= $q == \frac{1}{4}$

Out[143]=



Back substitute to get r and K. Check the sign of L_3.

In[144]:=

```
qsubst = {q→1/4}
rsubst/.qsubst
Ksubst/.rsubst/.qsubst
L₃ = Simplify[L3/.derivativessimplified/.rsubst/.qsubst]
```

Out[144]= $\{q \to \frac{1}{4}\}$

Out[145]= $\{r \to \frac{15}{8}\}$

Out[146]= $\{K \to \frac{4}{3}\}$

Out[147]= $-\frac{625\sqrt{\frac{7}{2}}\,\pi}{110\,592}$

4.2 Reversible center

Check κ_1, κ_2, κ_3 (with overline), the r.h.s. of the ODE, and the Jacobian at the equilibrium (1,1).

In[148]:=

```
f = (a2-a1) κ1 x^a1 y^b1 + (a3-a2) κ2 x^a2 y^b2 + (a4-a3) κ3 x^a3 y^b3;
g = K((b2-b1) κ1 x^a1 y^b1 + (b3-b2) κ2 x^a2 y^b2 + (b4-b3) κ3 x^a3 y^b3);
absubst = {a1→0, b1→0, a2→p, b2→q, a3→q, b3→p, a4→q-p, b4→p+q^2/p};
Ksubst = {K→-p/q};
lambdasubst = {λ→-1/(p^2-q^2)};
Simplify[{κ1,κ2,κ3}/.kappasubst/.hsubst/.absubst/.Ksubst/.lambdasubst]
fg = {f,g}/.kappasubst/.hsubst/.absubst/.Ksubst/.lambdasubst;
Simplify[fg]
MatrixForm[Simplify[J/.kappasubst/.hsubst/.absubst/.Ksubst/.lambdasubst]]
```

Out[153]=
$$\left\{1-\frac{q}{p},\ -\frac{q}{p-q},\ 1\right\}$$

Out[155]=
$$\left\{p-p\,x^q y^p+q\left(-1+x^p y^q\right),\ q-q\,x^q y^p+p\left(-1+x^p y^q\right)\right\}$$

Out[156]//MatrixForm=
$$\begin{pmatrix} 0 & -p^2+q^2 \\ p^2-q^2 & 0 \end{pmatrix}$$

Plot the toric rays that are asymptotic to the homoclinic orbit.

In[157]:=
```
pqsubst = {p→3/2, q→-1};
strpl = StreamPlot[fg/.pqsubst, {x,0,1/10}, {y,0,1/10}];
toric1 = Plot[x^(-p/q) ((p-q)/p)^(1/q)/.pqsubst, {x,0,1/10}, PlotStyle→Red];
toric2 = Plot[x^(-q/p) ((p-q)/p)^(1/p)/.pqsubst, {x,0,1/10}, PlotStyle→Red];
Show[strpl,toric1,toric2]
```

Out[161]=


5 Three reactions

Verify equation (18).

In[162]:=
```
f = κ₁c₁x^a₁y^b₁ + κ₂c₂x^a₂y^b₂ + κ₃c₃x^a₃y^b₃;
g = κ₁d₁x^a₁y^b₁ + κ₂d₂x^a₂y^b₂ + κ₃d₃x^a₃y^b₃;
Simplify[d₂f - c₂g - (+(c₁d₂-c₂d₁)κ₁x^a₁y^b₁ - (c₂d₃-c₃d₂)κ₃x^a₃y^b₃/.hsubst)]
Simplify[d₃f - c₃g - (-(c₃d₁-c₁d₃)κ₁x^a₁y^b₁ + (c₂d₃-c₃d₂)κ₂x^a₂y^b₂/.hsubst)]
```

Out[164]=
```
0
```

Out[165]=
```
0
```

Verify equation (23) on the determinant of the Jacobian matrix.

In[166]:=
```
f = κ₁c₁x^a₁y^b₁ + κ₂c₂x^a₂y^b₂ + κ₃c₃x^a₃y^b₃;
g = K(κ₁d₁x^a₁y^b₁ + κ₂d₂x^a₂y^b₂ + κ₃d₃x^a₃y^b₃);
kappasubst = {κ₁→λ(c₂d₃-c₃d₂), κ₂→λ(c₃d₁-c₁d₃), κ₃→λ(c₁d₂-c₂d₁)};
xstar = 1;
ystar = 1;
J = D[{f,g},{{x,y}}]/.{x→1,y→1};
Simplify[(Det[J]-1/λ K κ₁κ₂κ₃(a₁(b₂-b₃)+a₂(b₃-b₁)+a₃(b₁-b₂)))/.kappasubst]
```

Out[172]= 0

5.1 Four limit cycles

Check where the trace vanishes.

In[173]:=
```
f = κ₁c₁x^a₁y^b₁ + κ₂c₂x^a₂y^b₂ + κ₃c₃x^a₃y^b₃;
g = K(κ₁d₁x^a₁y^b₁ + κ₂d₂x^a₂y^b₂ + κ₃d₃x^a₃y^b₃);
abcdsubst = {a₁→0, b₁→0, c₁→0, d₁→-1, a₂→0, b₂→-1, c₂→1, d₂→-1, a₃→a, b₃→b, c₃→-1, d₃→d};
kappasubst = {κ₁→λ(c₂d₃-c₃d₂), κ₂→λ(c₃d₁-c₁d₃), κ₃→λ(c₁d₂-c₂d₁)};
lambdasubst = {λ→1};
fg = {f,g}/.kappasubst/.abcdsubst/.lambdasubst
J = D[fg,{{x,y}}]/.{x→1,y→1};
Reduce[Tr[J]==0 && a>0 && b>-1 && d>0 && 1+b d>0,K]
```

Out[178]= $\left\{\frac{1}{y}-x^a y^b,\ K\left(1-d-\frac{1}{y}+d\,x^a y^b\right)\right\}$

Out[180]= $\left(\left(0<d\leq 1\ \&\&\ b>-1\ \&\&\ a>0\right)\ ||\ \left(d>1\ \&\&\ b>-\frac{1}{d}\ \&\&\ a>0\right)\right)\ \&\&\ K==\frac{a}{1+b\,d}$

Store the value for K that makes the trace zero.

In[181]:=
```
Ksubst = {K→a/(1+b d)};
```

Move the equilibrium to (0, 0), bring the equation to canonical form (i.e., the linearization at the origin is $\dot{x}=-y$, $\dot{y}=x$), and compute the partial derivatives (up to order 9). These partial derivatives will then be substituted into the focal value formulas.

In[182]:=
```
f = κ₁c₁x^a₁y^b₁ + κ₂c₂x^a₂y^b₂ + κ₃c₃x^a₃y^b₃;
g = K(κ₁d₁x^a₁y^b₁ + κ₂d₂x^a₂y^b₂ + κ₃d₃x^a₃y^b₃);
J = D[{f,g},{{x,y}}]/.{x→1, y→1};
xyshift = {x→x+1, y→y+1};
f = f/.xyshift;
g = g/.xyshift;
T = {{1,0},{-aa/ω,-bb/ω}};
omegasubst = {ω→Sqrt[Det[J]]};
Tinv = Inverse[T];
Tinvuv = Tinv.{u,v};
FG = (T . {f,g})/ω /.{x→Tinvuv[[1]],y→Tinvuv[[2]]}/.{aa→J[[1,1]],bb→ J[[1,2]]};
F = FG[[1]];
G = FG[[2]];
equil = {u→0,v→0};
Clear[f,g];
m = 4;
derivatives = {};
For[i=0, i≤2m+1, i++, For[j=0, j≤2m+1-i, j++, derivatives = Join[derivatives,
{f_{i,j}→(D[F,{u,i},{v,j}]/((i!)*( j!))/.equil), g_{i,j}→(D[G,{u,i},{v,j}]/((i!)*( j!))/.equil)}]]];
```

Compute the first focal value and check where it vanishes.

In[200]:=
```
derivativessimplified =
Simplify[derivatives/.omegasubst/.Ksubst/.kappasubst/.abcdsubst/.lambdasubst];
L₁ = Simplify[L1/.derivativessimplified]
Reduce[L₁==0 && a>0 && d>1 && 1+b d>0,b]
```

Out[201]=

$$\frac{a\left(a+2a^2 d+a b d-(1+b d)^2\right)\pi}{8\sqrt{\frac{a^2(-1+d)}{1+b d}}\,(1+b d)^2}$$

Out[202]=

$$a>0\ \&\&\ d>1\ \&\&\ b==\frac{-2+a}{2d}+\frac{1}{2}\sqrt{\frac{a^2+8a^2 d}{d^2}}$$

Let' s see the second, third, and fourth focal values. They get complicated. (The computation of L_4 may take a few minutes.)

In[203]:=
```
bsubst = {b→(-2+a(1+√(1+8 d)))/(2 d)};
L₂ = Simplify[FullSimplify[L2/.derivativessimplified]/.bsubst]
L₃ = Simplify[FullSimplify[L3/.derivativessimplified]/.bsubst]
L₄ = Simplify[FullSimplify[L4/.derivativessimplified]/.bsubst]
```

Out[204]=

$$\left(a\left(-(-1+d)\left(68 d^2+2 d\left(-5+\sqrt{1+8 d}\right)-3\left(1+\sqrt{1+8 d}\right)\right)+a^2(-1+d)\left(3\left(1+\sqrt{1+8 d}\right)+2 d^2\left(11+\sqrt{1+8 d}\right)+d\left(23+11 \sqrt{1+8 d}\right)\right)+a\left(12 d^3 \sqrt{1+8 d}+6\left(1+\sqrt{1+8 d}\right)+3 d\left(9+\sqrt{1+8 d}\right)-d^2\left(57+29 \sqrt{1+8 d}\right)\right)\right) \pi\right) \Big/ \left(18 \sqrt{2} \sqrt{\frac{a(-1+d)}{1+\sqrt{1+8 d}}}\left(1+\sqrt{1+8 d}\right)^4\left(-2+a+2 d+a \sqrt{1+8 d}\right)\right)$$

Out[205]=

$$-\left(\left(a^2\left((-1+d)^3\left(972 d^3+d\left(893-55 \sqrt{1+8 d}\right)+237\left(1+\sqrt{1+8 d}\right)-2 d^2\left(2431+491 \sqrt{1+8 d}\right)\right)+2 a^2(-1+d)\left(264 d^6+702\left(1+\sqrt{1+8 d}\right)-40 d^2\left(-149+52 \sqrt{1+8 d}\right)+2 d^5\left(613+483 \sqrt{1+8 d}\right)+6 d\left(1037+569 \sqrt{1+8 d}\right)-3 d^3\left(8351+3687 \sqrt{1+8 d}\right)-d^4\left(24805+3769 \sqrt{1+8 d}\right)\right)-a(-1+d)^2\left(-942\left(1+\sqrt{1+8 d}\right)+d^4\left(9758+450 \sqrt{1+8 d}\right)-2 d\left(2975+1091 \sqrt{1+8 d}\right)+d^3\left(21647+4931 \sqrt{1+8 d}\right)+d^2\left(5813+7397 \sqrt{1+8 d}\right)\right)+a^4(-1+d)\left(231\left(1+\sqrt{1+8 d}\right)+4 d^5\left(703+31 \sqrt{1+8 d}\right)+11 d\left(335+251 \sqrt{1+8 d}\right)+14 d^4\left(1599+257 \sqrt{1+8 d}\right)+d^2\left(19083+9887 \sqrt{1+8 d}\right)+d^3\left(36475+11623 \sqrt{1+8 d}\right)\right)+a^3\left(120 d^7+930\left(1+\sqrt{1+8 d}\right)+28 d^6\left(493+44 \sqrt{1+8 d}\right)+10 d\left(1061+689 \sqrt{1+8 d}\right)+d^2\left(25893+5773 \sqrt{1+8 d}\right)+d^5\left(533+6985 \sqrt{1+8 d}\right)-d^4\left(71103+15067 \sqrt{1+8 d}\right)-d^3\left(23123+20855 \sqrt{1+8 d}\right)\right)\right) \pi\right) \Big/ \left(72 \sqrt{2}\left(\frac{a(-1+d)}{1+\sqrt{1+8 d}}\right)^{3/2}\left(1+\sqrt{1+8 d}\right)^7\left(-2+a+2 d+a \sqrt{1+8 d}\right)^2\right)\right)$$

Out[206]=

$$
\begin{aligned}
&\Bigg(\sqrt{\frac{a\,(-1+d)}{1+\sqrt{1+8\,d}}}\\
&\quad\Big(3\,(-1+d)^5\,\Big(-1\,350\,168\,d^4+1\,825\,044\,\big(1+\sqrt{1+8\,d}\big)+8\,d\,\big(1\,039\,715+127\,193\,\sqrt{1+8\,d}\big)+\\
&\qquad\quad 4\,d^3\,\big(-1\,942\,467+275\,354\,\sqrt{1+8\,d}\big)-d^2\,\big(32\,194\,101+13\,237\,315\,\sqrt{1+8\,d}\big)\Big)+\\
&\qquad 2\,a\,(-1+d)^4\,\Big(16\,314\,192\,\big(1+\sqrt{1+8\,d}\big)+20\,d^5\,\big(746\,612+52\,299\,\sqrt{1+8\,d}\big)+\\
&\qquad\quad 6\,d\,\big(20\,372\,149+9\,496\,021\,\sqrt{1+8\,d}\big)-d^4\,\big(363\,681\,203+36\,949\,049\,\sqrt{1+8\,d}\big)-\\
&\qquad\quad 2\,d^3\,\big(321\,024\,423+99\,959\,585\,\sqrt{1+8\,d}\big)-d^2\,\big(33\,552\,995+133\,752\,833\,\sqrt{1+8\,d}\big)\Big)+\\
&\qquad 12\,a^6\,(-1+d)\,\Big(437\,415\,\big(1+\sqrt{1+8\,d}\big)+8\,d^8\,\big(799\,993+23\,231\,\sqrt{1+8\,d}\big)+\\
&\qquad\quad 16\,d^7\,\big(8\,299\,934+811\,265\,\sqrt{1+8\,d}\big)+d\,\big(10\,587\,487+8\,837\,827\,\sqrt{1+8\,d}\big)+\\
&\qquad\quad 19\,d^5\,\big(64\,786\,897+18\,235\,757\,\sqrt{1+8\,d}\big)+6\,d^3\,\big(75\,001\,813+39\,832\,137\,\sqrt{1+8\,d}\big)+\\
&\qquad\quad 2\,d^6\,\big(328\,894\,755+59\,699\,281\,\sqrt{1+8\,d}\big)+d^2\,\big(98\,782\,836+66\,930\,848\,\sqrt{1+8\,d}\big)+\\
&\qquad\quad d^4\,\big(1\,052\,537\,463+419\,188\,895\,\sqrt{1+8\,d}\big)\Big)-\\
&\qquad 4\,a^2\,(-1+d)^3\,\Big(2\,333\,160\,d^7-20\,252\,799\,\big(1+\sqrt{1+8\,d}\big)+\\
&\qquad\quad d^6\,\big(132\,972\,559+11\,159\,685\,\sqrt{1+8\,d}\big)-d^2\,\big(377\,742\,403+18\,933\,343\,\sqrt{1+8\,d}\big)-\\
&\qquad\quad 3\,d\,\big(70\,406\,353+43\,402\,621\,\sqrt{1+8\,d}\big)+2\,d^5\,\big(416\,513\,618+59\,027\,017\,\sqrt{1+8\,d}\big)+\\
&\qquad\quad 5\,d^4\,\big(390\,575\,233+108\,024\,339\,\sqrt{1+8\,d}\big)+d^3\,\big(894\,225\,897+567\,337\,183\,\sqrt{1+8\,d}\big)\Big)+\\
&\qquad 4\,a^5\,\Big(1\,129\,632\,d^{10}+d^4\,\big(267\,696\,186-904\,979\,838\,\sqrt{1+8\,d}\big)+7\,930\,944\,\big(1+\sqrt{1+8\,d}\big)+\\
&\qquad\quad 8\,d^9\,\big(22\,872\,659+1\,162\,683\,\sqrt{1+8\,d}\big)+18\,d^7\,\big(38\,132\,353+33\,877\,975\,\sqrt{1+8\,d}\big)+\\
&\qquad\quad 3\,d\,\big(50\,904\,457+40\,329\,865\,\sqrt{1+8\,d}\big)+4\,d^8\,\big(351\,507\,807+57\,831\,427\,\sqrt{1+8\,d}\big)-\\
&\qquad\quad 13\,d^5\,\big(471\,734\,401+192\,198\,281\,\sqrt{1+8\,d}\big)+4\,d^3\,\big(611\,285\,790+203\,687\,477\,\sqrt{1+8\,d}\big)+\\
&\qquad\quad d^2\,\big(1\,006\,640\,779+586\,129\,951\,\sqrt{1+8\,d}\big)-d^6\,\big(5\,484\,544\,993+796\,821\,389\,\sqrt{1+8\,d}\big)\Big)+\\
&\qquad 4\,a^3\,(-1+d)^2\,\Big(d^3\,\big(56\,591\,601-679\,837\,695\,\sqrt{1+8\,d}\big)+26\,815\,896\,\big(1+\sqrt{1+8\,d}\big)+\\
&\qquad\quad 72\,d^8\,\big(220\,198+5515\,\sqrt{1+8\,d}\big)+120\,d^2\,\big(10\,519\,564+3\,933\,431\,\sqrt{1+8\,d}\big)+\\
&\qquad\quad 4\,d^7\,\big(10\,082\,069+6\,086\,434\,\sqrt{1+8\,d}\big)+6\,d\,\big(59\,746\,561+41\,869\,297\,\sqrt{1+8\,d}\big)-\\
&\qquad\quad d^6\,\big(1\,454\,017\,215+132\,100\,123\,\sqrt{1+8\,d}\big)-d^5\,\big(5\,198\,444\,217+1\,151\,299\,573\,\sqrt{1+8\,d}\big)-\\
&\qquad\quad d^4\,\big(4\,753\,010\,123+2\,026\,566\,983\,\sqrt{1+8\,d}\big)\Big)+\\
&\qquad 4\,a^4\,(-1+d)\,\Big(19\,970\,109\,\big(1+\sqrt{1+8\,d}\big)+12\,d^9\,\big(825\,499+7431\,\sqrt{1+8\,d}\big)-\\
&\qquad\quad 27\,d^6\,\big(175\,810\,871+19\,459\,271\,\sqrt{1+8\,d}\big)+12\,d\,\big(27\,142\,822+20\,486\,119\,\sqrt{1+8\,d}\big)+\\
&\qquad\quad d^8\,\big(378\,439\,718+34\,608\,994\,\sqrt{1+8\,d}\big)+d^7\,\big(378\,911\,879+246\,975\,083\,\sqrt{1+8\,d}\big)+\\
&\qquad\quad d^3\,\big(2\,278\,163\,397+257\,610\,265\,\sqrt{1+8\,d}\big)-4\,d^5\,\big(2\,323\,240\,174+705\,361\,543\,\sqrt{1+8\,d}\big)+d^2\\
&\qquad\qquad \big(1\,660\,617\,107+837\,044\,267\,\sqrt{1+8\,d}\big)-d^4\,\big(3\,428\,684\,549+2\,434\,223\,729\,\sqrt{1+8\,d}\big)\Big)\Big)\,\pi\Bigg)\Bigg/\\
&\Big(64\,800\,\sqrt{2}\,(-1+d)^3\,\big(1+\sqrt{1+8\,d}\big)^7\,\big(-2+a+2\,d+a\,\sqrt{1+8\,d}\big)^3\Big)
\end{aligned}
$$

Plot the curve in the (a, d) plane where L_2 vanishes.

In[207]:=
```
ContourPlot[L2==0, {a,1/2,2}, {d,1,4}, GridLines→Automatic]
```

Out[207]=



Take the enumerator of L_2, it is quadratic in "a",
so we can solve $L_2 = 0$ explicitly. We choose that root,
which gives positive value for "a".

In[208]:=
```
L2enumerator = (-(-1+d) (68 d^2+2 d (-5+Sqrt[1+8 d])-3 (1+Sqrt[1+8 d]))+
a^2 (-1+d) (3 (1+Sqrt[1+8 d])+2 d^2 (11+Sqrt[1+8 d])+d (23+11 Sqrt[1+8 d]))+
a (12 d^3 Sqrt[1+8 d]+6 (1+Sqrt[1+8 d])+3 d (9+Sqrt[1+8 d])-d^2 (57+29 Sqrt[1+8 d])));
c = CoefficientList[L2enumerator,a];
asubst = Simplify[{a→(-c[[2]]+Sqrt[c[[2]]^2-4 c[[3]]×c[[1]]])/(2 c[[3]])}]
Clear[c]
Reduce[(a/.asubst)>0 && d>1]
```

Out[210]=

$$\left\{a \to \left(-6-6\sqrt{1+8d}-12d^3\sqrt{1+8d}-3d\left(9+\sqrt{1+8d}\right)+d^2\left(57+29\sqrt{1+8d}\right)+\sqrt{2}\sqrt{\left(d^2\left(576d^5+169\left(1+\sqrt{1+8d}\right)+8d^4\left(43+34\sqrt{1+8d}\right)+2d\left(407+69\sqrt{1+8d}\right)+4d^3\left(36+79\sqrt{1+8d}\right)-d^2\left(1471+703\sqrt{1+8d}\right)\right)\right)}\right)\Big/\left(2(-1+d)\left(3\left(1+\sqrt{1+8d}\right)+2d^2\left(11+\sqrt{1+8d}\right)+d\left(23+11\sqrt{1+8d}\right)\right)\right)\right\}$$

Out[212]=

$d > 1$

Plug in the above "a" to L_3 and L_4 and let's see numerically what the sign of L_4 is,
where L_3 vanishes.

In[218]:=
```
L3d = Simplify[L₃/.asubst];
L4d = Simplify[L₄/.asubst];
p1 = Plot[{L3d,L4d}, {d,1.01,4}, PlotLegends→{"L₃","L₄"}, GridLines→Automatic,
ImageSize→Medium];
p2 = Plot[{L3d,L4d}, {d,3.2885,3.2890}, PlotLegends→{"L₃","L₄"}, GridLines→Automatic,
ImageSize→Medium];
Row[{p1,p2}]
```

Out[222]=





Thus, we numerically found that there exist parameter values such that trJ = $L_1 = L_2 = L_3 = 0$ and $L_4 < 0$.

5.2 Reversible center

Verify the trace and the determinant formula in the proof of Proposition 6.

In[223]:=
```
f = κ1 c1 x^a1 y^b1 + κ2 c2 x^a2 y^b2 + κ3 c3 x^a3 y^b3;
g = K (κ1 d1 x^a1 y^b1 + κ2 d2 x^a2 y^b2 + κ3 d3 x^a3 y^b3);
absubst = {a1→0, b1→0, a2→p, b2→q, a3→q, b3→p};
kappasubst = {κ1→λ(c2d3-c3d2), κ2→λ(c3d1-c1d3), κ3→λ(c1d2-c2d1)};
J=D[{f,g},{{x,y}}]/.{x→1,y→1};
Simplify[(Det[J]-1/λ K κ1κ2κ3(p^2-q^2))/.kappasubst/.absubst]
Simplify[(Tr[J]-(p(c2κ2+K d3κ3)+q(c3κ3+K d2κ2)))/.kappasubst/.absubst]
```

Out[228]= 0

Out[229]= 0

5.3 Liénard center

Verify the trace and the determinant formula in the proof of Proposition 7.

In[230]:=
```
absubst = {a1→1, b1→0, a2→0, b2→-1/2, a3→0, b3→-2};
xdot = (κ1 c1 x^a1 y^b1 + κ2 c2 x^a2 y^b2 + κ3 c3 x^a3 y^b3)/.absubst;
ydot = (K (κ1 d1 x^a1 y^b1 + κ2 d2 x^a2 y^b2 + κ3 d3 x^a3 y^b3))/.absubst;
kappasubst = {κ1→λ(c2d3-c3d2), κ2→λ(c3d1-c1d3), κ3→λ(c1d2-c2d1)};
J=D[{xdot,ydot},{{x,y}}]/.{x→1,y→1};
Simplify[(Det[J]-3/2 1/λ K κ1κ2κ3)/.kappasubst/.absubst]
Simplify[(Tr[J]-(c1κ1-1/2 K d2κ2-2K d3κ3))/.kappasubst/.absubst]
```

Out[235]= 0

Out[236]= 0

Check $\ddot{y} + f(y)\,\dot{y} + g(y) = 0$ in the proof of Proposition 7.

In[237]:=
```
xyshift = {x→x[τ]+1, y→y[τ]+1};
xdot = xdot/.xyshift;
ydot = ydot/.xyshift;
f = -c1κ1 + 1/2 K d2κ2 (y[τ]+1)^(-3/2) + 2K d3κ3 (y[τ]+1)^-3;
g = 1/λ K κ1κ2κ3 ((y[τ]+1)^(-1/2) - (y[τ]+1)^-2);
yddot=D[ydot,τ];
FullSimplify[(yddot+f ydot+g)/.{x'[τ]→xdot, y'[τ]→ydot}/.{κ2→λ(c3d1-c1d3), κ3→λ(c1d2-c2d1)}]
```

Out[243]= 0

Compute the integrals $F(x) = \int_0^x f(y)\,dy$ and $G(x) = \int_0^x g(y)\,dy$ in the proof of Proposition 7. Then verify $F = \alpha G^2 + \beta G$.

In[244]:=
```
F = Assuming[x>0, Integrate[f/.{y[τ]→y}, {y,0,x}]]
G = Assuming[x>0, Integrate[g/.{y[τ]→y}, {y,0,x}]]
Simplify[α G^2+β G-F/.{α→-λ^2/4 c1κ1/(K κ1κ2κ3)^2, β→-3λ/2 c1κ1/(K κ1κ2κ3)}/.{κ2→c1κ1/(K d2), κ3→c1κ1/(4K d3)}]
```

Out[244]= $-x\,c_1\,\kappa_1 + \frac{1}{2} K \left(2 - \frac{2}{\sqrt{1+x}}\right) d_2\,\kappa_2 + \frac{K\,x\,(2+x)\,d_3\,\kappa_3}{(1+x)^2}$

Out[245]= $\frac{K\left(-2 - 3x + 2\sqrt{1+x} + 2x\sqrt{1+x}\right)\kappa_1\,\kappa_2\,\kappa_3}{(1+x)\,\lambda}$

Out[246]= 0

6 Zigzag

Find the positive equilibrium.

In[247]:=
```
f = y^3-3x y^2+(1+κ)y;
g = -y^3+x y^2+(1-κ)y;
Reduce[f==0 && g==0 && x>0 && y>0 && κ>0, {y,x}]
```

Out[249]= $0 < \kappa < 2$ && $y == \sqrt{2-\kappa}$ && $x == -\frac{-y - y^3 - y\,\kappa}{3\,y^2}$

In[250]:=
```
Simplify[-(-y-y^3-y κ)/(3 y^2)/.{y→Sqrt[2-κ]}]
```

Out[250]= $\frac{1}{\sqrt{2-\kappa}}$

Compute the trace at the equilibrium.

In[251]:=
```
equilibrium={x→1/Sqrt[2-κ], y→Sqrt[2-κ]};
J = D[{f,g},{{x,y}}]/.equilibrium;
Simplify[Tr[J]]
```

Out[253]= $-9 + 5\,\kappa$

Compute the first focal value for $\kappa = \frac{9}{5}$.

In[254]:=

```
f = y^3-3x y^2+(1+κ)y;
g = -y^3+x y^2+(1-κ)y;
omegasubst = {ω→Sqrt[Det[J]]};
shift = {x→x+(x/.equilibrium), y→y+(y/.equilibrium)};
f = f/.shift;
g = g/.shift;
T = {{1,0},{-aa/ω,-bb/ω}};
Tinv = Inverse[T];
Tinvuv = Tinv.{u,v};
FG = (T . {f,g})/ω /.{x→Tinvuv[[1]],y→Tinvuv[[2]]}/.{aa→J[[1,1]],bb→ J[[1,2]]};
F = FG[[1]];
G = FG[[2]];
equil = {u→0,v→0};
Clear[f,g];
m = 1;
derivatives = {};
For[i=0, i≤2m+1, i++, For[j=0, j≤2m+1-i, j++, derivatives = Join[derivatives,
{f_{i,j}→(D[F,{u,i},{v,j}]/((i!)*( j!))/.{u→0,v→0}), g_{i,j}→(D[G,{u,i},{v,j}]/((i!)*( j!))/.{u→0,v→0})}]]];
Simplify[L1/.derivatives/.omegasubst/.{κ→9/5}]
```

Out[271]=

$$\frac{5\pi}{13}$$

Since $L_1 > 0$, the Andronov – Hopf bifurcation is subcritical.
We demonstrate the existence of an
unstable limit cycle for κ slightly smaller than $\frac{9}{5}$.

In[272]:=
```
f = y^3-3x y^2+(1+κ)y;
g = -y^3+x y^2+(1-κ)y;
xytime = {x→x[t], y→y[t]};
kappaspec = {κ→1.79};

tmax = 50; x0 = 2.8; y0 = 0.4;
s = NDSolve[{x'[t]==(f/.xytime/.kappaspec), y'[t]==(g/.xytime/.kappaspec), x[0]==x0, y[0]==y0},
{x,y}, {t,tmax}];
p1 = ParametricPlot[Evaluate[{x[t],y[t]}/.s], {t,0,tmax}, PlotRange→All, PlotStyle→Red];

tmax = 50; x0 = 3.1; y0 = 0.3;
s = NDSolve[{x'[t]==(f/.xytime/.kappaspec), y'[t]==(g/.xytime/.kappaspec), x[0]==x0, y[0]==y0},
{x,y}, {t,tmax}];
p2 = ParametricPlot[Evaluate[{x[t],y[t]}/.s], {t,0,tmax}, PlotRange→All, PlotStyle→Brown];

pequil = ListPlot[{{1/Sqrt[2-κ],Sqrt[2-κ]}/.kappaspec}, PlotStyle→{Orange, PointSize[Large]}];
strpl = StreamPlot[{f,g}/.kappaspec, {x,1,4}, {y,0,1}, LabelStyle→"Subtitle"];
Show[strpl,p1,p2,pequil]
```

Out[284]=