

Some minimal bimolecular mass-action systems with limit cycles

Balázs Boros and Josef Hofbauer

Department of Mathematics, University of Vienna, Austria

This Mathematica Notebook is a supplementary material to the paper which has the same title as this document.

It contains some of the calculations appearing in the paper.

0 The first and the second focal values

Below we collect some functions that will be used later on for computing the first and the second focal values. It is based on the Scholarpedia articles

http://www.scholarpedia.org/article/Andronov-Hopf_bifurcation (by Y. A. Kuznetsov) and

http://www.scholarpedia.org/article/Bautin_bifurcation (by J. Guckenheimer and Y. A. Kuznetsov).

```
In[1]:= Idx[set_, n_] := Module[{seq}, seq = (Table[Count[set, i], {i, n}] /. List -> Sequence);
      seq];

GetDerivatives[f_, equilibrium_] :=
  Module[{derivatives, order, deriv, i, j, k, A, B, CC, DD, EE},
    n = Length[f];
    A = D[f, {{x, y, z}}] /. equilibrium;
    derivatives = {};
    order = 2;
    For[i = 0, i ≤ order, i++,
      For[j = 0, j ≤ order - i, j++, For[k = 0, k ≤ order - i - j, k++,
        deriv = Simplify[D[f, {x, i}, {y, j}, {z, k}] /. equilibrium];
        derivatives = Join[derivatives,
          Simplify[{Fi,j,k → deriv[[1]], Gi,j,k → deriv[[2]], Hi,j,k → deriv[[3]]}]]];
    ];
    B[x_, y_] :=
      Sum[{FIdx[{k,1},n], GIdx[{k,1},n], HIdx[{k,1},n]} x[[k]] y[[1]] /. derivatives, {k, n}, {1, n}];
    (* CC[x_, y_, z_] := Sum[{FIdx[{k,1,m},n], GIdx[{k,1,m},n], HIdx[{k,1,m},n]} x[[k]] y[[1]] z[[m]] /.
      derivatives, {k, n}, {1, n}, {m, n}]; *)
    (* because of the bimolecularity, all derivatives of order 3 and higher vanish *)
    CC[x_, y_, z_] := {0, 0, 0};
```

```

DD[x_, y_, z_, s_] := {0, 0, 0};
EE[x_, y_, z_, s_, t_] := {0, 0, 0};
(* the reason for the double symbols CC,
DD, EE is that C, D, E are protected in Mathematica *)
{A, B, CC, DD, EE}
];

GetHopf[A_] := Module[{a, charpol, A2A1minusA0},
  charpol = Collect[-CharacteristicPolynomial[A, λ], λ];
  {a0, a1, a2, a3} = Simplify[CoefficientList[charpol, λ]];
  A2A1minusA0 = Simplify[a1 a2 - a0];
  {charpol, a0, a1, a2, a3, A2A1minusA0}
];

GetEigvectors[A_, om_] := Module[{n, mtx, pconj, q, qconj, normalize},
  n = Length[A];
  mtx = A - om I IdentityMatrix[n];
  q = NullSpace[mtx[[Range[1, n - 1]]]] [[1]];
  (* Notice that q is not normalised. The normalisation has relevance
  only when the second focal value is computed for parameter values,
  where the first focal value does not vanish. *)
  mtx = A^T - om I IdentityMatrix[n];
  pconj = NullSpace[mtx[[Range[1, n - 1]]]] [[1]];
  normalize = FullSimplify[pconj.q];
  pconj = pconj / normalize;
  qconj = FullSimplify[ComplexExpand[q*]];
  {pconj, q, qconj}
];

GetL1[A_, B_, CC_] :=
Module[{n, pconj, q, qconj, v1, v2, v3, c1, numer, denom, a, b, c, d, L1κω},
  n = Length[A];
  {pconj, q, qconj} = GetEigvectors[A, ω];
  v1 = CC[q, q, qconj];
  v2 = Simplify[B[q, Inverse[-A].B[q, qconj]]];
  v3 = Simplify[B[qconj, Inverse[2 I ω IdentityMatrix[n] - A].B[q, q]]];
  c1 = Simplify[pconj. (1/2 v1 + v2 + 1/2 v3)];
  (* We take the real part in a bit complicated way:  $\text{Re } \frac{a+bi}{c+di} = \frac{ac+bd}{c^2+d^2}$ .
  It seems to be faster than the standard solution would be. *)
  numer = Numerator[c1];
  denom = Denominator[c1];
  a = Simplify[ComplexExpand[Re[numer]]];
  b = Simplify[ComplexExpand[Im[numer]]];
  c = Simplify[ComplexExpand[Re[denom]]];
  d = Simplify[ComplexExpand[Im[denom]]];

```

```

L1 $\omega$  = Simplify[ $\frac{a c + b d}{c^2 + d^2}$ ];

L1 $\omega$ 
];

GetL2[A_, B_, CC_, DD_, EE_] :=
Module[{n, Id, omega, invA, inv2, inv3, pconj, q, qconj, h, prec, c, invbig},
  n = Length[A];
  Id = IdentityMatrix[n];
  omega =  $\sqrt{\frac{\text{Det}[A]}{\text{Tr}[A]}}$ ;
  invA = Inverse[A];
  inv2 = Simplify[Inverse[2 omega I Id - A]];
  inv3 = Simplify[Inverse[3 omega I Id - A]];
  {pconj, q, qconj} = GetEigvectors[A, omega];
  q = FullSimplify[q /. { $\omega \rightarrow \text{omega}$ }];
  pconj = FullSimplify[pconj /. { $\omega \rightarrow \text{omega}$ }];
  h2,0 = FullSimplify[inv2.B[q, q]];
  h1,1 = FullSimplify[-invA.B[q, q*]];
  prec = FullSimplify[CC[q, q, q*] + 2 B[q, h1,1] + B[q*, h2,0]];
  c1 = FullSimplify[1 / 2 (pconj.prec)];
  invbig =
    FullSimplify[Inverse[Join[Join[omega I Id - A, {q}^T, 2], {Join[pconj, {0}]}]]];
  h2,1 = FullSimplify[invbig.Join[FullSimplify[prec - 2 c1 q], {0}]] [[1 ;; n]];
  h3,0 = FullSimplify[inv3.(CC[q, q, q] + 3 B[q, h2,0])];
  h3,1 = FullSimplify[inv2.(DD[q, q, q, q*] + 3 CC[q, q, h1,1] +
    3 CC[q, q*, h2,0] + 3 B[h2,0, h1,1] + B[q*, h3,0] + 3 B[q, h2,1] - 6 c1 h2,0)];
  h2,2 = FullSimplify[-invA.(DD[q, q, q*, q*] + 4 CC[q, q*, h1,1] + CC[q*, q*, h2,0] +
    CC[q, q, h2,0*] + 2 B[h1,1, h1,1] + 2 B[q, h2,1*] + 2 B[q*, h2,1] + B[h2,0*, h2,0])];
  c2 = FullSimplify[
    1 / 12 (pconj.(EE[q, q, q, q*, q*] + DD[q, q, q, h2,0*] + 3 DD[q, q*, q*, h2,0] +
      6 DD[q, q, q*, h1,1] + CC[q*, q*, h3,0] + 3 CC[q, q, h2,1*] + 6 CC[q, q*, h2,1] +
      3 CC[q, h2,0*, h2,0] + 6 CC[q, h1,1, h1,1] + 6 CC[q*, h2,0, h1,1] + 2 B[q*, h3,1] +
      3 B[q, h2,2] + B[h2,0*, h3,0] + 3 B[h2,1*, h2,0] + 6 B[h1,1, h2,1]))];
  ComplexExpand[Re[c2]]
];

```

3 Feinberg-Berner oscillator

Theorem 10

```
In[7]:= f =  $\kappa_1 x y \{-1, -1, 1\} + \kappa_2 z \{1, 1, -1\} + \kappa_3 z \{1, 0, -1\} + \kappa_4 x \{-1, 0, 1\} +$   

 $\kappa_5 x \{1, 0, 0\} + \kappa_6 x^2 \{-1, 0, 0\} + \kappa_7 z \{0, 1, -1\} + \kappa_8 y \{0, -1, 1\};$ 
```

(ii) The system is competitive.

After performing the $w = -z$ coordinate transformation, all the off-diagonal entries are negative everywhere on $R_+ \times R_+ \times R_-$.

```
In[8]:= Print[MatrixForm[D[(f /. {z → -w})] {1, 1, -1}, {{x, y, w}}]]];
```

$$\begin{pmatrix} -y \kappa_1 - \kappa_4 + \kappa_5 - 2 x \kappa_6 & -x \kappa_1 & -\kappa_2 - \kappa_3 \\ -y \kappa_1 & -x \kappa_1 - \kappa_8 & -\kappa_2 - \kappa_7 \\ -y \kappa_1 - \kappa_4 & -x \kappa_1 - \kappa_8 & -\kappa_2 - \kappa_3 - \kappa_7 \end{pmatrix}$$

(iv) The unique positive equilibrium is (1, 1, 1).

```
In[9]:= equilibrium = {x → 1, y → 1, z → 1};  

 $\kappa_{\text{subst}} = \text{Solve}[(f /. \text{equilibrium}) == 0, \{\kappa_3, \kappa_5, \kappa_7\}] [[1]];$   

Print[ $\kappa_{\text{subst}}$ ];
```

$$\{\kappa_3 \rightarrow \kappa_4, \kappa_5 \rightarrow \kappa_1 - \kappa_2 + \kappa_6, \kappa_7 \rightarrow \kappa_1 - \kappa_2 + \kappa_8\}$$

(v) Linear stability for $\kappa_2 + 2 \kappa_4 \geq \kappa_1$.

```
In[12]:= g = f /.  $\kappa_{\text{subst}}$ ;  

J = D[g, {{x, y, z}}] /. equilibrium;  

{charpol, a0, a1, a2, a3, A2A1minusA0} = GetHopf[J];  

Print["Jacobian matrix: ", MatrixForm[J]];  

Print["coefficients of the characteristic polynomial  $\lambda^3 + a_2 \lambda^2 + a_1 \lambda + a_0$ :"];  

Print["    a2 =", a2];  

Print["    a1 =", a1];  

Print["    a0 =", a0];  

linstable = Reduce[Expand[A2A1minusA0] > 0 &&  $\kappa_1 \leq \kappa_2 + 2 \kappa_4$  &&  

 $\kappa_1 > 0$  &&  $\kappa_2 > 0$  &&  $\kappa_4 > 0$  &&  $\kappa_6 > 0$  &&  $\kappa_8 > 0$  &&  $\kappa_1 == 1$ , { $\kappa_2, \kappa_4, \kappa_6, \kappa_8, \kappa_1$ }];  

(* setting  $\kappa_1=1$  does not restrict generality,  

but reduces the number of unknowns and thus helps Reduce *)  

Print["all eigenvalues have negative real part: ", linstable];
```

Jacobian matrix:
$$\begin{pmatrix} -\kappa_2 - \kappa_4 - \kappa_6 & -\kappa_1 & \kappa_2 + \kappa_4 \\ -\kappa_1 & -\kappa_1 - \kappa_8 & \kappa_1 + \kappa_8 \\ \kappa_1 + \kappa_4 & \kappa_1 + \kappa_8 & -\kappa_1 - \kappa_4 - \kappa_8 \end{pmatrix}$$

coefficients of the characteristic polynomial $\lambda^3 + a_2 \lambda^2 + a_1 \lambda + a_0$:

$$a_2 = 2 \kappa_1 + \kappa_2 + 2 \kappa_4 + \kappa_6 + 2 \kappa_8$$

$$a_1 = -\kappa_1^2 + \kappa_1 (\kappa_2 + 2 (\kappa_4 + \kappa_6)) + 2 (\kappa_2 + \kappa_6) \kappa_8 + \kappa_4 (\kappa_6 + 3 \kappa_8)$$

$$a_0 = \kappa_4 (\kappa_6 \kappa_8 + \kappa_1 (\kappa_6 + \kappa_8))$$

all eigenvalues have negative real part:

$$\left(\left(0 < \kappa_2 < 1 \ \&\& \ \kappa_4 \geq \frac{1}{2} (1 - \kappa_2) \ \&\& \ \kappa_6 > 0 \ \&\& \ \kappa_8 > 0 \right) \mid \mid \left(\kappa_2 \geq 1 \ \&\& \ \kappa_4 > 0 \ \&\& \ \kappa_6 > 0 \ \&\& \ \kappa_8 > 0 \right) \right) \ \&\& \ \kappa_1 = 1$$

(vi)(a) The Routh-Hurwitz criterion.

In[22]:=

```
h = Expand[25 (a2 a1 - a0) /. {κ1 → 1, κ2 → 1/5, κ4 → 1/5}];
Print["h (κ6, κ8) = ", h];
```

$$h(\kappa_6, \kappa_8) = -26 + 128 \kappa_6 + 55 \kappa_6^2 + 40 \kappa_8 + 260 \kappa_6 \kappa_8 + 50 \kappa_6^2 \kappa_8 + 50 \kappa_8^2 + 100 \kappa_6 \kappa_8^2$$

Let us plot the sign of h .


```

In[24]:= κ6ub = κ6 /. Solve[(h /. {κ8 → 0}) == 0 && κ6 > 0][[1]];
κ8subst = Normal[Solve[h == 0 && κ6 > 0 && κ8 > 0][[1]]];
rpl1 = RegionPlot[h < 0, {κ6, 0, 1/5}, {κ8, 0, 1/2},
  PlotStyle → {Darker[Red], Opacity[0.1]}, BoundaryStyle → None, GridLines → Automatic,
  AxesLabel → {Style[κ6, Bold, 21, Black], Style[κ8, Bold, 21, Black]},
  Frame → None, Axes → True,
  PlotLabel → Style["The sign of h for κ1=1, κ2=1/5, κ4=1/5", Bold, 20, Black],
  TicksStyle → Directive[Bold, 14], ImageSize → Medium];
rpl2 = RegionPlot[h > 0, {κ6, 0, 1/5}, {κ8, 0, 1/2},
  PlotStyle → {Darker[Green], Opacity[0.1]}, BoundaryStyle → None];
pl = Plot[κ8 /. κ8subst, {κ6, 0, κ6ub}, PlotStyle → {Blue, Thick}];
txt = Graphics[{Text[Style[
  "unstable\nequilibrium\n(h(κ6,κ8)<0)", Bold, Darker[Red], 20], {0.05, 0.10}],
  Text[
    Style["stable\nequilibrium\n(h(κ6,κ8)>0)", Bold, Darker[Green], 20], {0.15, 0.40}]
  ]};
Show[rpl1, rpl2, pl, txt]

```

Out[30]=

The sign of h for $\kappa_1=1$, $\kappa_2=1/5$, $\kappa_4=1/5$



(vi) (b) Andronov-Hopf bifurcation (compute the first focal value).


Here we compute the first focal value in the special case that is taken in part (vi) of Theorem 10. Then we plot L_1 as a function of κ_6 (along the $h(\kappa_6, \kappa_8) = 0$ curve).

```

In[31]:= {A, B, CC, DD, EE} = GetDerivatives[g /. {κ1 → 1, κ2 →  $\frac{1}{5}$ , κ4 →  $\frac{1}{5}$ }, equilibrium];
L1κω = GetL1[A, B, CC];
omega = Simplify[ $\sqrt{\frac{\text{Det}[A]}{\text{Tr}[A]}}$ ];
L1 = Simplify[L1κω /. {ω → omega}];
L1κ6 = Simplify[L1 /. κ8subst];
Plot[L1κ6, {κ6, 0, κ6ub}, GridLines → Automatic,
  AxesLabel → {Style[κ6, Bold, 21, Black], None},
  PlotLabel → Style["L1 along h(κ6, κ8)=0\n(shown as a fuction of κ6)", Bold, 15, Black]]

```

Out[36]=



Finally, we plot the bifurcation diagram that is also shown in the paper.


In[37]:=

```

κ6degen = Solve[L1κ6 == 0 && 0 < κ6 < κ6ub, κ6][[1]];
rpl1 = RegionPlot[h < 0, {κ6, 0, 1/5}, {κ8, 0, 1/2},
  PlotStyle → {Darker[Red], Opacity[0.1]}, BoundaryStyle → None, GridLines → Automatic,
  AxesLabel → {Style[κ6, Bold, 21, Black], Style[κ8, Bold, 21, Black]},
  Frame → None, Axes → True,
  PlotLabel → Style["The sign of h for κ1=1, κ2=1/5, κ4=1/5", Bold, 21, Black],
  TicksStyle → Directive[Bold, 14], ImageSize → Large];
p11 = Plot[κ8 /. κ8subst, {κ6, 0, κ6 /. κ6degen}, PlotStyle → {Magenta, Thick}];
p12 = Plot[κ8 /. κ8subst, {κ6, κ6 /. κ6degen, κ6ub}, PlotStyle → {Orange, Thick}];
lpl = ListPlot[{κ6 /. κ6degen, κ8 /. κ8subst /. κ6degen},
  PlotStyle → Blue, PlotMarkers → Automatic];
txt2 = Graphics[{
  Text[Style["supercritical\nAndronov-Hopf\n(L1<0)", Bold, Magenta, 20], {0.09, 0.27}],
  Text[Style["degenerate\nAndronov-Hopf\n(L1=0)", Bold, Blue, 20], {0.125, 0.04}],
  Text[Style["subcritical\nAndronov-Hopf\n(L1>0)", Bold, Orange, 20], {0.17, 0.12}]
}];
arrow = Graphics[{Orange, Arrow[{0.181, 0.08}, {0.181, 0.015}]}],
  {Blue, Arrow[{0.15, 0.02}, {0.173, 0.02}]}];
shw = Show[rpl1, rpl2, p11, p12, lpl, txt, txt2, arrow]
(*Export["C:/bboros/Dropbox/dfc1thm/3d/bimolecular_paper/F_RH_L1.jpg", shw];*)

```


Out[44]=



(viii) Compute the second focal value.

In[45]=

```
{A, B, CC, DD, EE} =
  GetDerivatives[g /. {κ1 → 1, κ2 → 1/5, κ4 → 1/5} /. κ8subst /. N[κ6degen, 20], equilibrium];
L2 = GetL2[A, B, CC, DD, EE];
Print["the second focal value: ", L2];
```

```
the second focal value: -0.01487679976244
```

Thus, the second focal value is negative.

Remark

Here we verify the claim that is made in the paper following the proof of Theorem 10. Namely, we show that whenever $\kappa_2 + 2\kappa_4 < \kappa_1$, along the Routh-Hurwitz curve, the first focal value is negative at $\kappa_6 = 0$, while it is positive at $\kappa_8 = 0$. W.l.o.g. we set $\kappa_1 = 1$. The calculation of L_1 might take about four minutes.

```
In[48]:= {A, B, CC, DD, EE} = GetDerivatives[Simplify[g /. {κ1 → 1}], equilibrium];
L1κω = GetL1[A, B, CC];

triangle = 0 < κ2 < 1 && 0 < κ4 <  $\frac{1}{2}(1 - \kappa_2)$ ;
```

Case $\kappa_6 = 0$

```
In[51]:= omega = Simplify[ $\sqrt{\frac{\text{Det}[A]}{\text{Tr}[A]}}$  /. {κ6 → 0}];

L1 = Simplify[Simplify[L1κω /. {κ6 → 0}] /. {ω → omega}];
RH = Simplify[A2A1minusA0 /. {κ1 → 1, κ6 → 0}];
L1simp = Simplify[L1, RH == 0];
Print["at κ6 = 0 ..."];
Print[" ... Routh-Hurwitz: ", Reduce[RH == 0 && triangle && κ8 > 0, {κ2, κ4, κ8}]];
Print[" ... L1 is negative: ",
  Reduce[L1simp < 0 && RH == 0 && triangle && κ8 > 0, {κ2, κ4, κ8}]];
Print[" ... L1 is nonnegative: ",
  Reduce[L1simp ≥ 0 && RH == 0 && triangle && κ8 > 0, {κ2, κ4, κ8}]]];
```

at $\kappa_6 = 0 \dots$

$$\dots \text{Routh-Hurwitz: } 0 < \kappa_2 < 1 \ \&\& \ 0 < \kappa_4 < \frac{1}{2} (1 - \kappa_2) \ \&\&$$

$$\kappa_8 = \frac{2 - 6 \kappa_2 - 2 \kappa_2^2 - 9 \kappa_4 - 7 \kappa_2 \kappa_4 - 6 \kappa_4^2}{4 (2 \kappa_2 + 3 \kappa_4)} + \frac{1}{4} \sqrt{\left(\frac{1}{(2 \kappa_2 + 3 \kappa_4)^2} (4 + 8 \kappa_2 + 12 \kappa_2^2 + 8 \kappa_2^3 + 4 \kappa_2^4 + 12 \kappa_4 + \right.$$

$$\left. 24 \kappa_2 \kappa_4 + 32 \kappa_2^2 \kappa_4 + 28 \kappa_2^3 \kappa_4 + 9 \kappa_4^2 + 38 \kappa_2 \kappa_4^2 + 73 \kappa_2^2 \kappa_4^2 + 12 \kappa_4^3 + 84 \kappa_2 \kappa_4^3 + 36 \kappa_4^4) \right)}$$

$$\dots L_1 \text{ is negative: } 0 < \kappa_2 < 1 \ \&\& \ 0 < \kappa_4 < \frac{1}{2} (1 - \kappa_2) \ \&\&$$

$$\kappa_8 = \frac{2 - 6 \kappa_2 - 2 \kappa_2^2 - 9 \kappa_4 - 7 \kappa_2 \kappa_4 - 6 \kappa_4^2}{4 (2 \kappa_2 + 3 \kappa_4)} + \frac{1}{4} \sqrt{\left(\frac{1}{(2 \kappa_2 + 3 \kappa_4)^2} (4 + 8 \kappa_2 + 12 \kappa_2^2 + 8 \kappa_2^3 + 4 \kappa_2^4 + 12 \kappa_4 + \right.$$

$$\left. 24 \kappa_2 \kappa_4 + 32 \kappa_2^2 \kappa_4 + 28 \kappa_2^3 \kappa_4 + 9 \kappa_4^2 + 38 \kappa_2 \kappa_4^2 + 73 \kappa_2^2 \kappa_4^2 + 12 \kappa_4^3 + 84 \kappa_2 \kappa_4^3 + 36 \kappa_4^4) \right)}$$

$\dots L_1$ is nonnegative: False

Case $\kappa_8 = 0$

In[59]:=

```
omega = Simplify[ $\sqrt{\frac{\text{Det}[A]}{\text{Tr}[A]}}$  /. { $\kappa_8 \rightarrow 0$ }];
L1 = Simplify[Simplify[L1 $\kappa\omega$  /. { $\kappa_8 \rightarrow 0$ }] /. { $\omega \rightarrow \text{omega}$ }];
RH = Simplify[A2A1minusA0 /. { $\kappa_1 \rightarrow 1$ ,  $\kappa_8 \rightarrow 0$ }];
L1simp = Simplify[L1, RH == 0];
Print["at  $\kappa_8 = 0 \dots$ "];
Print[" ... Routh-Hurwitz: ", Reduce[RH == 0 && triangle &&  $\kappa_6 > 0$ , { $\kappa_2$ ,  $\kappa_4$ ,  $\kappa_6$ }]];
Print[" ...  $L_1$  is positive: ",
  Reduce[L1simp > 0 && RH == 0 && triangle &&  $\kappa_6 > 0$ , { $\kappa_2$ ,  $\kappa_4$ ,  $\kappa_6$ }]];
Print[" ...  $L_1$  is nonpositive: ",
  Reduce[L1simp ≤ 0 && RH == 0 && triangle &&  $\kappa_6 > 0$ , { $\kappa_2$ ,  $\kappa_4$ ,  $\kappa_6$ }]];
```

at $\kappa_8 = 0 \dots$

$$\dots \text{Routh-Hurwitz: } 0 < \kappa_2 < 1 \ \&\& \ 0 < \kappa_4 < \frac{1}{2} (1 - \kappa_2) \ \&\& \ \kappa_6 = \frac{-3 - 3 \kappa_2 - 7 \kappa_4 - \kappa_2 \kappa_4 - 2 \kappa_4^2}{2 (2 + \kappa_4)} + \frac{1}{2} \sqrt{\frac{25 + 10 \kappa_2 + \kappa_2^2 + 34 \kappa_4 + 12 \kappa_2 \kappa_4 + 2 \kappa_2^2 \kappa_4 + 21 \kappa_4^2 + 10 \kappa_2 \kappa_4^2 + \kappa_2^2 \kappa_4^2 + 12 \kappa_4^3 + 4 \kappa_2 \kappa_4^3 + 4 \kappa_4^4}{(2 + \kappa_4)^2}}$$

$$\dots L_1 \text{ is positive: } 0 < \kappa_2 < 1 \ \&\& \ 0 < \kappa_4 < \frac{1}{2} (1 - \kappa_2) \ \&\& \ \kappa_6 = \frac{-3 - 3 \kappa_2 - 7 \kappa_4 - \kappa_2 \kappa_4 - 2 \kappa_4^2}{2 (2 + \kappa_4)} + \frac{1}{2} \sqrt{\frac{25 + 10 \kappa_2 + \kappa_2^2 + 34 \kappa_4 + 12 \kappa_2 \kappa_4 + 2 \kappa_2^2 \kappa_4 + 21 \kappa_4^2 + 10 \kappa_2 \kappa_4^2 + \kappa_2^2 \kappa_4^2 + 12 \kappa_4^3 + 4 \kappa_2 \kappa_4^3 + 4 \kappa_4^4}{(2 + \kappa_4)^2}}$$

$\dots L_1$ is nonpositive: False

Remark

Consider the mass-action system (8.5) on page 109 in the book Foundations of Chemical Reaction Network Theory by Martin Feinberg.

(This is the mass-action system we call Feinberg-Berner oscillator.)

Unfortunately, with the rate constants given in the book, the unique positive equilibrium is linearly stable, as opposed to what is claimed there.

```
In[67]:=  $\kappa_{\text{subst}} = \left\{ \kappa_1 \rightarrow 1, \kappa_2 \rightarrow 1, \kappa_3 \rightarrow \frac{8}{100}, \kappa_4 \rightarrow \frac{1}{100}, \kappa_5 \rightarrow \frac{85}{10}, \kappa_6 \rightarrow 1, \kappa_7 \rightarrow 1, \kappa_8 \rightarrow \frac{2}{10} \right\};$ 
equilibrium = Simplify[Solve[(f /.  $\kappa_{\text{subst}}$ ) == 0 && x > 0 && y > 0 && z > 0] [[1]]];
Print["the eigenvalues: ", N[Eigenvalues[D[f /.  $\kappa_{\text{subst}}$ , {{x, y, z}}] /. equilibrium]]];
```

the eigenvalues: {-11.402, -7.70911, -0.0654958}

After we have pointed this mistake out (on page 5 in <https://arxiv.org/pdf/2202.11034v1.pdf>), Martin Feinberg looked into it. He found (and kindly let us know via private communication on February 27, 2022) that the values of κ_3 and κ_4 are accidentally swapped in the book due to a typographical error. Indeed, setting the values of κ_3 and κ_4 correctly, one finds that the equilibrium is unstable, as claimed in Section 8.3 of the book.

```
In[70]:=  $\kappa_{\text{subst}} = \left\{ \kappa_1 \rightarrow 1, \kappa_2 \rightarrow 1, \kappa_3 \rightarrow \frac{1}{100}, \kappa_4 \rightarrow \frac{8}{100}, \kappa_5 \rightarrow \frac{85}{10}, \kappa_6 \rightarrow 1, \kappa_7 \rightarrow 1, \kappa_8 \rightarrow \frac{2}{10} \right\};$ 
equilibrium = Simplify[Solve[(f /.  $\kappa_{\text{subst}}$ ) == 0 && x > 0 && y > 0 && z > 0] [[1]]];
Print["the eigenvalues: ", N[Eigenvalues[D[f /.  $\kappa_{\text{subst}}$ , {{x, y, z}}] /. equilibrium]]];
```

the eigenvalues: {-14.3295, 0.0470024 + 0.0530062 i, 0.0470024 - 0.0530062 i}

Theorem 12

In[73]:=
$$\begin{aligned} f = & \kappa_1 x y \{-1, -1, 1, 1\} + \kappa_2 z w \{1, 1, -1, -1\} + \kappa_3 z \{1, 0, -1, 0\} + \kappa_4 x \{-1, 0, 1, 0\} + \\ & \kappa_5 x w \{1, 0, 0, -1\} + \kappa_6 x^2 \{-1, 0, 0, 1\} + \kappa_7 z \{0, 1, -1, 0\} + \kappa_8 y \{0, -1, 1, 0\}; \\ x_{\text{positive}} = & \kappa_1 > 0 \& \& \kappa_2 > 0 \& \& \kappa_3 > 0 \& \& \kappa_4 > 0 \& \& \kappa_5 > 0 \& \& \kappa_6 > 0 \& \& \kappa_7 > 0 \& \& \kappa_8 > 0; \end{aligned}$$

(i) Find the curve of positive equilibria.

In[75]:= `Print[Reduce[f == 0 && xpositive && x > 0 && y > 0 && z > 0 && w > 0, {x, y, z, w}]];`

$$\begin{aligned} & \kappa_1 > 0 \& \& \kappa_2 > 0 \& \& \kappa_3 > 0 \& \& \kappa_4 > 0 \& \& \kappa_5 > 0 \& \& \kappa_6 > 0 \& \& \kappa_7 > 0 \& \& \kappa_8 > 0 \& \& x > 0 \& \& \\ y = & \frac{x^2 \kappa_2 \kappa_3 \kappa_4 \kappa_6 + x \kappa_2 \kappa_4^2 \kappa_7 + x \kappa_3 \kappa_4 \kappa_5 \kappa_7}{x \kappa_1 \kappa_3^2 \kappa_5 + \kappa_2 \kappa_3 \kappa_4 \kappa_8 + \kappa_3^2 \kappa_5 \kappa_8} \& \& z = \frac{x \kappa_4}{\kappa_3} \& \& w = \frac{x y \kappa_1 - z \kappa_7 + y \kappa_8}{z \kappa_2} \end{aligned}$$

Let us now verify that the formula given in the paper is correct.

In[76]:=
$$\begin{aligned} \text{equilibrium} = & \left\{ x \rightarrow t, y \rightarrow \frac{\kappa_2 \kappa_3 \kappa_6 t + \kappa_7 (\kappa_2 \kappa_4 + \kappa_3 \kappa_5)}{\kappa_1 \kappa_3 \kappa_5 t + \kappa_8 (\kappa_2 \kappa_4 + \kappa_3 \kappa_5)} \frac{\kappa_4}{\kappa_3} t, \right. \\ & \left. z \rightarrow \frac{\kappa_4}{\kappa_3} t, w \rightarrow \frac{\kappa_1 \kappa_3 \kappa_6 t + (\kappa_1 \kappa_4 \kappa_7 + \kappa_3 \kappa_6 \kappa_8)}{\kappa_1 \kappa_3 \kappa_5 t + (\kappa_2 \kappa_4 + \kappa_3 \kappa_5) \kappa_8} t \right\}; \\ \text{Print[Simplify[f /. equilibrium]]}; \end{aligned}$$

{0, 0, 0, 0}

(ii) Existence and uniqueness of positive equilibria.

Suffices to check that $t \mapsto \frac{a t^2 + b t}{c t + d}$ is strictly increasing on $(0, \infty)$ for any positive a, b, c, d .

In[78]:= `Print["the derivative is ", Simplify[D[$\frac{a t^2 + b t}{c t + d}$, t]], ", which is positive"];`

$$\text{the derivative is } \frac{b d + a t (2 d + c t)}{(d + c t)^2}, \text{ which is positive}$$

(iii) Toricity is equivalent to complex balancing.

In[79]:= `Print[Reduce[$\frac{\kappa_2 \kappa_6}{\kappa_1 \kappa_5} = \frac{\kappa_7}{\kappa_8} \& \& x_{\text{positive}}$]];`
`Print[Reduce[$\frac{\kappa_6}{\kappa_5} = \frac{\kappa_1 \kappa_4 \kappa_7 + \kappa_3 \kappa_6 \kappa_8}{(\kappa_2 \kappa_4 + \kappa_3 \kappa_5) \kappa_8} \& \& x_{\text{positive}}$]];`

$$\kappa_3 > 0 \ \&\& \ \kappa_4 > 0 \ \&\& \ \kappa_2 > 0 \ \&\& \ \kappa_5 > 0 \ \&\& \ \kappa_6 > 0 \ \&\& \ \kappa_7 > 0 \ \&\& \ \kappa_8 > 0 \ \&\& \ \kappa_1 = \frac{\kappa_2 \ \kappa_6 \ \kappa_8}{\kappa_5 \ \kappa_7}$$

$$\kappa_2 > 0 \ \&\& \ \kappa_3 > 0 \ \&\& \ \kappa_4 > 0 \ \&\& \ \kappa_5 > 0 \ \&\& \ \kappa_6 > 0 \ \&\& \ \kappa_7 > 0 \ \&\& \ \kappa_8 > 0 \ \&\& \ \kappa_1 = \frac{\kappa_2 \ \kappa_6 \ \kappa_8}{\kappa_5 \ \kappa_7}$$

(v) The Jacobian matrix and its determinant.

```
In[81]:= J = Simplify[D[f[[1 ;; 3]] /. {w -> c - x - y - z}, {{x, y, z}}] /. {x -> 0, y -> 0, z -> 0}];
Print["the Jacobian matrix of the reduced system: ", MatrixForm[J]];
Print["its determinant: ", Simplify[Det[J]]];
```

the Jacobian matrix of the reduced system:
$$\begin{pmatrix} -\kappa_4 + c \ \kappa_5 & 0 & c \ \kappa_2 + \kappa_3 \\ 0 & -\kappa_8 & c \ \kappa_2 + \kappa_7 \\ \kappa_4 & \kappa_8 & -c \ \kappa_2 - \kappa_3 - \kappa_7 \end{pmatrix}$$

its determinant: $c \ (\kappa_2 \ \kappa_4 + \kappa_3 \ \kappa_5) \ \kappa_8$

4 Wilhelm-Heinrich oscillator

Theorem 13

```
In[84]:= f = \kappa_1 x \{1, 0, 0\} + \kappa_2 x y \{-1, 0, 0\} + \kappa_3 y \{0, -1, 0\} + \kappa_4 x \{-1, 0, 1\} + \kappa_5 z \{0, 1, -1\};
```

(i) The unique positive equilibrium.

```
In[85]:= Print[Reduce[
  f == 0 \&\& \kappa_1 > 0 \&\& \kappa_2 > 0 \&\& \kappa_3 > 0 \&\& \kappa_4 > 0 \&\& \kappa_5 > 0 \&\& x > 0 \&\& y > 0 \&\& z > 0, {y, x, z}]]];
```

$$\kappa_1 > 0 \ \&\& \ 0 < \kappa_4 < \kappa_1 \ \&\& \ \kappa_2 > 0 \ \&\& \ \kappa_3 > 0 \ \&\& \ \kappa_5 > 0 \ \&\& \ y = -\frac{-\kappa_1 + \kappa_4}{\kappa_2} \ \&\& \ x = \frac{y \ \kappa_3}{\kappa_4} \ \&\& \ z = \frac{y \ \kappa_3}{\kappa_5}$$

```
In[86]:= equilibrium = {x -> \frac{\kappa_1 - \kappa_4}{\kappa_2} \frac{\kappa_3}{\kappa_4}, y -> \frac{\kappa_1 - \kappa_4}{\kappa_2}, z -> \frac{\kappa_1 - \kappa_4}{\kappa_2} \frac{\kappa_3}{\kappa_5}};
```

(ii) The system is competitive.

After performing the $w = -z$ coordinate transformation, all the off-diagonal entries are nonpositive everywhere on $\mathbb{R}_+ \times \mathbb{R}_+ \times \mathbb{R}_-$.

In[87]:=

```
Print[MatrixForm[D[(f /. {z → -w}) {1, 1, -1}, {{x, y, w}}]]];
```

$$\begin{pmatrix} \kappa_1 - y \kappa_2 - \kappa_4 & -x \kappa_2 & 0 \\ 0 & -\kappa_3 & -\kappa_5 \\ -\kappa_4 & 0 & -\kappa_5 \end{pmatrix}$$

(iv) The Routh-Hurwitz criterion and the first focal value.

In[88]:=

```
{A, B, CC, DD, EE} = GetDerivatives[f, equilibrium];
{charpol, a0, a1, a2, a3, A2A1minusA0} = GetHopf[A];
Print["Jacobian matrix: ", MatrixForm[Simplify[A]]];
Print["characteristic polynomial: ", charpol];
Print["with  $a_i$  being the coefficient of  $\lambda^i$ , the expression  $a_2 a_1 - a_0$  equals ",
      Collect[A2A1minusA0, t]];
Hopf = FullSimplify[Normal[Solve[A2A1minusA0 == 0 &&  $\kappa_3 > 0$ ]][1]];
Print["Routh-Hurwitz: ", Hopf];
```

Jacobian matrix:

$$\begin{pmatrix} 0 & \kappa_3 \left(1 - \frac{\kappa_1}{\kappa_4}\right) & 0 \\ 0 & -\kappa_3 & \kappa_5 \\ \kappa_4 & 0 & -\kappa_5 \end{pmatrix}$$

characteristic polynomial: $\lambda^3 + \lambda \kappa_3 \kappa_5 + \kappa_1 \kappa_3 \kappa_5 - \kappa_3 \kappa_4 \kappa_5 + \lambda^2 (\kappa_3 + \kappa_5)$

with a_i being the coefficient of λ^i , the expression $a_2 a_1 - a_0$ equals $\kappa_3 \kappa_5 (-\kappa_1 + \kappa_3 + \kappa_4 + \kappa_5)$

Routh-Hurwitz: $\{\kappa_1 \rightarrow \kappa_3 + \kappa_4 + \kappa_5\}$

Next, we verify that the first focal value is indeed negative.

In[95]:=

```
L1pqrsω = GetL1[A, B, CC];
omega = Simplify[ $\sqrt{\frac{\text{Det}[A]}{\text{Tr}[A]}}$ ];
L1 = Factor[L1pqrsω /. {ω → omega} /. Hopf];
Print["the first focal value: ", L1];
```

the first focal value: $-\frac{\kappa_2^2 \kappa_5^3}{(\kappa_3^2 + 3 \kappa_3 \kappa_5 + \kappa_5^2) (\kappa_3^2 + 6 \kappa_3 \kappa_5 + \kappa_5^2)}$

Theorem 14

```
In[99]:= f = x w {1, 0, 0, -1} + p x y {-1, 0, 0, 1} +
          q y {0, -1, 0, 1} + r x {-1, 0, 1, 0} + s z {0, 1, -1, 0};
```

(i) Line of equilibria.

```
In[100]:= Print[Simplify[Reduce[
          f == 0 && p > 0 && q > 0 && r > 0 && s > 0 && x > 0 && y > 0 && z > 0 && w > 0, {x, y, z, w}]]];
```

$$s > 0 \&\& r > 0 \&\& q > 0 \&\& p > 0 \&\& x > 0 \&\& y = \frac{r x}{q} \&\& \frac{q y}{s} = z \&\& r + p y = w$$

```
In[101]:= equilibrium = {x -> t, y -> t \frac{r}{q}, z -> t \frac{r}{s}, w -> \left(t \frac{p}{q} + 1\right) r};
Print[Simplify[f /. equilibrium]];
```

```
{0, 0, 0, 0}
```

(ii) The Routh-Hurwitz criterion.

```
In[103]:= c = (x + y + z + w /. equilibrium);
f2 = f[[1 ;; 3]] /. {w -> c - x - y - z};
Clear[c];
{A, B, CC, DD, EE} = GetDerivatives[f2, equilibrium];
{charpol, a0, a1, a2, a3, A2A1minusA0} = GetHopf[A];
Print["Jacobian matrix: ", MatrixForm[Simplify[A]]];
Print["characteristic polynomial: ", charpol];
Print["with  $a_i$  being the coefficient of  $\lambda^i$ , the expression  $a_2 a_1 - a_0$  equals ",
      Collect[A2A1minusA0, t]];
Hopf = FullSimplify[Normal[Solve[A2A1minusA0 == 0][[1]]]];
Print["Routh-Hurwitz: ", Hopf];
```


$$\text{Jacobian matrix: } \begin{pmatrix} -t & -(1+p)t & -t \\ 0 & -q & s \\ r & 0 & -s \end{pmatrix}$$

$$\text{characteristic polynomial: } qrt + qst + rst + prst + (qs + qt + rt + st)\lambda + (q + s + t)\lambda^2 + \lambda^3$$

with a_i being the coefficient of λ^i , the expression $a_2 a_1 - a_0$ equals

$$q^2 s + q s^2 + (q^2 + 2qs - prs + s^2)t + (q + r + s)t^2$$

$$\text{Routh-Hurwitz: } \left\{ p \rightarrow \frac{qs(q+s) + (q+s)^2 t + (q+r+s)t^2}{rst} \right\}$$

(iii) The first focal value.

In[113]:=

```
L1pqrsω = GetL1[A, B, CC];
omega = Simplify[ $\sqrt{\frac{\text{Det}[A]}{\text{Tr}[A]}}$ ];
L1 = Simplify[L1pqrsω /. {ω → omega} /. Hopf];
Print["the first focal value: ", L1];
```

the first focal value:

$$-\left(\left((q+t)(s+t)(s^2+rt+st+q(s+t))((r+s)(s-t)t+q^2(2s+t)+q(2s^2+2st+(r-t)t))\right)/\right. \\ \left.(2r^2t^2(q^4+s^4+9s^3t+9st^2(2r+t)+9q^3(s+t)+5s^2t(r+4t)+t^2(4r^2+5rt+t^2)+\right. \\ \left.9q(s+t)(s^2+4st+t(2r+t))+5q^2(4s^2+9st+t(r+4t))))\right)$$

We omit the factors with purely positive terms.

In[117]:=

```
L1qrst = -((r+s)(s-t)t+q^2(2s+t)+q(2s^2+2st+(r-t)t));
Print[Simplify[CoefficientList[L1qrst, t]]];
```

$$\{-2qs(q+s), -(q+s)(q+r+s), q+r+s\}$$

We now find values of p, q, r, s for which there are two Hopf points, and the sign of the first focal value at those points have opposite sign.

In[119]:=

```
pqrsubst = {p → 8, q → 1, r → 2, s → 1};
roots = Solve[A2A1minusA0 /. {p → 8, q → 1, r → 2, s → 1} == 0];
Print["the two roots: ", roots];
Print["L1 at the 1st root: ", Simplify[L1qrst /. pqrsubst /. roots[[1]]];
Print["L1 at the 2nd root: ", Simplify[L1qrst /. pqrsubst /. roots[[2]]];
```

the two roots: $\left\{ \left\{ t \rightarrow \frac{1}{2} (3 - \sqrt{7}) \right\}, \left\{ t \rightarrow \frac{1}{2} (3 + \sqrt{7}) \right\} \right\}$

L_1 at the 1st root: $-2\sqrt{7}$

L_1 at the 2nd root: $2\sqrt{7}$

Next, we show an instance, where the first focal value vanishes. We analyse this situation further in (vi) below.

In[124]:=

```
pqrstsubst = {p -> 3 (1 + Sqrt[2]), q -> 1, r -> 2, s -> 1, t -> 1 + Sqrt[2]};
Print["a1 a2 - a0 = ", Simplify[A2A1minusA0 /. pqrstsubst]];
Print["L1 = ", Simplify[L1qrst /. pqrstsubst]];
```

$a_1 a_2 - a_0 = 0$

$L_1 = 0$

(v) The first focal value is negative at the lower Hopf point.

Here we check using computer algebra that the first focal value at the lower Hopf point never nonnegative, i.e., it is always negative. For a formal proof, see the paper.

In[127]:=

```
coeffs = Simplify[CoefficientList[A2A1minusA0, t]];
Print[Reduce[
  A2A1minusA0 == 0 && L1qrst ≥ 0 && p > 0 && q > 0 && r > 0 && s > 0 && t > 0 && t < -  $\frac{\text{coeffs}[[2]]}{2 \text{coeffs}[[3]]} \]]];$ 
```

False

(vi) The second focal value.

The computation below might take a little longer, approximately 1-2 minutes. This is because a number of times FullSimplify is employed for involved nonreal expressions with square roots.

In[129]:=

```

pqrsubst = {p → 3 (1 + √2), q → 1, r → 2, s → 1, t → 1 + √2};
{A, B, CC, DD, EE} = GetDerivatives[f2 /. pqrsubst, equilibrium /. pqrsubst];
A = Simplify[A];

omega = Simplify[√(Det[A]/Tr[A])];

L2 = GetL2[A, B, CC, DD, EE];
Print["the second focal value: ", L2, " ≈ ", N[L2]];

```

the second focal value: $-\frac{55}{4116} + \frac{5}{343\sqrt{2}} \approx -0.00305481$

Finally, we plot the bifurcation diagram that is also shown in the paper.


In[135]:=

```

qrsubst = {q → 1, r → 2, s → 1};
rpl1 = RegionPlot[(A2A1minusA0 /. qrsubst) < 0, {t, 0, 4}, {p, 0, 10},
  PlotStyle → {Darker[Red], Opacity[0.1]}, BoundaryStyle → None, GridLines → Automatic,
  AxesLabel → {Style[t, Bold, 21, Black], Style[p, Bold, 21, Black]}, Frame → None, Axes →
  True, PlotLabel → Style["Bifurcation diagram for q=1, r=2, s=1", Bold, 21, Black],
  TicksStyle → Directive[Bold, 14], ImageSize → Large];
rpl2 = RegionPlot[(A2A1minusA0 /. qrsubst) > 0, {t, 0, 4}, {p, 0, 10},
  PlotStyle → {Darker[Green], Opacity[0.1]}, BoundaryStyle → None];
pl1 = Plot[p /. Hopf /. qrsubst, {t, 0, 1 + √2},
  PlotStyle → {Magenta, Thick}, PlotRange → {{0, 10}}];
pl2 = Plot[p /. Hopf /. qrsubst, {t, 1 + √2, 4},
  PlotStyle → {Orange, Thick}, PlotRange → {{0, 10}}];
lpl = ListPlot[{{1 + √2, 3 (1 + √2)}}, PlotStyle → Blue, PlotMarkers → Automatic];
txt = Graphics[{
  Text[Style["unstable\nequilibrium\n(h(p,1,2,1,t)<0)",
    Bold, Darker[Red], 20], {1.5, 9}],
  Text[Style["stable\nequilibrium\n(h(p,1,2,1,t)>0)",
    Bold, Darker[Green], 20], {1.5, 1}],
  Text[Style["supercritical\nAndronov-Hopf\n(L1<0)", Bold, Magenta, 20], {0.6, 4}],
  Text[Style["degenerate\nAndronov-Hopf\n(L1=0)", Bold, Blue, 20], {2.6, 6.2}],
  Text[Style["subcritical\nAndronov-Hopf\n(L1>0)", Bold, Orange, 20], {3.4, 7.75}]
}];
shw = Show[rpl1, rpl2, pl1, pl2, lpl, txt]
(*Export["C:/bboros/Dropbox/dfc1thm/3d/bimolecular_paper/WH_RH_L1.jpg", shw];*)

```

Out[142]=



(vii) The eigenvalues at the corner equilibrium $(0, 0, 0, c)$.

In[143]=

```
Clear[c];
J = Simplify[D[f[[1 ;; 3]] /. {w -> c - x - y - z}, {{x, y, z}}];
Print[
  "the eigenvalues of the reduced system at the corner equilibrium  $E_W = (0, 0, 0, c)$  : ",
  Eigenvalues[J /. {x -> 0, y -> 0, z -> 0}]];

```

the eigenvalues of the reduced system at the corner equilibrium $E_W = (0, 0, 0, c)$: $\{-q, c - r, -s\}$

5 Wilhelm oscillator

Theorem 15

In[146]:=

$$f = \kappa_1 y \{0, 1, 0\} + \kappa_2 x^2 \{-2, 0, 1\} + \kappa_3 y z \{1, -1, 0\} + \kappa_4 z^2 \{0, 0, -2\};$$

(i) The system is competitive.

After performing the $w = -x$ coordinate transformation, all the off-diagonal entries are nonpositive everywhere on $\mathbb{R}_- \times \mathbb{R}_+ \times \mathbb{R}_+$.

In[147]:=

```
Print[MatrixForm[D[(f /. {x → -w}) {-1, 1, 1}, {{w, y, z}}]]];
```

$$\begin{pmatrix} 4 w \kappa_2 & -z \kappa_3 & -y \kappa_3 \\ 0 & \kappa_1 - z \kappa_3 & -y \kappa_3 \\ 2 w \kappa_2 & 0 & -4 z \kappa_4 \end{pmatrix}$$

(ii) The unique positive equilibrium.

In[148]:=

```
equilibrium =  
Simplify[Normal[Solve[f == 0 &&  $\kappa_1 > 0$  &&  $\kappa_2 > 0$  &&  $\kappa_3 > 0$  &&  $\kappa_4 > 0$  &&  $x > 0$  &&  $y > 0$  &&  $z > 0$ ,  
{x, y, z}][[1]],  $\kappa_1 > 0$  &&  $\kappa_3 > 0$ ];  
Print["equilibrium: ", equilibrium];
```

$$\text{equilibrium: } \left\{ x \rightarrow \frac{\sqrt{2} \kappa_1 \sqrt{\frac{\kappa_4}{\kappa_2}}}{\kappa_3}, y \rightarrow \frac{4 \kappa_1 \kappa_4}{\kappa_3^2}, z \rightarrow \frac{\kappa_1}{\kappa_3} \right\}$$

(iii) The first focal value.

In[150]:=

```

reparam = {κ1 → κ2  $\frac{p}{r}$ , κ3 → κ2 p, κ4 → κ2  $\frac{q^2}{2}$ };
f2 = Expand[Simplify[ $\frac{\kappa_3}{\kappa_1 \kappa_2}$  f /. reparam]];
Print["reparametrised ODE: ", MatrixForm[{ẋ, ẏ, ż}], "=", MatrixForm[f2]];
equilibrium2 = Simplify[equilibrium /. reparam, q > 0];
Print["equilibrium: ", equilibrium2];
{A, B, CC, DD, EE} = GetDerivatives[f2, equilibrium2];
{charpol, a0, a1, a2, a3, A2A1minusA0} = GetHopf[A];
Print["Jacobian matrix: ", MatrixForm[A]];
Print["characteristic polynomial: ", charpol];
linstable = Reduce[Expand[A2A1minusA0] > 0 && p > 0 && q > 0 && r > 0];
Print["all eigenvalues have negative real part: ", linstable];
Hopf = Simplify[Normal[Solve[A2A1minusA0 == 0 && q > 0][[1]]]];
L1pqrω = GetL1[A, B, CC];
omega = Simplify[ $\sqrt{\frac{\text{Det}[A]}{\text{Tr}[A]}}$ ];
L1 = Factor[Simplify[L1pqrω /. {ω → omega} /. Hopf]];
Print["the first focal value: ", L1];

```

$$\text{reparametrised ODE: } \begin{pmatrix} \dot{x} \\ \dot{y} \\ \dot{z} \end{pmatrix} = \begin{pmatrix} -2rx^2 + pryz \\ py - pryz \\ rx^2 - q^2rz^2 \end{pmatrix}$$

$$\text{equilibrium: } \left\{ x \rightarrow \frac{q}{r}, y \rightarrow \frac{2q^2}{pr}, z \rightarrow \frac{1}{r} \right\}$$

$$\text{Jacobian matrix: } \begin{pmatrix} -4q & p & 2q^2 \\ 0 & 0 & -2q^2 \\ 2q & 0 & -2q^2 \end{pmatrix}$$

$$\text{characteristic polynomial: } 4pq^3 + 4q^3\lambda + (4q + 2q^2)\lambda^2 + \lambda^3$$

$$\text{all eigenvalues have negative real part: } q > 0 \&\& 0 < p < 4q + 2q^2 \&\& r > 0$$

$$\text{the first focal value: } -\frac{q(4 + 12q + 17q^2 + 11q^3 + 3q^4)r^2}{(1+q)(4+q)(4+8q+q^2)}$$

Theorem 16

In[166]:=

```
f = κ1 y w {0, 1, 0, -1} + κ2 x2 {-2, 0, 1, 1} + κ3 y z {1, -1, 0, 0} + κ4 z2 {0, 0, -2, 2};
```

(ii) The line of positive equilibria.

In[167]:=

```
Print[Simplify[Normal[Solve[f == 0 && κ1 > 0 && κ2 > 0 &&  
κ3 > 0 && κ4 > 0 && x > 0 && y > 0 && z > 0 && w > 0, {x, y, w}][[1]], z > 0]]];
```

$$\left\{ x \rightarrow \sqrt{2} z \sqrt{\frac{\kappa_4}{\kappa_2}}, y \rightarrow \frac{4 z \kappa_4}{\kappa_3}, w \rightarrow \frac{z \kappa_3}{\kappa_1} \right\}$$

In[168]:=

$$\text{equilibrium} = \left\{ x \rightarrow \sqrt{\frac{2 \kappa_4}{\kappa_2}} t, y \rightarrow \frac{4 \kappa_4}{\kappa_3} t, z \rightarrow t, w \rightarrow \frac{\kappa_3}{\kappa_1} t \right\};$$

(iii) The Routh-Hurwitz criterion.

We start by reparametrising the ODE.

In[169]:=

```
f2 = Expand[Simplify[ $\frac{\kappa_3}{\kappa_1 \kappa_2}$  f /. reparam]];
Print["reparametrised ODE: ", MatrixForm[{ẋ, ẏ, ż, ẇ}], "=", MatrixForm[f2]];
equilibrium2 = Simplify[equilibrium /. reparam, q > 0];
Print["equilibrium: ", equilibrium2];
equilibrium2 = equilibrium2 /. {t → 1};
c = (x + y + z + w /. equilibrium2);
f3 = f2[[1 ;; 3]] /. {w → c - x - y - z};
Clear[c];
{A, B, CC, DD, EE} = GetDerivatives[f3, equilibrium2];
{charpol, a0, a1, a2, a3, A2A1minusA0} = GetHopf[A];
Print["Jacobian matrix: ", MatrixForm[A]];
Print["characteristic polynomial: ", charpol];
linstable = Reduce[Expand[A2A1minusA0] > 0 && p > 0 && q > 0 && r > 0];
Print["all eigenvalues have negative real part: ", linstable];
Hopf = FullSimplify[Normal[Solve[A2A1minusA0 == 0 && q > 0][[1]]];
Print["Routh-Hurwitz: ", Hopf];
```

$$\text{reparametrised ODE: } \begin{pmatrix} \dot{x} \\ \dot{y} \\ \dot{z} \\ \dot{w} \end{pmatrix} = \begin{pmatrix} -2rx^2 + pryz \\ pwy - pryz \\ rx^2 - q^2rz^2 \\ rx^2 - pwy + q^2rz^2 \end{pmatrix}$$

$$\text{equilibrium: } \left\{ x \rightarrow qt, y \rightarrow \frac{2q^2t}{p}, z \rightarrow t, w \rightarrow rt \right\}$$

$$\text{Jacobian matrix: } \begin{pmatrix} -4qr & pr & 2q^2r \\ -2q^2 & -2q^2 & -2q^2 - 2q^2r \\ 2qr & 0 & -2q^2r \end{pmatrix}$$

characteristic polynomial:

$$4pq^3r^2 + 4pq^4r^2 + 8q^5r^2 + 4pq^3r^3 + (2pq^2r + 8q^3r + 4q^4r + 4q^3r^2)\lambda + (2q^2 + 4qr + 2q^2r)\lambda^2 + \lambda^3$$

all eigenvalues have negative real part: $(0 < r \leq 1 \&\& q > 0 \&\& p > 0) \mid \mid \left(r > 1 \&\&$

$$\left(\left(0 < q < -r + r^2 \&\& 0 < p < \frac{-4q^2 - 2q^3 - 8qr - 8q^2r - 2q^3r - 4qr^2 - 2q^2r^2}{q + r - r^2} \right) \mid \mid (q \geq -r + r^2 \&\& p > 0) \right) \right)$$

$$\text{Routh-Hurwitz: } \left\{ p \rightarrow -\frac{2q(2+q)(q+(2+q)r+r^2)}{q+r-r^2} \right\}$$

(iv) The first focal value.

This might take 1-2 minutes.

In[185]:=

```
L1pqrw = GetL1[A, B, CC];
omega = Simplify[ $\sqrt{\frac{\text{Det}[A]}{\text{Tr}[A]}}$ ];
L1 = Simplify[L1pqrw /. {w -> omega} /. Hopf];
Print["the first focal value: ", L1];
Print["L1 negative: ", Reduce[L1 < 0 && r > 1 && 0 < q < r (r - 1)]];
Print["L1 vanishes: ", Reduce[L1 == 0 && r > 1 && 0 < q < r (r - 1)]];
Print["L1 positive: ", Reduce[L1 > 0 && r > 1 && 0 < q < r (r - 1)]];
```


the first focal value:

$$\begin{aligned} & \left((q r (2 + q^2 + 3 r + r^2 + 2 q (1 + r)) (8 r^5 (2 - 3 r + r^3) + 8 q r^5 (-4 - 13 r + 14 r^2 + 3 r^3) + \right. \\ & \quad 2 q^7 r (2 - 5 r + r^2 + r^3 + r^4) + 2 q^6 r (10 - 15 r - 26 r^2 + 18 r^3 + 13 r^4 + 4 r^5) + \\ & \quad 2 q^2 r^3 (14 + 19 r - 62 r^2 + 176 r^3 + 160 r^4 + 17 r^5) + 2 q^3 r^2 (6 + 41 r - 98 r^2 + 136 r^3 + 371 r^4 + \\ & \quad 137 r^5 + 11 r^6) + q^4 r (-2 + 51 r - 154 r^2 - 213 r^3 + 493 r^4 + 414 r^5 + 101 r^6 + 6 r^7) + \\ & \quad \left. q^5 (-2 + 25 r + 5 r^2 - 210 r^3 + 15 r^4 + 209 r^5 + 86 r^6 + 16 r^7) \right) / \\ & \left(2 (q + r - r^2) (16 (-1 + r)^2 r^6 + q^6 (1 + 4 r + r^2 - 6 r^3) + 4 q r^5 (6 - 21 r + 2 r^2 + 13 r^3) + \right. \\ & \quad 4 q^2 r^4 (10 - 28 r - 15 r^2 + 31 r^3 + 12 r^4) + q^5 r (10 + 20 r - 27 r^2 - 35 r^3 + 5 r^4 + 3 r^5) + \\ & \quad \left. q^3 r^3 (38 - 85 r - 174 r^2 + 60 r^3 + 84 r^4 + 13 r^5) + q^4 r^2 (31 - 3 r - 130 r^2 - 45 r^3 + 42 r^4 + 12 r^5 + r^6) \right) \end{aligned}$$

L_1 negative: $r > 1 \&\& 0 < q < -r + r^2$

L_1 vanishes: False

L_1 positive: False

(vi) The eigenvalues at the corner equilibria $E_Y = (0, c, 0, 0)$ and $E_W = (0, 0, 0, c)$.

In[192]:=

```
J = Simplify[D[f[[1 ;; 3]] /. {w -> c - x - y - z}, {{x, y, z}}]];
Print["the eigenvalues of the reduced system at the corner equilibrium ..."];
Print[" ...  $E_W = (0, 0, 0, c)$ : ", Eigenvalues[J /. {x -> 0, y -> 0, z -> 0}]];
Print[" ...  $E_Y = (0, c, 0, 0)$ : ", Eigenvalues[J /. {x -> 0, y -> c, z -> 0}]];
```

the eigenvalues of the reduced system at the corner equilibrium ...

... $E_W = (0, 0, 0, c)$: $\{c \kappa_1, 0, 0\}$

... $E_Y = (0, c, 0, 0)$: $\{0, 0, -c \kappa_1\}$