Oscillationsin three-reaction quadraticmass-actionsystems

Murad Banaji, Balázs Boros, Josef Hofbauer

This document contains the computations of the first focal values in Sections 4.4 and 4.5 in the paper titled "Oscillations in three-reaction quadratic mass-action systems".
In all cases, the first focal value turns out to be negative, and hence, all the occurring Andronov-Hopf bifurcations are supercritical, leading to stable limit cycles.

0 The first focal value

Theory

The first focal value for the differential equation

$$\dot{x} = -\omega y + \sum_{i+j\geq 2} \frac{F_{ij}}{i!j!} x^i y^j$$

$$\dot{y} = \omega x + \sum_{i+j\geq 2} \frac{G_{ij}}{i!j!} x^i y^j$$

is given by

$L_1 = F_{30} + F_{12} + G_{03} + G_{21} + \frac{1}{\omega}[F_{11}(F_{20} + F_{02}) - G_{11}(G_{20} + G_{02}) + F_{02}G_{02} - F_{20}G_{20}]$, see e.g. http://www.scholarpedia.org/article/Andronov-Hopf_bifurcation.

Practice

The function **GetDerivatives** brings the ODE to canonical form, and computes the necessary derivatives.

In[1]:=
```
GetDerivatives[fg_, equilibrium_] :=
  Module[{J, xyshift, T, Tinvuv, FG, derivatives, a, b, u, v, i, j},
   J = Simplify[D[fg, {{x, y}}] /. equilibrium];
   xyshift = {x → x + (x /. equilibrium), y → y + (y /. equilibrium)};
   T = {{1, 0}, {-a / ω, -b / ω}};
   Tinvuv = Inverse[T].{u, v};
   FG = (T . fg /. xyshift) /. {x → Tinvuv[[1]], y → Tinvuv[[2]]} /.
     {a → J[[1, 1]], b → J[[1, 2]]};
   derivatives = {};
   For[i = 0, i ≤ 3, i++, For[j = 0, j ≤ 3 - i, j++,
     derivatives =
      Join[derivatives, {F_{i,j} → (D[FG[[1]], {u, i}, {v, j}] /. {u → 0, v → 0}),
        G_{i,j} → (D[FG[[2]], {u, i}, {v, j}] /. {u → 0, v → 0})}]]];
   derivatives];
```

In[2]:=

$$L_1 = F_{3,0} + F_{1,2} + G_{0,3} + G_{2,1} + \frac{1}{\omega}\left(F_{1,1}\left(F_{2,0} + F_{0,2}\right) - G_{1,1}\left(G_{2,0} + G_{0,2}\right) + F_{0,2}\,G_{0,2} - F_{2,0}\,G_{2,0}\right);$$

4 The analysis of quadratic (2,3,2) systems

4.4 Discussion on Theorem 7

Network on the left (four reactions)

In[3]:=
```
fg = κ1 x^2 {1, 0} + κ2 x y {-1, 1} + κ3 y {0, -1} + κ4 y {1, -1};
equil = Simplify[Solve[fg == 0 && {κ1, κ2, κ3, κ4} > 0 && {x, y} > 0, {x, y}]];
Print["positive equilibrium: ", equil];
```

positive equilibrium: $\left\{\left\{x \to \boxed{\frac{\kappa_3+\kappa_4}{\kappa_2} \quad \text{if } \kappa_3>0\,\&\&\,\kappa_4>0\,\&\&\,\kappa_1>0\,\&\&\,\kappa_2>0},\right.\right.$

$\left.\left. y \to \boxed{\frac{\kappa_1\,(\kappa_3+\kappa_4)^2}{\kappa_2^2\,\kappa_3} \quad \text{if } \kappa_3>0\,\&\&\,\kappa_4>0\,\&\&\,\kappa_1>0\,\&\&\,\kappa_2>0}\right\}\right\}$

In[6]:=
```
equilibrium = Normal[equil[[1]]];
J = D[fg, {{x, y}}] /. equilibrium;
trJ = Simplify[Tr[J]];
detJ = Simplify[Det[J]];
hopf = Normal[Solve[trJ == 0 && {κ1, κ2, κ3, κ4} > 0][[1]]];
derivatives = Simplify[GetDerivatives[fg, equilibrium] /. hopf];
L1 = Simplify[L_1 /. derivatives];
Print["det J =  ", detJ];
Print["tr J =  ", trJ];
Print["L_1 =  ", L1];
```

$$\det J = \frac{\kappa_1 (\kappa_3 + \kappa_4)^2}{\kappa_2}$$

$$\mathrm{tr}\, J = \frac{\kappa_1 (\kappa_3 - \kappa_4)(\kappa_3 + \kappa_4)}{\kappa_2 \kappa_3}$$

$$L_1 = -\frac{2 \kappa_1 \kappa_2}{\kappa_3}$$

Network in the middle (trimolecular source)

In[16]:=

```
fg = κ₁ {1, 0} + κ₂ x y^2 {-1, 1} + κ₃ y {0, -1};
equil = Simplify[Solve[fg == 0 && {κ₁, κ₂, κ₃} > 0 && {x, y} > 0, {x, y}]];
Print["positive equilibrium: ", equil];
```

positive equilibrium: $\left\{\left\{x \to \frac{\kappa_3^2}{\kappa_1 \kappa_2} \text{ if } \kappa_1 > 0 \,\&\&\, \kappa_2 > 0 \,\&\&\, \kappa_3 > 0,\ y \to \frac{\kappa_1}{\kappa_3} \text{ if } \kappa_1 > 0 \,\&\&\, \kappa_2 > 0 \,\&\&\, \kappa_3 > 0\right\}\right\}$

In[19]:=

```
equilibrium = Normal[equil[[1]]];
J = D[fg, {{x, y}}] /. equilibrium;
trJ = Simplify[Tr[J]];
detJ = Simplify[Det[J]];
hopf = Normal[Solve[trJ == 0 && {κ₁, κ₂, κ₃} > 0][[1]]];
ωsubst = Simplify[{ω → Sqrt[detJ /. hopf]}];
derivatives = Simplify[GetDerivatives[fg, equilibrium] /. hopf];
L1 = Simplify[L₁ /. derivatives /. ωsubst];
Print["det J =  ", detJ];
Print["tr J =  ", trJ];
Print["L₁ =  ", L1];
```

$$\det J = \frac{\kappa_1^2 \kappa_2}{\kappa_3}$$

$$\mathrm{tr}\, J = -\frac{\kappa_1^2 \kappa_2}{\kappa_3^2} + \kappa_3$$

$$L_1 = -\frac{\kappa_3^3}{\kappa_1^2}$$

Network on the right (tetramolecular target)

In[30]:=
```
fg = κ1 x^2 {1, 1} + κ2 x y {-1, 0} + κ3 y {0, -1};
equil = Simplify[Solve[fg == 0 && {κ1, κ2, κ3} > 0 && {x, y} > 0, {x, y}]];
Print["positive equilibrium: ", equil];
```

positive equilibrium: $\left\{\left\{x \to \frac{\kappa_3}{\kappa_2} \text{ if } \kappa_1 > 0 \,\&\&\, \kappa_2 > 0 \,\&\&\, \kappa_3 > 0,\ y \to \frac{\kappa_1 \kappa_3}{\kappa_2^2} \text{ if } \kappa_1 > 0 \,\&\&\, \kappa_2 > 0 \,\&\&\, \kappa_3 > 0\right\}\right\}$

In[33]:=
```
equilibrium = Normal[equil[[1]]];
J = D[fg, {{x, y}}] /. equilibrium;
trJ = Simplify[Tr[J]];
detJ = Simplify[Det[J]];
hopf = Normal[Solve[trJ == 0 && {κ1, κ2, κ3} > 0][[1]]];
derivatives = Simplify[GetDerivatives[fg, equilibrium] /. hopf];
L1 = Simplify[L1 /. derivatives];
Print["det J =  ", detJ];
Print["tr J =  ", trJ];
Print["L1 =  ", L1];
```

$\det J = \frac{\kappa_1 \kappa_3^2}{\kappa_2}$

$\operatorname{tr} J = \left(-1 + \frac{\kappa_1}{\kappa_2}\right) \kappa_3$

$L_1 = -\frac{2 \kappa_2^2 \kappa_3}{\omega^2}$

4.5 Andronov-Hopf bifurcations

Case 9

In[43]:=
```
κpositive = κ1 > 0 && κ2 > 0 && κ3 > 0;
absubst = {a1 → 2, b1 → 0, a2 → 1, b2 → 1, a3 → 0, b3 → 1};
fg = Sum[κi x^ai y^bi {ci, di}, {i, 1, 3}] /. absubst;
xyμ = Normal[Solve[μ {u1, u2, u3} == ({κ1 x^a1 y^b1, κ2 x^a2 y^b2, κ3 x^a3 y^b3} /. absubst) &&
      κpositive && {x, y} > 0][[1]]];
Print["positive equilibrium and μ: ", xyμ];
cross = Cross[{c1, c2, c3}, {d1, d2, d3}];
usubst = {u1 → cross[[1]], u2 → cross[[2]], u3 → cross[[3]]};
```

positive equilibrium and μ: $\left\{x \to \frac{u_2 \kappa_3}{u_3 \kappa_2},\ y \to \frac{u_2^2 \kappa_1 \kappa_3}{u_1 u_3 \kappa_2^2},\ \mu \to \frac{u_2^2 \kappa_1 \kappa_3^2}{u_1 u_3^2 \kappa_2^2}\right\}$

In[50]:=
```
equilibrium = xyμ[[{1, 2}]];
J = Simplify[D[fg, {{x, y}}] /. equilibrium];
trJ = Simplify[Tr[J]];
detJ = Simplify[Det[J]];
hopf = Solve[trJ == 0, κ1][[1]];
ωsubst = {ω → Sqrt[detJ]};
derivatives = Simplify[GetDerivatives[fg, equilibrium]];
L1 = Simplify[Simplify[L1 /. derivatives] /. ωsubst /. hopf /. usubst];
L1nonneg1 = Reduce[L1 ≥ 0 && c1 > 0 && c2 == -1 && c3 > 0 &&
    d1 > 0 && d2 ≥ -1 && d3 ≥ -1 && 1/2 (d3/c3 + d1/c1) < d2/c2 < d1/c1 && {κ2, κ3} > 0];
L1nonneg2 = Reduce[L1 ≥ 0 && c1 > 0 && c2 == -1 &&
    c3 == 0 && d1 > 0 && d2 ≥ -1 && d3 == -1 && d2/c2 < d1/c1 && {κ2, κ3} > 0];
Print["tr J = 0 for  ", hopf];
Print["L1=  ", L1];
Print["L1 ≥ 0 when c3 > 0:  ", L1nonneg1];
Print["L1 ≥ 0 when c3 = 0:  ", L1nonneg2];
```

tr $J = 0$ for $\left\{\kappa_1 \to -\frac{u_1 (d_2 u_2 + d_3 u_3)\, \kappa_2}{u_2 (2 c_1 u_1 + c_2 u_2)}\right\}$

$L_1 = \Big(\big(2 (c_2 d_1 - c_1 d_2)^2 \big(4 c_1^2 c_3^2 d_2^2 d_3 + c_2^2 d_3 \left(2 c_3^2 d_1^2 + c_1 c_3 d_1 d_3 + c_1^2 d_3^2\right) - c_2 c_3 d_2 \left(c_3^2 d_1^2 + 3 c_1 c_3 d_1 d_3 + 4 c_1^2 d_3^2\right)\big) \kappa_2^2\big) \Big/ \big(c_1 (c_3 d_1 - c_1 d_3)^2 (c_3 d_2 - c_2 d_3) (2 c_1 c_3 d_2 - c_2 (c_3 d_1 + c_1 d_3))\, \kappa_3\big)\Big)$

$L_1 \geq 0$ when $c_3 > 0$: False

$L_1 \geq 0$ when $c_3 = 0$: False

Hence, L_1 is negative.

Case 10

In[64]:=
```
κpositive = κ₁ > 0 && κ₂ > 0 && κ₃ > 0;
absubst = {a₁ → 2, b₁ → 0, a₂ → 1, b₂ → 1, a₃ → 0, b₃ → 0};
fg = Sum[κᵢ x^aᵢ y^bᵢ {cᵢ, dᵢ}, {i, 1, 3}] /. absubst;
equil = Reduce[μ {u₁, u₂, u₃} == ({κ₁ x^a₁ y^b₁, κ₂ x^a₂ y^b₂, κ₃ x^a₃ y^b₃} /. absubst) &&
    κpositive && {x, y} > 0 && {μ, u₁, u₂, u₃} > 0, {x, y, μ}];
Print["positive equilibrium: ", equil];
cross = Cross[{c₁, c₂, c₃}, {d₁, d₂, d₃}];
usubst = {u₁ → cross[[1]], u₂ → cross[[2]], u₃ → cross[[3]]};
```

positive equilibrium:

$$u_1 > 0 \,\&\&\, u_2 > 0 \,\&\&\, u_3 > 0 \,\&\&\, \kappa_1 > 0 \,\&\&\, \kappa_2 > 0 \,\&\&\, \kappa_3 > 0 \,\&\&\, x == \sqrt{\frac{u_1 \kappa_3}{u_3 \kappa_1}} \,\&\&\, y == \frac{x\, u_2 \kappa_1}{u_1 \kappa_2} \,\&\&\, \mu == \frac{x^2 \kappa_1}{u_1}$$

In[71]:=
```
equilibrium = {x → Sqrt[(u₁ κ₃)/(u₃ κ₁)], y → u₂/κ₂ Sqrt[(κ₁ κ₃)/(u₁ u₃)]};
J = Simplify[D[fg, {{x, y}}] /. equilibrium, κpositive];
trJ = Simplify[Tr[J]];
detJ = Simplify[Det[J], κpositive && {u₁, u₂, u₃} > 0];
hopf =
  Normal[Simplify[Solve[trJ == 0 && κpositive && {u₁, u₂, u₃} > 0], {u₁, u₂, u₃} > 0][[1]]];
ωsubst = {ω → Sqrt[detJ]};
Print["tr J = 0  for  ", hopf];
derivatives =
  Simplify[GetDerivatives[fg, equilibrium], κpositive && u₁ > 0 && u₂ > 0 && u₃ > 0];
L1 = Simplify[Simplify[L₁ /. derivatives] /. ωsubst /. hopf,
   κpositive && u₁ > 0 && u₂ > 0 && u₃ > 0];
Print["L₁ =  ", L1];
```

$$\mathrm{tr}\, J = 0 \ \text{ for } \ \left\{c_1 \to -\frac{c_2 u_2}{2 u_1} - \frac{d_2 \kappa_2}{2 \kappa_1}\right\}$$

$$L_1 = \frac{2 u_1 u_3 \left(c_2 u_2 \kappa_1 - d_2 u_1 \kappa_2\right)}{\sqrt{\frac{u_1^5 u_3 \kappa_3}{\kappa_1}}}$$

The expression in the enumerator of L_1 is never nonnegative:

In[81]:=

```
Reduce[Simplify[c2 u2 κ1 - d2 u1 κ2 /. usubst] ≥ 0 &&
  c1 == (-c2 u2/(2 u1) - d2 κ2/(2 κ1) /. usubst) && κpositive && c1 > 0 && c2 == -1 &&
  c3 > 0 && d1 > 0 && d2 == -1 && d3 ≥ 0 && 1/2 (d3/c3 + d1/c1) < d2/c2 < d1/c1]
```

$$\text{Reduce}\left[\text{Simplify}[c_2\,u_2\,\kappa_1 - d_2\,u_1\,\kappa_2 \;/.\; \text{usubst}] \geq 0 \;\&\&\; c_1 == \left(-\frac{c_2\,u_2}{2\,u_1} - \frac{d_2\,\kappa_2}{2\,\kappa_1} \;/.\; \text{usubst}\right) \;\&\&\; \kappa\text{positive} \;\&\&\; c_1 > 0 \;\&\&\; c_2 == -1 \;\&\&\; c_3 > 0 \;\&\&\; d_1 > 0 \;\&\&\; d_2 == -1 \;\&\&\; d_3 \geq 0 \;\&\&\; \frac{1}{2}\left(\frac{d_3}{c_3} + \frac{d_1}{c_1}\right) < \frac{d_2}{c_2} < \frac{d_1}{c_1}\right]$$

Out[81]=

```
False
```

Hence, L_1 is negative.