Adding species to chemical reaction networks: preserving rank preserves nondegenerate behaviours

Murad Banaji, Balázs Boros, Josef Hofbauer

This Mathematica Notebook is a supplementary material to the paper which has the same title as this document. It contains some of the calculations appearing in the paper.

0 Focal values

Derive L_1 and L_2, the first and the second focal values for the differential equation

$$\dot{x} = -y + \sum_{i+j\geq 2} f_{i,j}\, x^i\, y^j,$$

$$\dot{y} = x + \sum_{i+j\geq 2} g_{i,j}\, x^i\, y^j.$$

Theoretical background: Chapter 4 in Dumortier, Llibre, Artés: Qualitative Theory of Planar Differential Systems.

In[1]:=
```
m = 2;
cd = {}; R2cd = {};
For[k = 2, k ≤ 2 m + 1, k++, For[i = 0, i ≤ k, i++,
  {cd = Join[cd, {c_{k,i}, d_{k,i}}], R2cd = Join[R2cd, {R_{k,i} → c_{k,i} + d_{k,i} I}]}]];
coeffsxy =
  CoefficientList[ComplexExpand[Sum[Sum[R_{k,i} z^(k-i) (z*)^i, {i, 0, k}], {k, 2, 2 m + 1}]
   /. R2cd /. {z → x + y I}], {x, y}];
cond = True;
For[k = 2, k ≤ 2 m + 1, k++, For[i = 0, i ≤ k, i++,
  {cond = cond && (f_{i,k-i} == ComplexExpand[Re[coeffsxy[[i + 1, k - i + 1]]]]) &&
  (g_{i,k-i} == ComplexExpand[Im[coeffsxy[[i + 1, k - i + 1]]]])}]];
cd2fg = Solve[cond, cd][[1]];

For[k = 2, k ≤ 2 m + 1, k++, R_k = Sum[R_{k,i} z^(k-i) w^i, {i, 0, k}]];

(* F[i,j] computes the polynomial F_i(h_j) *)
F[i_, j_] := Module[{coeffs, M, mtx},
   coeffs = CoefficientList[D[R_i h_j, {z, 1}], {z, w}];
   M = Dimensions[coeffs][[1]] - 1;
   mtx = (coeffs + Transpose[coeffs*])
     Table[If[k + l == M && k ≠ l, 1/(k - l), 0], {k, 0, M}, {l, 0, M}];
   I z^Range[0,M].mtx.w^Range[0,M]];

h_0 = 1;
For[k = 1, k ≤ 2 m - 1, k++, h_k = Sum[F[k + 1 - l, l], {l, 0, k - 1}]];

(* H[k,j] computes H_k(h_j), note that one of k and j is even,
the other one is odd in all of the interesting cases *)
H[k_, j_] := Module[{},
   coeffs = CoefficientList[h_j, {z, w}];
   Sum[Coefficient[R_k, z^a w^(k-a)] × coeffs[[((k - 2 a + 1) + j)/2 + 1, (j - (k - 2 a + 1))/2 + 1]],
    {a, (k + 1 - j)/2, (k + 1 + j)/2}]];
For[j = 1, j ≤ m, j++, L_j = Simplify[ComplexExpand[
     2 π Re[Sum[H[2 j + 1 - l, l], {l, 0, 2 j - 1}]] /. R2cd /. cd2fg]]];
```

Display the first focal value, L_1. The second focal value, L_2, is somewhat longer. Important note: $f_{i,j}$ and $g_{i,j}$ include the division by $i!\,j!$, so they are the Taylor coefficients (not simply the respective derivatives).

In[14]:=
```
Print["𝐿₁ = ", L₁];
```

$$L_1 = \frac{1}{4}\pi\,(f_{1,2} + f_{1,1}\,f_{2,0} + 3\,f_{3,0} + f_{0,2}\,(f_{1,1} + 2\,g_{0,2}) + 3\,g_{0,3} - g_{0,2}\,g_{1,1} - 2\,f_{2,0}\,g_{2,0} - g_{1,1}\,g_{2,0} + g_{2,1})$$

Good practice: compute the focal values once, and save them in a file. When needed, load from that file.

In[15]:=
```
L1 = L₁; L2 = L₂; (* when storing in a file, better avoiding subscripts *)
path = "C://bboros/Dropbox/murad/lifting/tex/computations/focal_values.mx";
DumpSave[path, {L1, L2}];
Protect[path];
Off[Remove::rmptc];
Remove["Global`*"]; (* clear all variables *)
On[Remove::rmptc];
Unprotect[path];
```

Define a module that computes the necessary partial derivatives. To be used later.

In[23]:=
```
GetDerivatives[fg_, equilibrium_, m_] := Module[{J, xyshift, T, Tinvuv, FG, derivatives},
  J = Simplify[D[fg, {{x, y}}] /. equilibrium];
  xyshift = {x → x + (x /. equilibrium), y → y + (y /. equilibrium)};
  T = {{1, 0}, {-a / ω, -b / ω}};
  Tinvuv = Inverse[T].{u, v};
  FG =
   (T . fg /. xyshift) / ω /. {x → Tinvuv[[1]], y → Tinvuv[[2]]} /. {a → J[[1, 1]], b → J[[1, 2]]};
  derivatives = {};
  For[i = 0, i ≤ 2 m + 1, i++, For[j = 0, j ≤ 2 m + 1 - i, j++,
    derivatives =
     Join[derivatives, Simplify[{f_{i,j} → (D[FG[[1]], {u, i}, {v, j}] / ((i!) * (j!)) /. {u → 0, v → 0}),
        g_{i,j} → (D[FG[[2]], {u, i}, {v, j}] / ((i!) * (j!)) /. {u → 0, v → 0})}]]]];
  derivatives];
```

4.1 Schlögl model

On any compact subinterval of $(0, \infty)$, the vectorfield $-x^3 + 6\,x^2\,(1 - \epsilon\,x) - 11\,x\,(1 - \epsilon\,x) + 6\,(1 - \epsilon\,x)^2$ converges uniformly to $-x^3 + 6\,x^2 - 11\,x + 6$ as $\epsilon \to 0$.

In[24]:=
```
eps = 10^-4;
feps = -x^3 + 6 x^2 (1 - ϵ x) - 11 x (1 - ϵ x) + 6 (1 - ϵ x)^2;
Plot[{feps /. {ϵ → 0}, feps /. {ϵ → eps}}, {x, 0, 4},
 PlotStyle → {Orange, {Dashed, Blue}}, PlotRange → All, PlotLegends →
  Placed[{feps /. {ϵ → 0}, "-x^3 + 6x^2 (1-ϵx) - 11x(1-ϵx) + 6(1-ϵx)^2 with ϵ = " <>
     ToString[TraditionalForm[eps]]}, {0.55, 0.8}],
 LabelStyle → Directive[FontSize → 16], AxesLabel → {x,},
 GridLines → Automatic, ImageSize → Large]
```

Out[26]=



4.2 Lotka-Volterra Autocatalator (LVA)

Find the positive equilibrium of the LVA with one reversible reaction (when it exists).

In[27]:=
```
fg = κ1 x^2 {1, 0} + κ2 x^3 {-1, 0} + κ3 x y {-1, 1} + κ4 y {0, -1};
κpositive = κ1 > 0 && κ2 > 0 && κ3 > 0 && κ4 > 0;
xypositive = x > 0 && y > 0;
Print["positive equilibrium: ", Simplify[Reduce[fg == 0 && κpositive && xypositive]]];
```

positive equilibrium: $\kappa_1 > 0 \,\&\&\, \kappa_2 > 0 \,\&\&\, \kappa_3 > 0 \,\&\&\, 0 < \kappa_4 < \frac{\kappa_1 \kappa_3}{\kappa_2} \,\&\&\, y == \frac{\kappa_4 (\kappa_1 \kappa_3 - \kappa_2 \kappa_4)}{\kappa_3^3} \,\&\&\, x == \frac{\kappa_4}{\kappa_3}$

Check the sign of the determinant and the trace of the Jacobian matrix at the positive equilibrium. The determinant is always positive, while the trace changes sign at $\kappa_1 \kappa_3 = \kappa_2 \kappa_4$.

In[31]:=
```
equilibrium = {x → κ4/κ3, y → κ4 (κ1 κ3 - κ2 κ4)/κ3^3};
J = D[fg, {{x, y}}] /. equilibrium;
Print["det J = ", Simplify[Det[J]]];
omegasubst = Simplify[{ω → Sqrt[Det[J]]}, κpositive];
Print["tr J = ", Simplify[Tr[J]]];
```

$$\det J = \frac{\kappa_4^2 (\kappa_1 \kappa_3 - \kappa_2 \kappa_4)}{\kappa_3^2}$$

$$\operatorname{tr} J = \frac{\kappa_4 (\kappa_1 \kappa_3 - 2 \kappa_2 \kappa_4)}{\kappa_3^2}$$

When the trace vanishes, compute the first focal value. It is negative, the Andronov-Hopf bifurcation is supercritical.

In[36]:=
```
tracevanish = {κ4 → κ1 κ3/(2 κ2)};
derivatives = GetDerivatives[fg, equilibrium, 1];
derivativessimp = Simplify[derivatives /. omegasubst /. tracevanish, κpositive];
Get[path];
Print["L1 = ", Simplify[L1 /. derivativessimp, κpositive]];
```

$$L_1 = -\frac{\sqrt{2}\, \pi\, \kappa_2^2}{\sqrt{\kappa_1^3 \kappa_3}}$$

Now homogenise the LVA and find the curve of positive equilibria.

In[41]:=
```
fgh = κ1 x^2 z {1, 0, -1} + κ2 x^3 {-1, 0, 1} + κ3 x y {-1, 1, 0} + κ4 y {0, -1, 1};
κpositive = κ1 > 0 && κ2 > 0 && κ3 > 0 && κ4 > 0;
xyzpositive = x > 0 && y > 0 && z > 0;
Print["positive equilibria: ",
  Simplify[Reduce[fgh == 0 && κpositive && xyzpositive, {x, y, z}]]];
```

positive equilibria: $\kappa_1 > 0 \;\&\&\; \kappa_2 > 0 \;\&\&\; \kappa_3 > 0 \;\&\&\; \kappa_4 > 0 \;\&\&\; x == \frac{\kappa_4}{\kappa_3} \;\&\&\; y > 0 \;\&\&\; z == \frac{x \kappa_2 + \frac{y \kappa_3}{x}}{\kappa_1}$

Parametrise the curve of positive equilibria by $t > 0$. For each positive equilibrium, restrict the dynamics to its positive stoichiometric class and eliminate z. Check the sign of the determinant and the trace of the Jacobian matrix at the positive equilibrium. The determinant is always positive, while the trace changes sign at $t = \frac{(\kappa_1 + \kappa_2)\kappa_4^2}{\kappa_3^3}$.

In[45]:=
```
equilibrium = {x → κ4/κ3, y → t, z → (κ4/κ3) (κ2/κ1) + (κ3/κ1) (κ3/κ4) t};
fg = fgh[[1 ;; 2]] /. {z → (x + y + z /. equilibrium) - x - y};
J = D[fg, {{x, y}}] /. equilibrium;
Print["det J = ", Simplify[Det[J]]];
omegasubst = Simplify[{ω → Sqrt[Det[J]]}, κpositive];
Print["tr J = ", Simplify[Tr[J]]];
```

$$\det J = \frac{t\,\kappa_4\left(\kappa_3^2+\kappa_1\kappa_4\right)}{\kappa_3}$$

$$\mathrm{tr}\,J = t\,\kappa_3 - \frac{(\kappa_1+\kappa_2)\,\kappa_4^2}{\kappa_3^2}$$

When the trace vanishes, compute the first focal value. It is negative, the Andronov-Hopf bifurcation is supercritical.

In[51]:=
```
tracevanish = {t → (κ1 + κ2) κ4^2/κ3^3};
derivatives = GetDerivatives[fg, equilibrium, 1];
derivativessimp = Simplify[derivatives /. omegasubst /. tracevanish, κpositive];
Get[path];
Print["L1 = ", Simplify[L1 /. derivativessimp, κpositive]];
```

$$L_1 = -\frac{\pi\sqrt{\kappa_1+\kappa_2}\,\kappa_3^2\left(2\kappa_3^2+\kappa_1\kappa_4\right)}{4\left(\kappa_4\left(\kappa_3^2+\kappa_1\kappa_4\right)\right)^{3/2}}$$

4.3 Lotka reactions

The ODE of the homogenised Lotka reactions has negative divergence on the positive orthant $\mathbb{R}^3_+$. Thus, there is no periodic solution.

In[56]:=
```
fgh = κ1 x z {1, 0, -1} + κ2 x y {-1, 1, 0} + κ3 y {0, -1, 1};
Print["divergence = ", Simplify[Div[fgh/(x y z), {x, y, z}]]];
```

$$\text{divergence} = -\frac{\kappa_3}{x\,z^2}$$

The unique positive equilibrium in the positive stoichiometric class $x+y+z=c$ with $c>\frac{\kappa_3}{\kappa_2}$ is asymptotically stable.

In[58]:=
```
equilibrium = {x → κ3/κ2, y → κ1/κ2 t, z → t};
fg = fgh[[1 ;; 2]] /. {z → (x + y + z /. equilibrium) - x - y};
J = D[fg, {{x, y}}] /. equilibrium;
Print["det J = ", Simplify[Det[J]]];
Print["tr J = ", Simplify[Tr[J]]];
```

$\det J = \frac{t\,\kappa_1\,(\kappa_1+\kappa_2)\,\kappa_3}{\kappa_2}$

$\mathrm{tr}\,J = -\frac{\kappa_1\,\kappa_3}{\kappa_2}$

Since each of the corner equilibria $(c, 0, 0)$ and $(0, 0, c)$ is a saddle (assuming $c > \kappa_3/\kappa_2$), the ω - limit set of each positive initial condition is disjoint from the boundary. Thus, by the Poincaré-Bendixson Theorem, the system is permanent.

In[63]:=
```
fg = fgh[[1 ;; 2]] /. {z → c - x - y};
J = D[fg, {{x, y}}];
Print["at the corner equilibrium (c, 0), det J = ",
  Simplify[Det[J] /. {x → c, y → 0}]];
Print["at the corner equilibrium (0, 0), det J = ", Simplify[Det[J] /. {x → 0, y → 0}]];
```

at the corner equilibrium $(c, 0)$, $\det J = c\,\kappa_1\,(-c\,\kappa_2+\kappa_3)$

at the corner equilibrium $(0, 0)$, $\det J = -c\,\kappa_1\,\kappa_3$

4.4 Brusselator

We start with verifying for the Brusselator that the Andronov-Hopf bifurcation is supercritical.

In[67]:=
```
fg = κ1 {1, 0} + κ2 x {-1, 0} + κ3 x {-1, 1} + κ4 x^2 y {1, -1};
κpositive = κ1 > 0 && κ2 > 0 && κ3 > 0 && κ4 > 0;
equilibrium = {x → κ1/κ2, y → (κ2 κ3)/(κ1 κ4)};
J = D[fg, {{x, y}}] /. equilibrium;
Print["det J = ", Det[J]];
omegasubst = Simplify[{ω → Sqrt[Det[J]]}, κpositive];
Print["tr J = ", Tr[J]];
```

$$\det J = \frac{\kappa_1^2 \kappa_4}{\kappa_2}$$

$$\mathrm{tr}\, J = -\kappa_2 + \kappa_3 - \frac{\kappa_1^2 \kappa_4}{\kappa_2^2}$$

In[74]:=

```
tracevanish = {κ3 → κ2 + κ1^2 κ4/κ2^2};
derivatives = GetDerivatives[fg, equilibrium, 1];
derivativessimp = Simplify[derivatives /. omegasubst /. tracevanish, κpositive];
Get[path];
Print["L1 = ", Simplify[L1 /. derivativessimp, κpositive]];
```

$$L_1 = -\frac{\pi \left(2 \kappa_2^4 + \kappa_1^2 \kappa_2 \kappa_4\right)}{4 \kappa_1^3 \sqrt{\kappa_2 \kappa_4}}$$

Homogenise the Brusselator and observe first that $(0, c, 0)$ is an equilibrium on the boundary for every $c > 0$.

Since the two eigenvalues within the triangle $\{(x, y, z) \in \mathbb{R}^3_{\geq 0} : x + y + z = c\}$ are negative reals, the corner equilibrium $(0, c, 0)$ is asymptotically stable relative to its positive stoichiometric class. Therefore, the homogenised Brusselator is not permanent.

In[79]:=

```
fgh = κ1 z {1, 0, -1} + κ2 x {-1, 0, 1} + κ3 x {-1, 1, 0} + κ4 x^2 y {1, -1, 0};
eigs = Eigenvalues[D[fgh, {{x, y, z}}] /. {x → 0, z → 0}];
Print["eigenvalues: ", eigs];
Print["two negative real eigenvalues for ",
  Reduce[eigs[[2]] < 0 && eigs[[3]] < 0 && κpositive]];
```

eigenvalues:

$$\left\{0, \frac{1}{2}\left(-\kappa_1 - \kappa_2 - \kappa_3 - \sqrt{-4 \kappa_1 \kappa_3 + (\kappa_1 + \kappa_2 + \kappa_3)^2}\right), \frac{1}{2}\left(-\kappa_1 - \kappa_2 - \kappa_3 + \sqrt{-4 \kappa_1 \kappa_3 + (\kappa_1 + \kappa_2 + \kappa_3)^2}\right)\right\}$$

two negative real eigenvalues for $\kappa_1 > 0$ && $\kappa_2 > 0$ && $\kappa_3 > 0$ && $\kappa_4 > 0$

Let us now turn to the positive equilibria of the homogenised Brusselator.

In[83]:=

```
xyzpositive = x > 0 && y > 0 && z > 0;
Print["positive equilibria: ",
  Simplify[Reduce[fgh == 0 && κpositive && xyzpositive, {x, y, z}]]];
```

positive equilibria: $\kappa_1 > 0$ && $\kappa_2 > 0$ && $\kappa_3 > 0$ && $\kappa_4 > 0$ && $x > 0$ && $y == \frac{\kappa_3}{x \kappa_4}$ && $z == \frac{x (\kappa_2 + \kappa_3 - x y \kappa_4)}{\kappa_1}$

In[85]:=
```
equilibrium = {x → t, y → κ3/κ4 1/t, z → κ2/κ1 t};
```

Eliminate z using $x + y + z = c$, where $c = t\dfrac{\kappa_1 + \kappa_2}{\kappa_1} + \dfrac{\kappa_3}{\kappa_4}\dfrac{1}{t}$. Then compute the Jacobian matrix at the positive equilibrium, as well as its trace and determinant.

In[86]:=
```
fg = fgh[[1 ;; 2]] /. {z → (x + y + z /. equilibrium) - x - y};
J = D[fg, {{x, y}}] /. equilibrium;
detJ = Det[J];
trJ = Tr[J];
Print["J = ", MatrixForm[J]];
Print["det J = ", detJ];
Print["tr J = ", trJ];
```

$J = \begin{pmatrix} -\kappa_1 - \kappa_2 + \kappa_3 & -\kappa_1 + t^2 \kappa_4 \\ -\kappa_3 & -t^2 \kappa_4 \end{pmatrix}$

$\det J = -\kappa_1 \kappa_3 + t^2 \kappa_1 \kappa_4 + t^2 \kappa_2 \kappa_4$

$\mathrm{tr}\, J = -\kappa_1 - \kappa_2 + \kappa_3 - t^2 \kappa_4$

For the fold bifurcation of the positive equilibrium, find where the determinant is zero and the trace is nonzero .

In[93]:=
```
Print["det J = 0 and tr J ≠ 0 : ", Simplify[Reduce[detJ == 0 && trJ ≠ 0 &&
    t > 0 && κpositive && (x + y + z /. equilibrium) == c && c > 0, {t, κ4}]]];
```

$\det J = 0$ and $\mathrm{tr}\, J \neq 0$:

$$t == \frac{c\,\kappa_1}{2\,(\kappa_1 + \kappa_2)} \;\&\&\; \kappa_4 == \frac{\kappa_1 \kappa_3}{t^2\,(\kappa_1 + \kappa_2)} \;\&\&\; \left(0 < \kappa_3 < \frac{(\kappa_1 + \kappa_2)^2}{\kappa_2} \;||\; \kappa_3 > \frac{(\kappa_1 + \kappa_2)^2}{\kappa_2}\right) \;\&\&\; c > 0 \;\&\&\; \kappa_1 > 0 \;\&\&\; \kappa_2 > 0$$

To check the transversality condition for the fold bifurcation, the verify the regularity of the map $(x, y, \kappa_i) \mapsto (f, g, \det J)$, here i can be any of 1, 2, 3, 4.

In[94]:=
```
fold = {t → c κ1/(2 (κ1 + κ2)), κ4 → 4 (κ1 + κ2) κ3/(c^2 κ1)};
fgc = fgh[[1 ;; 2]] /. {z → c - x - y};
map = Join[fgc, {Det[D[fgc, {{x, y}}]]}];
For[i = 1, i ≤ 4, i++,
  {Print["determinant of the R^3 to R^3 map when differentiated w.r.t. ",
    κi, ": ", Simplify[Det[D[map, {{x, y, κi}}] /. equilibrium /. fold]]]}];
```

determinant of the R^3 to R^3 map when differentiated w.r.t. κ_1: $\frac{2\kappa_1\kappa_2\kappa_3^2}{\kappa_1+\kappa_2}$

determinant of the R^3 to R^3 map when differentiated w.r.t. κ_2: $-\frac{2\kappa_1^2\kappa_3^2}{\kappa_1+\kappa_2}$

determinant of the R^3 to R^3 map when differentiated w.r.t. κ_3: $-2\kappa_1^2\kappa_3$

determinant of the R^3 to R^3 map when differentiated w.r.t. κ_4: $\frac{c^2\kappa_1^3\kappa_3}{2(\kappa_1+\kappa_2)}$

For the Andronov-Hopf bifurcation, find where the determinant is positive and the trace is zero (Routh-Hurwitz).

In[98]:=
```
Print["det J > 0 and tr J = 0 : ", Reduce[detJ > 0 && trJ == 0 && t > 0 && κpositive, κ4]];
```

det $J > 0$ and tr $J = 0$: $\kappa_1 > 0$ && $\kappa_2 > 0$ && $\kappa_3 > \frac{\kappa_1^2 + 2\kappa_1\kappa_2 + \kappa_2^2}{\kappa_2}$ && $t > 0$ && $\kappa_4 == -\frac{\kappa_1+\kappa_2-\kappa_3}{t^2}$

Compute the first focal value L_1.

In[99]:=
```
tracevanish = {κ4 → (κ3 - κ1 - κ2)/t^2};
omegasubst = Simplify[{ω → Sqrt[Det[J]]}];
derivatives = GetDerivatives[fg, equilibrium, 2];
derivativessimp = Simplify[derivatives /. omegasubst /. tracevanish, κpositive];
Get[path];
κ2ab = {κ2 → a κ1, κ3 → b κ1};
Print["L1 = ", Simplify[L1 /. derivativessimp /. κ2ab, κ1 > 0]];
```

$$L_1 = \frac{\left(4 + a^4 - a^3(-7+b) - 5b + 5b^2 - 2b^3 - a^2\left(-15 + 8b + b^2\right) + a\left(13 - 12b + 3b^2 + b^3\right)\right)\pi}{4\left(-1 - a^2 + a(-2+b)\right)^{3/2}(2 + a - b)\,t^2}$$

We now analyse the enumerator P and the denominator Q of L_1 (notice that we multiply both of them by -1). The denominator is positive, while the enumerator can have any sign. The Andronov-Hopf bifurcation therefore can be supercritical, subcritical, or degenerate.

In[106]:=
```
Pab = - (4 + a^4 - a^3 (-7 + b) - 5 b + 5 b^2 - 2 b^3 - a^2 (-15 + 8 b + b^2) + a (13 - 12 b + 3 b^2 + b^3));
Qab = - (-1 - a^2 + a (-2 + b))^(3/2) (2 + a - b);
abpositive = a > 0 && b > 0;
H = Simplify[(κ3 > (κ1 + κ2)^2/κ2) /. κ2ab, κ1 > 0 && abpositive];
Print["eigenvalues are ±ω i : ", Reduce[H && abpositive]];
Print["Q (a, b) > 0 : ", Reduce[Qab > 0 && H && abpositive]];
Print["coeffs of b in P(a, b) : ", MatrixForm[CoefficientList[Pab, b]]];
Print["P (a, b) = 0 : ", Reduce[Pab == 0 && H && abpositive, b]];
```

eigenvalues are $\pm\omega i$: $a > 0 \,\&\&\, b > \frac{1 + 2a + a^2}{a}$

$Q(a, b) > 0$: $a > 0 \,\&\&\, b > \frac{1 + 2a + a^2}{a}$

coeffs of b in $P(a, b)$: $\begin{pmatrix} -4 - 13a - 15a^2 - 7a^3 - a^4 \\ 5 + 12a + 8a^2 + a^3 \\ -5 - 3a + a^2 \\ 2 - a \end{pmatrix}$

$P(a, b) = 0$: (a == 1.70… &&
b == Root[4 + 13 a + 15 a^2 + 7 a^3 + a^4 + (-5 - 12 a - 8 a^2 - a^3) #1 + (5 + 3 a - a^2) #1^2 + (-2 + a) #1^3 &, 2]) || ((1.70… < a < 2 &&
(b == Root[4 + 13 a + 15 a^2 + 7 a^3 + a^4 + (-5 - 12 a - 8 a^2 - a^3) #1 + (5 + 3 a - a^2) #1^2 + (-2 + a) #1^3 &, 2] ||
b == Root[4 + 13 a + 15 a^2 + 7 a^3 + a^4 + (-5 - 12 a - 8 a^2 - a^3) #1 + (5 + 3 a - a^2) #1^2 + (-2 + a) #1^3 &, 3])) || (a == 2 && b == 6) || (a > 2 &&
b == Root[4 + 13 a + 15 a^2 + 7 a^3 + a^4 + (-5 - 12 a - 8 a^2 - a^3) #1 + (5 + 3 a - a^2) #1^2 + (-2 + a) #1^3 &, 3])

We now depict the sign of L_1 as a function of a and b.

In[114]:=

```
amax = 4;
bmax = 40;
bmaxsmall = 8;
Pneg = RegionPlot[Pab < 0 && a > 0 && b > (a + 1)^2/a,
    {a, 0, amax}, {b, 0, bmax}, GridLines → Automatic, BoundaryStyle → None,
    AxesLabel → {Style[a, Bold, 16], Style[b, Bold, 16]}, Frame → None, Axes → True];
Ppos1 = RegionPlot[Pab > 0 && a > 0 && b > (a + 1)^2/a, {a, 0, amax}, {b, bmaxsmall, bmax},
    PlotStyle → {Yellow, Opacity[0.5]}, BoundaryStyle → None];
Ppos2 = RegionPlot[Pab > 0 && a > 0 && b > (a + 1)^2/a, {a, 0, amax}, {b, 0, bmaxsmall},
    PlotStyle → {Yellow, Opacity[0.5]}, BoundaryStyle → None];
det0 = Plot[(a + 1)^2/a, {a, 0, amax}, PlotStyle → Black, PlotRange → {{0, amax}, {0, bmax}}];
polynom = 4 + 13 a + 15 a^2 + 7 a^3 + a^4 + (-5 - 12 a - 8 a^2 - a^3) #1 + (5 + 3 a - a^2) #1^2 + (-2 + a) #1^3 &;
b2 = Root[polynom, 2];
b3 = Root[polynom, 3];
plotb2 =
  Plot[b2, {a, 1.70…, 2}, PlotStyle → Red, PlotRange → {{0, amax}, {0, bmax}}];
plotb3a = Plot[b3, {a, 1.70…, 2}, PlotStyle → Red,
    PlotRange → {{0, amax}, {0, bmax}}];
plotb3b = Plot[b3, {a, 2, amax}, PlotStyle → Red, PlotRange → {{0, amax}, {0, bmax}}];
txt = Graphics[{Text[Style["L1>0\n(subcritical\nAndronov-Hopf)", 16], {0.9, 15}],
     Text[Style["L1<0\n(supercritical\nAndronov-Hopf)", 16], {3, 15}],
     Rotate[Text[Style["L1=0, L2>0 (Bautin)", 16, Red], {1.65, 30}], 90 Degree]}];
Show[Pneg, Ppos1, Ppos2, det0, plotb2, plotb3a, plotb3b, txt]
```

Out[128]=



Compute the second focal value, L_2, where $L_1 = 0$. Turns out it is always positive, justifying the red label on the above figure.
Thus, one can find parameter values for which the equilibrium is repelling and is surrounded by two limit cycles (the inner one is stable, the outer one is unstable).

In[129]:=

```
L2ab = Simplify[L2 /. derivativessimp /. κ2ab, κ1 > 0];
Print["L2 = ", L2ab];
Print["L2 < 0  : ", Reduce[L2ab < 0 && Pab == 0 && H && abpositive && t > 0]];
Print["L2 = 0  : ", Reduce[L2ab == 0 && Pab == 0 && H && abpositive && t > 0]];
Print["L2 > 0  : ", Reduce[L2ab > 0 && Pab == 0 && H && abpositive && t > 0, b]];
```

$$L_2 = \frac{1}{48\left(-1-a^2+a(-2+b)\right)^{7/2}(2+a-b)^2 t^4}$$

$$\left(684 + 139\,a^9 + a^8\,(1713 - 552\,b) - 1872\,b + 2845\,b^2 - 2714\,b^3 + 1655\,b^4 - 598\,b^5 + 96\,b^6 + a^7\left(8783 - 6352\,b + 718\,b^2\right) + a^6\left(25\,193 - 28\,496\,b + 8960\,b^2 - 89\,b^3\right) + a^5\left(45\,129 - 68\,392\,b + 36\,601\,b^2 - 5978\,b^3 - 649\,b^4\right) + a^4\left(52\,759 - 98\,200\,b + 73\,973\,b^2 - 25\,270\,b^3 + 2085\,b^4 + 662\,b^5\right) + a^3\left(40\,453 - 87\,392\,b + 83\,988\,b^2 - 43\,637\,b^3 + 11\,483\,b^4 - 680\,b^5 - 272\,b^6\right) + a\left(5528 - 14\,376\,b + 19\,285\,b^2 - 16\,108\,b^3 + 8707\,b^4 - 3039\,b^5 + 677\,b^6 - 86\,b^7\right) + a^2\left(19\,683 - 47\,392\,b + 54\,814\,b^2 - 37\,650\,b^3 + 15\,806\,b^4 - 3778\,b^5 + 338\,b^6 + 43\,b^7\right)\right)\pi$$

$L_2 < 0$: False

$L_2 = 0$: False

$L_2 > 0$: t > 0 && ((a == 1.70… && b == Root[4 + 13 a + 15 a^2 + 7 a^3 + a^4 + (-5 - 12 a - 8 a^2 - a^3) #1 + (5 + 3 a - a^2) #1^2 + (-2 + a) #1^3 &, 2]) || (1.70… < a < 2 && (b == Root[4 + 13 a + 15 a^2 + 7 a^3 + a^4 + (-5 - 12 a - 8 a^2 - a^3) #1 + (5 + 3 a - a^2) #1^2 + (-2 + a) #1^3 &, 2] || b == Root[4 + 13 a + 15 a^2 + 7 a^3 + a^4 + (-5 - 12 a - 8 a^2 - a^3) #1 + (5 + 3 a - a^2) #1^2 + (-2 + a) #1^3 &, 3])) || (a == 2 && b == 6) || (a > 2 && b == Root[4 + 13 a + 15 a^2 + 7 a^3 + a^4 + (-5 - 12 a - 8 a^2 - a^3) #1 + (5 + 3 a - a^2) #1^2 + (-2 + a) #1^3 &, 3]))

We now verify the nondegeneracy of the Bautin bifurcation (Section 8.3 in Kuznetsov: Elements of Applied Bifurcation Theory). The approach we follow here is that κ_1, κ_2, t are fixed, and κ_3, κ_4 serve as bifurcation parameters. To check that the map $(\kappa_3, \kappa_4) \mapsto (\mathrm{tr}\, J, L_1)$ is regular at the critical value, we need a more general formula of the first focal value, not assuming that the real part of the eigenvalues is zero (but still assuming the Jacobian matrix at the equilibrium is of the form $\begin{pmatrix} \mu & -\omega \\ \omega & \mu \end{pmatrix}$). Derivation of this more general first focal value formula can be carried out following http://www.scholarpedia.org/article/Bautin_bifurcation. Important note: below $F_{i,j}$ and $G_{i,j}$ are the partial derivatives, i.e., $f_{i,j} = \frac{F_{i,j}}{i!\,j!}$ and $g_{i,j} = \frac{G_{i,j}}{i!\,j!}$. We remark that the formula derived below is valid only for $\omega > 0$.

In[134]:=

```
n = 2;
A = {{μ, -ω}, {ω, μ}};
B[x_, y_] := Sum[
    {F_Count[{k,l},1],Count[{k,l},2], G_Count[{k,l},1],Count[{k,l},2]} x⟦k⟧ × y⟦l⟧, {k, 1, n}, {l, 1, n}];
CC[x_, y_, z_] := Sum[{F_Count[{k,l,m},1],Count[{k,l,m},2], G_Count[{k,l,m},1],Count[{k,l,m},2]}
     x⟦k⟧ × y⟦l⟧ × z⟦m⟧, {k, 1, n}, {l, 1, n}, {m, 1, n}];
q = {-1/Sqrt[2], I/Sqrt[2]};
p = q;
Q1 = Simplify[CC[q, q, q*]];
Q2 = Simplify[B[q, 1/Det[A] A.B[q, q*]]];
Q3 = Simplify[B[q*, Inverse[2 (μ + I ω) IdentityMatrix[2] - A].B[q, q]]];
c1 = Simplify[p*.Simplify[1/2 Q1 + Q2 + 1/2 Q3]];
L1general = Simplify[1/ω ComplexExpand[Re[c1]] - μ/ω^2 ComplexExpand[Im[c1]]];
Print["L1 = ", L1general];
```

$$L_1 = \frac{1}{8\,\omega^2\left(\mu^2+\omega^2\right)\left(\mu^2+9\,\omega^2\right)}
\Big(\left(3\mu^3\omega+11\mu\omega^3\right)F_{0,2}^2+\left(\mu^5+10\mu^3\omega^2+9\mu\omega^4\right)F_{0,3}+8\mu^3\omega F_{1,1}^2+8\mu\omega^3F_{1,1}^2+\mu^4\omega F_{1,2}+10\mu^2\omega^3F_{1,2}+
9\omega^5F_{1,2}+3\mu^4F_{1,1}F_{2,0}+28\mu^2\omega^2F_{1,1}F_{2,0}+9\omega^4F_{1,1}F_{2,0}+5\mu^3\omega F_{2,0}^2+29\mu\omega^3F_{2,0}^2+\mu^5F_{2,1}+
10\mu^3\omega^2F_{2,1}+9\mu\omega^4F_{2,1}+\mu^4\omega F_{3,0}+10\mu^2\omega^3F_{3,0}+9\omega^5F_{3,0}-6\mu^3\omega F_{1,1}G_{0,2}+10\mu\omega^3F_{1,1}G_{0,2}+
5\mu^3\omega G_{0,2}^2+29\mu\omega^3G_{0,2}^2+\mu^4\omega G_{0,3}+10\mu^2\omega^3G_{0,3}+9\omega^5G_{0,3}-6\mu^3\omega F_{2,0}G_{1,1}+10\mu\omega^3F_{2,0}G_{1,1}-
3\mu^4G_{0,2}G_{1,1}-28\mu^2\omega^2G_{0,2}G_{1,1}-9\omega^4G_{0,2}G_{1,1}+8\mu^3\omega G_{1,1}^2+8\mu\omega^3G_{1,1}^2+F_{0,2}\Big(\left(3\mu^4+28\mu^2\omega^2+9\omega^4\right)F_{1,1}+
32\mu\omega^3F_{2,0}+3\mu^4G_{0,2}+28\mu^2\omega^2G_{0,2}+9\omega^4G_{0,2}+10\mu^3\omega G_{1,1}+26\mu\omega^3G_{1,1}\Big)-\mu^5G_{1,2}-
10\mu^3\omega^2G_{1,2}-9\mu\omega^4G_{1,2}+10\mu^3\omega F_{1,1}G_{2,0}+26\mu\omega^3F_{1,1}G_{2,0}-3\mu^4F_{2,0}G_{2,0}-28\mu^2\omega^2F_{2,0}G_{2,0}-
9\omega^4F_{2,0}G_{2,0}+32\mu\omega^3G_{0,2}G_{2,0}-3\mu^4G_{1,1}G_{2,0}-28\mu^2\omega^2G_{1,1}G_{2,0}-9\omega^4G_{1,1}G_{2,0}+3\mu^3\omega G_{2,0}^2+
11\mu\omega^3G_{2,0}^2+\mu^4\omega G_{2,1}+10\mu^2\omega^3G_{2,1}+9\omega^5G_{2,1}-\mu^5G_{3,0}-10\mu^3\omega^2G_{3,0}-9\mu\omega^4G_{3,0}\Big)$$

Now that we have the general formula for L_1, let us apply it to the lifted Brusselator. After that, we compute the relevant determinant for checking the transversality condition of the Bautin bifurcation.

In[146]:=
```
xyshift = {x → x + (x /. equilibrium), y → y + (y /. equilibrium)};
T = {{ω, 0}, {(d - a) / 2, -b}};
Tinvuv = Inverse[T].{u, v};
FG = T . fg /. xyshift /. {x → Tinvuv[[1]], y → Tinvuv[[2]]} /.
   {a → J[[1, 1]], b → J[[1, 2]], d → J[[2, 2]]};
m = 1;
derivatives = {};
For[i = 0, i ≤ 2 m + 1, i++, For[j = 0, j ≤ 2 m + 1 - i, j++,
   derivatives =
    Join[derivatives, Simplify[{F_{i,j} → (D[FG[[1]], {u, i}, {v, j}] /. {u → 0, v → 0}),
       G_{i,j} → (D[FG[[2]], {u, i}, {v, j}] /. {u → 0, v → 0})}]]]];
musubst = Simplify[{μ → Tr[J]/2}];
omegasubst = Simplify[{ω → Sqrt[4 Det[J] - Tr[J]^2]/2}];
L1gen = Simplify[L1general /. Simplify[derivatives] /. musubst /. omegasubst];
det =
  Simplify[Det[Simplify[D[{Tr[J], L1gen}, {{κ3, κ4}}]] /. tracevanish] /. κ2ab], κ1 > 0];
Print["determinant of the R^3 to R^3 map: ", det];
Print["det > 0 : ", Reduce[det > 0 && Pab == 0 && H && abpositive && κ1 > 0, b]];
Print["det = 0 : ", Reduce[det == 0 && Pab == 0 && H && abpositive && κ1 > 0, b]];
Print["det < 0 : ", Reduce[det < 0 && Pab == 0 && H && abpositive && κ1 > 0, b]];
```

determinant of the R^3 to R^3 map:

$$-\left(\left(-12+5a^6+a^5(39-14b)+40b-34b^2+8b^3+a^4\left(107-83b+12b^2\right)+a^3\left(125-141b+51b^2-2b^3\right)-a^2\left(-48+46b-33b^2+9b^3+b^4\right)+a\left(-16+66b-39b^2+b^3+2b^4\right)\right)\Big/\left(8\left(-1-a^2+a(-2+b)\right)^{7/2}(2+a-b)^2\kappa_1^3\right)\right)$$

det > 0 : $\kappa_1 > 0$ && 1.70… $< a < 2$ &&
$b == \text{Root}\left[4+13a+15a^2+7a^3+a^4+\left(-5-12a-8a^2-a^3\right)\#1+\left(5+3a-a^2\right)\#1^2+(-2+a)\#1^3\&, 3\right]$

det = 0 : $\kappa_1 > 0$ && $a ==$ 1.70… &&
$b == \text{Root}\left[4+13a+15a^2+7a^3+a^4+\left(-5-12a-8a^2-a^3\right)\#1+\left(5+3a-a^2\right)\#1^2+(-2+a)\#1^3\&, 2\right]$

det < 0 : $\kappa_1 > 0$ &&
((1.70… $< a \le 2$ && $b == \text{Root}\left[4+13a+15a^2+7a^3+a^4+\left(-5-12a-8a^2-a^3\right)\#1+\left(5+3a-a^2\right)\#1^2+(-2+a)\#1^3\&, 2\right]$) || ($a > 2$ &&
$b == \text{Root}\left[4+13a+15a^2+7a^3+a^4+\left(-5-12a-8a^2-a^3\right)\#1+\left(5+3a-a^2\right)\#1^2+(-2+a)\#1^3\&, 3\right]$))

By the above, the transversality condition is fulfilled except when $a = \overline{a} =$ 1.70…. Below we find the symbolic value of 1.70... .

In[161]:=
```
Simplify[1.70...]
```

Out[161]=
$$\frac{1}{10}\times\left(-10+\sqrt{5\times\left(73+22\sqrt{11}\right)}\right)$$

We now turn to the Bogdanov-Takens bifurcation (http://www.scholarpedia.org/article/Bogdanov-Takens_bifurcation or Section 8.4 in Kuznetsov: Elements of Applied Bifurcation Theory). Let us start with finding the parameter values, where $\det J = 0$ and $\operatorname{tr} J = 0$.

In[162]:=
```
Print["det J = 0 and tr J = 0 : ",
  Simplify[Reduce[detJ == 0 && trJ == 0 && c == Simplify[(x + y + z /. equilibrium)] &&
    κpositive && t > 0 && c > 0, {t, κ3, κ4}]]];
```

det $J = 0$ and tr $J = 0$: $\kappa_1 > 0$ && $\kappa_2 > 0$ && $c > 0$ && $t == \frac{c\,\kappa_1}{2\,(\kappa_1+\kappa_2)}$ && $\kappa_3 == \frac{(\kappa_1+\kappa_2)^2}{\kappa_2}$ && $\kappa_4 == \frac{\kappa_1\,\kappa_3}{t^2\,(\kappa_1+\kappa_2)}$

We regard c, κ_1, κ_2 as fixed, while κ_3, κ_4 are parameters. Let us now find the vectors q_0, q_1, p_0, p_1 for which $A\,q_0 = 0$, $A\,q_1 = q_0$, $A^T\,p_1 = 0$, $A^T\,p_0 = p_1$ and $q_0^T\,q_0 = p_1^T\,p_1 = 1$, $q_1^T\,p_0 = q_0^T\,p_1 = 0$, where A is the Jacobian matrix. Below we divide the ODE by κ_2 and introduce the new parameter $d = \frac{\kappa_1}{\kappa_2}$ in order to get slightly simpler formulas. For finding q_0, q_1, p_0, p_1, we follow Appendix A in the article Kuznetsov: Practical computation of normal forms on center manifolds at degenerate Bogdanov-Takens bifurcations.

In[163]:=

```
BT = {t → c κ1/(2 (κ1 + κ2)), κ3 → (κ1 + κ2)^2/κ2, κ4 → 4 (κ1 + κ2)^3/(c^2 κ1 κ2)};
A = Simplify[J/κ2 /. BT /. {κ1 → d κ2}];
Print["A = ", MatrixForm[A]];
Clear[p, q];
q0 = {d, -1 - d};
q1 = {0, 1 / d};
p0 = {0, -1 / (d + 1)};
p1 = {d + 1, d};
μ = Sqrt[q0.q0];
q0 = Simplify[1/μ q0];
q1 = 1/μ q1;
q1 = Simplify[q1 - (q0.q1) q0];
ν = Simplify[q0.p0];
p1 = Simplify[1/ν p1];
p0 = p0 - (p0.q1) p1;
p0 = Simplify[1/ν p0];
```

$$A = \begin{pmatrix} d\ (1+d) & d^2 \\ -(1+d)^2 & -d\ (1+d) \end{pmatrix}$$

Notice that A, the Jacobian matrix, is indeed a nonzero matrix and its eigenvalue is zero (with multiplicity two). With this, the condition (BT.0) in Theorem 8.4 in Kuznetsov's book is fulfilled. Below we check (BT.1) and (BT.2).

In[179]:=

```
xyshift = {x → x + (x /. equilibrium), y → y + (y /. equilibrium)};
fgshifted = fg / κ2 /. xyshift;
derivatives = {};
For[i = 0, i ≤ 2, i++, For[j = 0, j ≤ 2 - i, j++,
   derivatives = Join[derivatives,
     Simplify[{F_{i,j} → (D[fgshifted[[1]], {x, i}, {y, j}] /. {x → 0, y → 0}),
       G_{i,j} → (D[fgshifted[[2]], {x, i}, {y, j}] /. {x → 0, y → 0})}]]]];
derivs = Simplify[derivatives /. BT /. {κ1 → d κ2}];
Print["what has to be nonzero in (BT.1): ",
  Simplify[p0.B[q0, q0] + p1.B[q0, q1] /. derivs]];
Print["what has to be nonzero in (BT.2): ", Simplify[p1.B[q0, q0] /. derivs]];
```

what has to be nonzero in (BT.1): $-\frac{4\,d\,(1+d)^2}{c\,\sqrt{1+2\,d+2\,d^2}}$

what has to be nonzero in (BT.2): $-\frac{4\,d\,(1+d)^3}{c\,\sqrt{1+2\,d+2\,d^2}}$

Since each of the above two quantities is negative, their product is positive. Thus, the Bogdanov-Takens bifurcation is subcritical, meaning the Andronov-Hopf bifurcation is subcritical. It is left to verify the transversality conditions (BT.3) in Theorem 8.4 in Kuznetsov's book.

In[186]:=
```
zsubst = {z → c - x - y};
J = D[fgh⟦1 ;; 2⟧ /. zsubst, {{x, y}}];
map = {fgh⟦1⟧ /. zsubst, fgh⟦2⟧ /. zsubst, Tr[J], Det[J]};
Print["determinant of the 4×4 matrix in (BT.3): ",
  Simplify[Det[D[map, {{x, y, κ3, κ4}}] /. equilibrium /. BT]]];
```

determinant of the 4×4 matrix in (BT.3): $\frac{1}{2}\,c^2\,\kappa_1^3$

Since the determinant above is nonzero (in fact, it is positive), the map in question is indeed regular. This concludes the analysis of the lifted Brusselator.