Limit cycles in mass-conserving deficiency-one mass-action systems

Balázs Boros, Josef Hofbauer

This Mathematica Notebook is a supplementary material to the paper which has the same title as this document.
It contains some of the calculations appearing in the paper.

0 Focal values

Derive L_1, L_2, and L_3, the first, the second, and the third focal values for the differential equation

$$\dot{x} = -y + \sum_{i+j\geq 2} f_{i,j}\, x^i\, y^j,$$

$$\dot{y} = x + \sum_{i+j\geq 2} g_{i,j}\, x^i\, y^j.$$

Theoretical background: Chapter 4 in Dumortier, Llibre, Artés: Qualitative Theory of Planar Differential Systems.

In[ ]:=
```
m = 3;
cd = {}; R2cd = {};
For[k = 2, k ≤ 2 m + 1, k++, For[i = 0, i ≤ k, i++,
  {cd = Join[cd, {c_{k,i}, d_{k,i}}], R2cd = Join[R2cd, {R_{k,i} → c_{k,i} + d_{k,i} I}]}]];
coeffsxy =
  CoefficientList[ComplexExpand[Sum[Sum[R_{k,i} z^(k-i) (z*)^i, {i, 0, k}], {k, 2, 2 m + 1}]
   /. R2cd /. {z → x + y I}], {x, y}];
cond = True;
For[k = 2, k ≤ 2 m + 1, k++, For[i = 0, i ≤ k, i++,
  {cond = cond && (f_{i,k-i} == ComplexExpand[Re[coeffsxy[[i + 1, k - i + 1]]]]) &&
  (g_{i,k-i} == ComplexExpand[Im[coeffsxy[[i + 1, k - i + 1]]]])}]];
cd2fg = Solve[cond, cd][[1]];

For[k = 2, k ≤ 2 m + 1, k++, R_k = Sum[R_{k,i} z^(k-i) w^i, {i, 0, k}]];

(* F[i,j] computes the polynomial F_i(h_j) *)
F[i_, j_] := Module[{coeffs, M, mtx},
   coeffs = CoefficientList[D[R_i h_j, {z, 1}], {z, w}];
   M = Dimensions[coeffs][[1]] - 1;
   mtx = (coeffs + Transpose[coeffs*])
     Table[If[k + l == M && k ≠ l, 1/(k - l), 0], {k, 0, M}, {l, 0, M}];
   I z^Range[0,M] .mtx.w^Range[0,M]];

h_0 = 1;
For[k = 1, k ≤ 2 m - 1, k++, h_k = Sum[F[k + 1 - l, l], {l, 0, k - 1}]];

(* H[k,j] computes H_k(h_j), note that one of k and j is even,
the other one is odd in all of the interesting cases *)
H[k_, j_] := Module[{},
   coeffs = CoefficientList[h_j, {z, w}];
   Sum[Coefficient[R_k, z^a w^(k-a)] × coeffs[[((k - 2 a + 1) + j)/2 + 1, (j - (k - 2 a + 1))/2 + 1]],
    {a, (k + 1 - j)/2, (k + 1 + j)/2}]];
For[j = 1, j ≤ m, j++, L_j = Simplify[ComplexExpand[
     2 π Re[Sum[H[2 j + 1 - l, l], {l, 0, 2 j - 1}]] /. R2cd /. cd2fg]]];
```

Display the first focal value, L_1. The second focal value, L_2, is somewhat longer. The third focal value, L_3, is very long. Important note: here $f_{i,j}$ and $g_{i,j}$ include the division by $i!\,j!$, so they are the Taylor coefficients (not simply the respective derivatives).

In[•]:=
```
Print["ℒ₁ = ", L₁];
```

$$\mathcal{L}_1 = \frac{1}{4}\pi\,(f_{1,2} + f_{1,1}\,f_{2,0} + 3\,f_{3,0} + f_{0,2}\,(f_{1,1} + 2\,g_{0,2}) + 3\,g_{0,3} - g_{0,2}\,g_{1,1} - 2\,f_{2,0}\,g_{2,0} - g_{1,1}\,g_{2,0} + g_{2,1})$$

Good practice: compute the focal values once, and save them in a file. When needed, load from that file.

In[•]:=
```
L1 = L₁; L2 = L₂; L3 = L₃;(* when storing in a file, better avoiding subscripts *)
path =
  "C://bboros/Dropbox/dfc1thm/3d/parallelogram_paper/mathematica/focal_values.mx";
DumpSave[path, {L1, L2, L3}];
Protect[path];
Off[Remove::rmptc];
Remove["Global`*"]; (* clear all variables *)
On[Remove::rmptc];
Unprotect[path];
```

Define a module that computes the necessary partial derivatives. To be used later.

In[•]:=
```
GetDerivatives[fg_, equilibrium_, m_] :=
  Module[{J, xyshift, T, Tinvuv, FG, derivatives, a, b, u, v},
   J = Simplify[D[fg, {{x, y}}] /. equilibrium];
   xyshift = {x → x + (x /. equilibrium), y → y + (y /. equilibrium)};
   T = {{1, 0}, {-a / ω, -b / ω}};
   Tinvuv = Inverse[T].{u, v};
   FG =
    (T . fg /. xyshift) / ω /. {x → Tinvuv⟦1⟧, y → Tinvuv⟦2⟧} /. {a → J⟦1, 1⟧, b → J⟦1, 2⟧};
   derivatives = {};
   For[i = 0, i ≤ 2 m + 1, i++, For[j = 0, j ≤ 2 m + 1 - i, j++,
     derivatives = Join[derivatives, {f_{i,j} → (D[FG⟦1⟧, {u, i}, {v, j}] / ((i !) * (j !)) /. {u → 0, v → 0}),
        g_{i,j} → (D[FG⟦2⟧, {u, i}, {v, j}] / ((i !) * (j !)) /. {u → 0, v → 0})}]]];
   derivatives];
```

3 Parallelograms

3.1 Supercritical Andronov-Hopf bifurcation

Start with the planar parallelogram ($\gamma = 0$) and compute the first focal value. Observe that it is negative. Thus, the Andronov-Hopf bifurcation is supercritical, and a stable limit cycle is born.

```
In[∘]:= κpositive = κ1 > 0 && κ2 > 0 && κ3 > 0 && κ4 > 0;
fg = κ1 y {1, -1} + κ2 x {0, 2} + κ3 x y^2 {-1, 1} + κ4 y^3 {0, -2};
equilibrium = {x → ((κ1^3 κ4)/(κ3^3 κ2))^(1/4), y → ((κ1 κ2)/(κ3 κ4))^(1/4)};
J = D[fg, {{x, y}}] /. equilibrium;
trJ = Simplify[Tr[J], κpositive];
tracevanish = Normal[Solve[trJ == 0 && κpositive][[1]]];
ωsubst = Simplify[{ω → Sqrt[Det[J]]}, κpositive];
derivatives = Simplify[GetDerivatives[fg, equilibrium, 1], κpositive];
Get[path];
Print["L1 = ", Simplify[L1 /. derivatives /. ωsubst /. tracevanish, κpositive]];
```

$$L_1 = -\frac{\pi\,(\kappa_3\,\kappa_4)^{3/2}}{\sqrt{2}\,\kappa_2\,(\kappa_3 + 6\,\kappa_4)^2}$$

Let us now lift the planar parallelogram by adding a new species in a way that rank of the network remains two (in fact, the Euclidean embedded graph remains a parallelogram).
Verify that the formula for the equilibria and the trace are correct.

```
In[∘]:= fgh = κ1 y z^γ {1, -1, 0} + κ2 x z^γ {0, 2, -γ} + κ3 x y^2 {-1, 1, 0} + κ4 y^3 {0, -2, γ};
equilibrium = {x → ((κ1 κ4)/(κ2 κ3))^(1/2) t^γ, y → t^γ, z → ((κ3 κ4)/(κ1 κ2))^(1/(2 γ)) t^2};
trJ = (Sqrt[(κ1 κ3 κ4)/κ2] - (κ3 + 6 κ4)) t^(2 γ) - γ^2 κ4 ((κ1 κ2)/(κ3 κ4))^(1/(2 γ)) t^(3 γ-2);
J = D[fgh, {{x, y, z}}];
Print["the equilibria are given correctly: ",
  Simplify[fgh /. equilibrium, κpositive && t > 0 && γ > 0] == {0, 0, 0}];
Print["the trace is given correctly: ",
  Simplify[trJ - Tr[J] /. equilibrium, κpositive && t > 0 && γ > 0] == 0];
```

```
the equilibria are given correctly: True
```

```
the trace is given correctly: True
```

We reparametrise the rate constants to make the formulas somewhat lighter.
Then we compute the derivatives that will be plugged in to the focal value formula.

In[∘]:=
```
abcdpositive = a > 0 && b > 0 && c > 0 && d > 0;
κ2abcd = {κ1 → a^(2γ), κ2 → b^(2γ), κ3 → c^(2γ), κ4 → d^(2γ)};
fg = Simplify[fgh[[1 ;; 2]] /. {z → ((γ x + γ y + 2 z /. equilibrium) - γ x - γ y)/2}];
derivatives =
  Simplify[GetDerivatives[fg, equilibrium, 1] /. κ2abcd, abcdpositive && t > 0 && γ > 0];
J = D[fg, {{x, y}}] /. equilibrium /. κ2abcd;
ωsubst = Simplify[{ω → Sqrt[Det[J]]}, abcdpositive && t > 0 && γ > 0];
```

3.1.1 Case $\gamma = 2$

We take $t = 1$ (when $\gamma = 2$, due to the homogeneity, the dynamics is the same in every stoichiometric class).

It turns out L_1 is negative, and thus, the Andronov-Hopf bifurcation is always supercritical.

In[∘]:=
```
γsubst = {γ → 2};
tsubst = {t → 1};
dersimple = Simplify[derivatives /. γsubst /. tsubst];
trJsimple = Simplify[trJ /. κ2abcd /. γsubst /. tsubst, abcdpositive];
ωsimple = Simplify[ωsubst /. γsubst /. tsubst, abcdpositive];
L1abcd = Simplify[L1 /. dersimple];
L1enum = Simplify[ω^3 L1abcd /. ωsimple];
(* the multiplication by ω^3 is to simplify the formula a bit *)
Print["L1 is nonnegative for: ", Reduce[L1enum ≥ 0 && trJsimple == 0 && abcdpositive]];
Print["L1 is negative for: ", Reduce[L1enum < 0 && trJsimple == 0 && abcdpositive]];
Print["the trace vanishes for: ", Reduce[trJsimple == 0 && abcdpositive]];
```

L_1 is nonnegative for: False

L_1 is negative for: $d > 0 \,\&\&\, c > 0 \,\&\&\, b > 0 \,\&\&\, a == \frac{2\,b^3\,d}{c^3} + \sqrt{\frac{b^2\,c^8 + 4\,b^6\,d^4 + 6\,b^2\,c^4\,d^4}{c^6\,d^2}}$

the trace vanishes for: $d > 0 \,\&\&\, c > 0 \,\&\&\, b > 0 \,\&\&\, a == \frac{2\,b^3\,d}{c^3} + \sqrt{\frac{b^2\,c^8 + 4\,b^6\,d^4 + 6\,b^2\,c^4\,d^4}{c^6\,d^2}}$

3.1.2 Case $\gamma \neq 2$

Below we compute and analyse and the first focal value for fixed values of γ. We find that it is negative.

Notice that we do another reparametrisation for convenience.

Further, the analysis of the sign of the first focal value is performed by investigating the enumerator and the denominator separately (this seems to be a lot faster).

In[ ]:=
```
gammas = {1, 3, 4, 5, 6};
For[i = 1, i ≤ Length[gammas], i++, {
   gamma = gammas[[i]];
   Print[Style[StringJoin["γ =", ToString[gamma]], {Blue, Bold}]];
   γsubst = {γ → gamma};
   tHopf = {t → (1/γ^2 ((a/b)^γ - (c/d)^γ - 6 (d/c)^γ) (c d)/(a b) (c/d)^γ)^(1/(γ-2))};
   (* solution of trJ=0 *)
   ωsimple = Simplify[ωsubst /. γsubst, abcdpositive && t > 0];
   dersimple = Simplify[derivatives /. γsubst, abcdpositive && t > 0];
   tsimple = Simplify[tHopf /. γsubst];
   L1simple = Simplify[L1 /. dersimple /. ωsimple];
   L1abcd = Simplify[L1simple /. tsimple, abcdpositive];
   abcd2ABCD = {a → A^(1/γ), b → B^(1/γ), c → C^(1/γ), d → D^(1/γ)} /. γsubst;
   (* another reparametrisation *)
   ABCDpositive = A > 0 && B > 0 && C > 0 && D > 0;
   L1ABCD = Simplify[L1abcd /. abcd2ABCD, ABCDpositive];
   condHopf = A/B - C/D - 6 D/C > 0; (* condition for Hopf *)
   L1enum = Numerator[L1ABCD];
   L1denom = Denominator[L1ABCD];
   Print["L1 = ", L1ABCD];
   If[gamma > 2, {
     Print["enumerator of L1 is positive for ",
      Reduce[L1enum > 0 && ABCDpositive && condHopf, A]];
     Print["denominator of L1 is negative for ",
      Reduce[L1denom < 0 && ABCDpositive && condHopf, A]];
    },
    {
     Print["enumerator of L1 is negative for ",
      Reduce[L1enum < 0 && ABCDpositive && condHopf, A]];
     Print["denominator of L1 is positive for ",
      Reduce[L1denom > 0 && ABCDpositive && condHopf, A]];
    }];
  }];
```

$\gamma = 1$

$$L_1 = -\left(\left(\sqrt{C}\ \left(A\,C\,D - B\,\left(C^2 + 6\,D^2\right)\right)^2\,\left(3\,A^6\,C^4\,D^6 - 2\,A^5\,B\,C^3\,D^5\,\left(3\,C^2 + 31\,D^2\right) + A^4\,B^2\,C^2\,D^4\,\left(-C^4 + 98\,C^2\,D^2 + 516\,D^4\right) + 8\,A^3\,B^3\,C\,D^3\,\left(C^6 + 4\,C^4\,D^2 - 65\,C^2\,D^4 - 261\,D^6\right) + B^6\,C^2\,\left(C^8 + 18\,C^6\,D^2 + 108\,C^4\,D^4 + 232\,C^2\,D^6 + 96\,D^8\right) + 2\,A\,B^5\,C\,D\,\left(-C^8 + 3\,C^6\,D^2 + 128\,C^4\,D^4 + 620\,C^2\,D^6 + 720\,D^8\right) + A^2\,B^4\,D^2\,\left(-3\,C^8 - 92\,C^6\,D^2 - 360\,C^4\,D^4 + 648\,C^2\,D^6 + 3456\,D^8\right)\right)\,\pi\right) \Big/ \left(8\,\sqrt{2}\,A^4\,B^4\,D^4\,\left(A\,C\,D - B\,\left(C^2 + 4\,D^2\right)\right)\,\left(A^2\,C\,D^2 - 6\,A\,B\,D^3 - B^2\,\left(C^3 + 2\,C\,D^2\right)\right)^{3/2}\right)\right)$$

enumerator of L_1 is negative for $C > 0\ \&\&\ D > 0\ \&\&\ B > 0\ \&\&\ A > \frac{B\,C^2 + 6\,B\,D^2}{C\,D}$

denominator of L_1 is positive for $C > 0\ \&\&\ D > 0\ \&\&\ B > 0\ \&\&\ A > \frac{B\,C^2 + 6\,B\,D^2}{C\,D}$

$\gamma = 3$

$$L_1 = \left(\left(59049\,B^8\,D^8\,\left(3\,A^6\,C^3\,D^6 - 2\,A^5\,B\,C^2\,D^5\,\left(C^2 - 9\,D^2\right) - 3\,A^4\,B^2\,C\,D^4\,\left(3\,C^4 - 2\,C^2\,D^2 + 180\,D^4\right) + 6\,A\,B^5\,C^2\,D\,\left(-C^6 + C^4\,D^2 + 28\,C^2\,D^4 + 84\,D^6\right) + 8\,A^3\,B^3\,D^3\,\left(C^6 - 2\,C^4\,D^2 - 18\,C^2\,D^4 + 243\,D^6\right) + A^2\,B^4\,C\,D^2\,\left(5\,C^6 - 36\,C^4\,D^2 + 384\,C^2\,D^4 + 1080\,D^6\right) + B^6\,\left(C^9 + 22\,C^7\,D^2 + 132\,C^5\,D^4 + 216\,C^3\,D^6\right)\right)\,\pi\right) \Big/ \left(8\,\sqrt{2}\,C^{5/2}\,\left(B\,C^2 - A\,C\,D + 6\,B\,D^2\right)^6\,\left(-A\,C\,D + B\,\left(C^2 + 4\,D^2\right)\right)\,\left(A^2\,C\,D^2 - 6\,A\,B\,D^3 - B^2\,\left(C^3 + 2\,C\,D^2\right)\right)^{3/2}\right)\right)$$

enumerator of L_1 is positive for $C > 0\ \&\&\ D > 0\ \&\&\ B > 0\ \&\&\ A > \frac{B\,C^2 + 6\,B\,D^2}{C\,D}$

denominator of L_1 is negative for $C > 0\ \&\&\ D > 0\ \&\&\ B > 0\ \&\&\ A > \frac{B\,C^2 + 6\,B\,D^2}{C\,D}$

$\gamma = 4$

$$L_1 = \left(\left(256\,\sqrt{2}\,B^5\,D^5\,\left(3\,A^6\,C^4\,D^6 + 88\,A^5\,B\,C^3\,D^7 - A^4\,B^2\,C^2\,D^4\,\left(13\,C^4 + 40\,C^2\,D^2 + 1464\,D^4\right) + 8\,A^3\,B^3\,C\,D^3\,\left(C^6 - 14\,C^4\,D^2 - 14\,C^2\,D^4 + 684\,D^6\right) + A^2\,B^4\,D^2\,\left(9\,C^8 + 16\,C^6\,D^2 + 1344\,C^4\,D^4 + 2592\,C^2\,D^6 - 3024\,D^8\right) + B^6\,C^2\,\left(C^8 + 24\,C^6\,D^2 + 120\,C^4\,D^4 - 32\,C^2\,D^6 - 624\,D^8\right) - 8\,A\,B^5\,C\,D\,\left(C^8 - 3\,C^6\,D^2 - 14\,C^4\,D^4 + 76\,C^2\,D^6 + 360\,D^8\right)\right)\,\pi\right) \Big/ \left(A\,C^{5/2}\,\left(B\,C^2 - A\,C\,D + 6\,B\,D^2\right)^4\,\left(-A\,C\,D + B\,\left(C^2 + 4\,D^2\right)\right)\,\left(A^2\,C\,D^2 - 6\,A\,B\,D^3 - B^2\,\left(C^3 + 2\,C\,D^2\right)\right)^{3/2}\right)\right)$$

enumerator of L_1 is positive for $C > 0\ \&\&\ D > 0\ \&\&\ B > 0\ \&\&\ A > \frac{B\,C^2 + 6\,B\,D^2}{C\,D}$

denominator of L_1 is negative for $C > 0\ \&\&\ D > 0\ \&\&\ B > 0\ \&\&\ A > \frac{B\,C^2 + 6\,B\,D^2}{C\,D}$

$\gamma = 5$

$L_1 =$

$$\Big(\Big(625\times 5^{2/3} B^4 D^4 \big(3 A^6 C^4 D^6 + 2 A^5 B C^3 D^5 (C^2 + 89 D^2) - A^4 B^2 C^2 D^4 (17 C^4 + 86 C^2 D^2 + 2652 D^4) + 8 A^3 B^3 C D^3 (C^6 - 32 C^4 D^2 - 23 C^2 D^4 + 1251 D^6) + A^2 B^4 D^2 (13 C^8 + 84 C^6 D^2 + 2696 C^4 D^4 + 4968 C^2 D^6 - 6912 D^8) + B^6 C^2 (C^8 + 26 C^6 D^2 + 92 C^4 D^4 - 440 C^2 D^6 - 1632 D^8) - 2 A B^5 C D (5 C^8 - 27 C^6 D^2 - 24 C^4 D^4 + 1108 C^2 D^6 + 3600 D^8)\big)\pi\Big) \Big/ \Big(8\sqrt{2} A^{4/3} C^{13/6} (-A C D + B (C^2 + 4 D^2)) (A C D - B (C^2 + 6 D^2))^{10/3} (A^2 C D^2 - 6 A B D^3 - B^2 (C^3 + 2 C D^2))^{3/2}\Big)\Big)$$

enumerator of L_1 is positive for $C > 0$ && $D > 0$ && $B > 0$ && $A > \frac{B C^2 + 6 B D^2}{C D}$

denominator of L_1 is negative for $C > 0$ && $D > 0$ && $B > 0$ && $A > \frac{B C^2 + 6 B D^2}{C D}$

$\gamma = 6$

$L_1 =$

$$\Big(\Big(81\sqrt{2} B^{5/2} D^{7/2} \big(3 A^6 C^4 D^6 + 4 A^5 B C^3 D^5 (C^2 + 72 D^2) - 3 A^4 B^2 C^2 D^4 (7 C^4 + 44 C^2 D^2 + 1368 D^4) + 8 A^3 B^3 C D^3 (C^6 - 56 C^4 D^2 - 45 C^2 D^4 + 1944 D^6) + A^2 B^4 D^2 (17 C^8 + 168 C^6 D^2 + 4440 C^4 D^4 + 8208 C^2 D^6 - 11\,664 D^8) + B^6 C^2 (C^8 + 28 C^6 D^2 + 48 C^4 D^4 - 1008 C^2 D^6 - 3024 D^8) - 12 A B^5 C D (C^8 - 8 C^6 D^2 + 2 C^4 D^4 + 360 C^2 D^6 + 1080 D^8)\big)\pi\Big) \Big/ \Big(A^{3/2} C^2 \sqrt{-C^3 + \frac{(A^2 - 2 B^2) C D^2}{B^2} - \frac{6 A D^3}{B}} (-A C D + B (C^2 + 4 D^2)) (-A C D + B (C^2 + 6 D^2))^3 (-A^2 C D^2 + 6 A B D^3 + B^2 (C^3 + 2 C D^2))\Big)\Big)$$

enumerator of L_1 is positive for $C > 0$ && $D > 0$ && $B > 0$ && $A > \frac{B C^2 + 6 B D^2}{C D}$

denominator of L_1 is negative for $C > 0$ && $D > 0$ && $B > 0$ && $A > \frac{B C^2 + 6 B D^2}{C D}$

3.2 Subcritical Andronov-Hopf bifurcation

Start with the planar parallelogram ($\gamma = 0$) and compute the first focal value. Observe that it can have any sign. Thus, the Andronov-Hopf bifurcation can be subcritical (unlike for the parallelograms in Section 3.1). Hence, an unstable limit cycle can be born, which is necessarily surrounded by a stable limit cycle (the system is permanent).

In[•]:=
```
fg = κ1 y {1, -1} + κ2 x {0, 2} + κ3 x y^2 {-1, 1} + κ4 y^3 {0, -2} + κ5 x y^2 {0, -2} + κ6 y {0, 2} /.
    {κ1 → 1, κ3 → 1, κ2 → a, κ5 → 4 a, κ6 → b, κ4 → 4 b};
equilibrium = Simplify[Solve[fg == 0 && a > 0 && b > 0 && x > 0 && y > 0][[1]], a > 0 && b > 0];
J = D[fg, {{x, y}}] /. equilibrium;
trJ = Tr[J];
tracevanish = Solve[trJ == 0 && a > 0 && b > 0, b][[1]];
Print["tr J = 0 for ", tracevanish];
ωsubst = {ω → Sqrt[Det[J]]};
derivatives = Simplify[GetDerivatives[fg, equilibrium, 1]];
Get[path];
L1a = Simplify[L1 /. derivatives /. ωsubst /. tracevanish];
Print["L1 = ", L1a];
Print["tr J = 0 and L1 > 0 for ", Reduce[L1a > 0 && trJ == 0 && a > 0 && b > 0, b]]
```

tr $J = 0$ for $\left\{b \to \boxed{\frac{1}{16} \times (3 - 64\,a) \text{ if } 0 < a < \frac{3}{64}}\right\}$

$L_1 = \boxed{\frac{(15 - 1312\,a + 15\,360\,a^2)\,\pi}{12\sqrt{3}} \text{ if } 0 < a < \frac{3}{64}}$

tr $J = 0$ and $L_1 > 0$ for $0 < a < \frac{1}{960} \times \left(41 - \sqrt{781}\right)$ && $b == \frac{1}{16} \times (3 - 64\,a)$

Let us now lift the planar parallelogram by adding a new species in a way that rank of the network remains two (in fact, the Euclidean embedded graph remains a parallelogram).
Verify that the formula for the equilibria and the trace are correct.

In[•]:=
```
fgh = κ1 y z^γ {1, -1, 0} + κ2 x z^γ {0, 2, -γ} +
     κ3 x y^2 {-1, 1, 0} + κ4 y^3 {0, -2, γ} + κ5 x y^2 {0, -2, γ} + κ6 y z^γ {0, 2, -γ} /.
    {κ1 → 1, κ3 → 1, κ2 → a, κ5 → 4 a, κ6 → b, κ4 → 4 b};
equilibrium = {x → 2 t^γ, y → 1/2 t^γ, z → t^2};
trJ = 4 t^(2 γ) ((3/16 - 4 a - b) - γ^2/8 (4 a + b) t^(γ-2));
J = D[fgh, {{x, y, z}}];
Print["the equilibria are given correctly: ",
  Simplify[fgh /. equilibrium, a > 0 && b > 0 && t > 0 && γ > 0] == {0, 0, 0}];
Print["the trace is given correctly: ",
  Simplify[trJ - Tr[J] /. equilibrium, a > 0 && b > 0 && t > 0 && γ > 0] == 0];
```

```
the equilibria are given correctly: True
```

```
the trace is given correctly: True
```

Next, we compute the first focal value.

In[∘]:=
```
fg = Simplify[fgh[[1 ;; 2]] /. {z → ((γ x + γ y + 2 z /. equilibrium) - γ x - γ y)/2}];
derivatives = Simplify[GetDerivatives[fg, equilibrium, 1], a > 0 && b > 0 && t > 0 && γ > 0];
J = D[fg, {{x, y}}] /. equilibrium;
ωsubst = Simplify[{ω → Sqrt[Det[J]]}, a > 0 && b > 0 && t > 0 && γ > 0];
L1abtγ = Simplify[Simplify[L1 /. derivatives] /. ωsubst];
Print["L1 = ", L1abtγ];
```

$$
\begin{aligned}
L_1 = {} & \frac{1}{32\sqrt{2}\,(4t^2+t^\gamma\gamma^2)^3\,\left((4a+b)\times(8t^2+5t^\gamma\gamma^2)\right)^{3/2}}\,\pi\, t^{-3-2\gamma} \\
& \Big(128\times\left(3+131072a^3+24b-1664b^2+12288b^3+4096a^2(-3+40b)+320a(-1-40b+256b^2)\right)t^{12}+ \\
& \quad 32\times\left(9+950272a^3+452b-11936b^2+89600b^3+512a^2(-209+2416b)+\right. \\
& \qquad \left.64a(-15-1588b+9504b^2)\right)t^{10+\gamma}\gamma^2-20(4a+b)^2t^{6\gamma}(-1+\gamma)\gamma^{10}(8a(-4+\gamma)+b(-8+7\gamma))+ \\
& \quad 4t^{8+2\gamma}\gamma^3\left(-9\times(3+\gamma)+32768a^3(2+121\gamma)+24b(-35+124\gamma)-320b^2(-13+145\gamma)+\right. \\
& \qquad 512b^3(-2+847\gamma)+1024a^2(29-526\gamma+8b(2+695\gamma))+32a\left(-3\times(8+15\gamma)+\right. \\
& \qquad\quad \left.\left.64b^2(-2+1421\gamma)-8b(-94+1949\gamma)\right)\right)-t^{4+4\gamma}\gamma^6\left(4096a^3(50-219\gamma+179\gamma^2)-\right. \\
& \qquad 4a\left(81-183\gamma-330\gamma^2+64b^2(-150+781\gamma+11\gamma^2)+16b(-117-71\gamma+96\gamma^2)\right)+ \\
& \qquad b\left(-81+318\gamma+275\gamma^2+b(936+736\gamma-10632\gamma^2)-64b^2(-50+281\gamma+265\gamma^2)\right)+ \\
& \qquad \left.128a^2\left(117+50\gamma-163\gamma^2+8b(150-719\gamma+433\gamma^2)\right)\right)- \\
& \quad 2t^{2+5\gamma}\gamma^8\left(512a^3(91-205\gamma+114\gamma^2)+b^2\left(9\times(3+19\gamma-54\gamma^2)+8b(91-270\gamma+139\gamma^2)\right)+\right. \\
& \qquad 16a^2\left(27+141\gamma-146\gamma^2+8b(273-680\gamma+417\gamma^2)\right)+ \\
& \qquad \left.8ab\left(27+156\gamma-191\gamma^2+4b(273-745\gamma+442\gamma^2)\right)\right)+2t^{6+3\gamma}\gamma^4 \\
& \quad\; \left(3\times(9-17\gamma)\gamma+12b\gamma(-133+101\gamma)+32768a^3(-2+17\gamma+19\gamma^2)-16b^2(36-335\gamma+145\gamma^2)+\right. \\
& \qquad 256b^3(-4+32\gamma+743\gamma^2)+256a^2\left(-36+167\gamma-737\gamma^2+64b(-3+25\gamma+105\gamma^2)\right)+ \\
& \qquad \left.8a\left(-\gamma(339+151\gamma)+128b^2(-12+98\gamma+1125\gamma^2)-16b(36-251\gamma+1476\gamma^2)\right)\right)\Big)
\end{aligned}
$$

3.2.1 Case $\gamma = 2$

In the $\gamma = 2$ case we set may $t = 1$. Further, we eliminate a using $\operatorname{tr} J = 0$. The first focal value formula then becomes very simple.

In[∘]:=
```
asubst = Solve[(trJ /. {γ → 2, t → 1}) == 0 && a > 0 && b > 0, a][[1]];
L1b = Simplify[L1abtγ /. {γ → 2, t → 1} /. asubst];
Print["L1 = ", L1b];
Plot[L1b, {b, 0, 1 / 8}, GridLines → Automatic]
```

$L_1 = \frac{1}{32}\sqrt{7}\left(-7 - 16\,b + 1280\,b^2\right)\pi$ if $0 < b < \frac{1}{8}$

Out[∘]=



3.2.2 Case $\gamma \neq 2$

Solve tr $J = 0$ for t. Also, define the region in the (a, b)-plane, which admits an Andronov-Hopf bifurcation.

In[∘]:=

$$\text{tsubst} = \left\{t \to \left(\frac{\frac{3}{16} - 4\,a - b}{\frac{1}{8} \times (4\,a + b)\,\gamma^2}\right)^{\frac{1}{\gamma - 2}}\right\};$$

$$\text{hopfab} = \left(\frac{3}{16} - 4\,a - b > 0\right) \,\&\&\, a > 0 \,\&\&\, b > 0;$$

Plot the sign of the first focal value for $\gamma = 1$ and $\gamma = 3$.

In[•]:=
```
γsubsts = {{γ → 1}, {γ → 3}};
toshow = {};
For[i = 1, i ≤ Length[γsubsts], i++, {
   γsubst = γsubsts⟦i⟧;
   L1ab = Simplify[L1abtγ /. tsubst /. γsubst];
   L1pos = Reduce[L1ab > 0 && hopfab];
   L1neg = Reduce[L1ab < 0 && hopfab];
   L1zero = Solve[L1ab == 0 && hopfab, b]⟦1⟧;
   amax = L1zero⟦1⟧⟦2⟧⟦2⟧⟦5⟧;

   rplpos = RegionPlot[L1pos, {a, 0, 0.05}, {b, 0, 0.2},
     PlotStyle → {Darker[Red], Opacity[0.1]}, BoundaryStyle → None,
     GridLines → Automatic, AxesLabel → {Style[a, Bold, 16], Style[b, Bold, 16]},
     Frame → None, Axes → True, PlotLabel → Style[StringJoin["The sign of L₁ for γ=",
        ToString[γ /. γsubst]], Bold, 16], ImageSize → Medium];
   rplnega = RegionPlot[L1neg, {a, 0, 0.05}, {b, 0, 0.12},
     PlotStyle → {Darker[Green], Opacity[0.1]}, BoundaryStyle → None];
   rplnegb = RegionPlot[L1neg, {a, 0.01, 0.02}, {b, 0.12, 0.14},
     PlotStyle → {Darker[Green], Opacity[0.1]}, BoundaryStyle → None];
   pl = Plot[b /. L1zero, {a, 0, amax}, PlotStyle → Blue];
   txt = Graphics[{Text[Style["L₁>0", Darker[Red], Bold, 16], {0.005, 0.125}],
      Text[Style["L₁=0", Blue, Bold, 16], {0.015, 0.142}],
      Text[Style["L₁<0", Darker[Green], Bold, 16], {0.025, 0.025}]}];
   toshow = Join[toshow, {Show[rplpos, rplnega, rplnegb, pl, txt]}];
  }];
Row[toshow]
```

Out[•]=





3.3 Two Andronov-Hopf points

Start with the planar parallelogram (set the stoichiometric coefficient of Z to zero in every complex) and compute the first focal value. Observe that it can have any sign. Thus, the Andronov-Hopf bifurcation can be subcritical (unlike for the parallelograms in Section 3.1). Hence, an unstable limit cycle can be born, which is necessarily surrounded by a stable limit cycle (the system is permanent).

In[ ]:=
```
fg = κ1 y {2, -1} + κ2 x^2 {0, 2} + κ3 x^2 y^2 {-2, 1} + κ4 y^3 {0, -2} + κ5 x^2 y^2 {0, -2} + κ6 y {0, 2} /.
   {κ1 → 1, κ3 → 1, κ2 → a, κ5 → 16 a, κ6 → b, κ4 → 16 b};
equilibrium = Simplify[Solve[fg == 0 && a > 0 && b > 0 && x > 0 && y > 0][[1]], a > 0 && b > 0];
J = D[fg, {{x, y}}] /. equilibrium;
trJ = Tr[J];
tracevanish = Solve[trJ == 0 && a > 0 && b > 0, b][[1]];
Print["tr J = 0 for ", tracevanish];
ωsubst = {ω → Sqrt[Det[J]]};
derivatives = Simplify[GetDerivatives[fg, equilibrium, 1]];
Get[path];
L1a = Simplify[L1 /. derivatives /. ωsubst /. tracevanish];
Print["L1 = ", L1a];
Print["tr J = 0 and L1 > 0 for ", Reduce[L1a > 0 && trJ == 0 && a > 0 && b > 0, b]]
```

$\mathrm{tr}\, J = 0$ for $\left\{b \to \boxed{\frac{1}{8} \times (1 - 128\,a) \text{ if } 0 < a < \frac{1}{128}}\right\}$

$L_1 = \boxed{\left(\frac{45}{32} - 592\,a + 36\,864\,a^2\right) \pi \text{ if } 0 < a < \frac{1}{128}}$

$\mathrm{tr}\, J = 0$ and $L_1 > 0$ for $0 < a < \frac{37 - \sqrt{559}}{4608}$ && $b == \frac{1}{8} \times (1 - 128\,a)$

Let us now lift the planar parallelogram by adding a new species in a way that rank of the network remains two (in fact, the Euclidean embedded graph remains a parallelogram).
Verify that the formula for the equilibria and the trace are correct.

In[ ]:=
```
fgh = κ1 y z^γ {2, -1, 0} + κ2 x^2 z^γ {0, 2, -γ} +
    κ3 x^2 y^2 {-2, 1, 0} + κ4 y^3 {0, -2, γ} + κ5 x^2 y^2 {0, -2, γ} + κ6 y z^γ {0, 2, -γ} /.
   {γ → 1, κ1 → 1, κ3 → 1, κ2 → a, κ5 → 4 a, κ6 → b, κ4 → 4 b};
equilibrium = {x → t , y → 1/4 t^2, z → 1/4 t^4};
trJ = 1/4 t^2 (-t^3 + (1 - 4 × (4 a + b)) t^2 - (4 a + b));
J = D[fgh, {{x, y, z}}];
Print["the equilibria are given correctly: ",
  Simplify[fgh /. equilibrium, a > 0 && b > 0 && t > 0 && γ > 0] == {0, 0, 0}];
Print["the trace is given correctly: ",
  Simplify[trJ - Tr[J] /. equilibrium, a > 0 && b > 0 && t > 0 && γ > 0] == 0];
```

```
the equilibria are given correctly: True
```

```
the trace is given correctly: True
```

Next, we find that there are exactly two distinct positive t's for which the trace vanishes if and only if $\boxed{4a+b<1/16}$. The two roots coincide (both of them equal to 1/2) when $4a+b=1/16$.

Further, for $4a+b<1/16$, one root is smaller than $\frac{2}{3}\left(1-4\times(4a+b)\right)$, while the other is larger. The situation is in fact slightly simpler: one root is smaller than 1/2, while the other is larger (this is because $p(1/2)>0$ for any $4a+b<1/16$, see below).

In[∘]:=
```
p = -t^3 + (1 - 4 c) t^2 - c;
Reduce[p == 0 && c > 0 && t > 0, t]
Reduce[D[p, t] == 0 && t > 0 && c > 0, t]
Reduce[Simplify[p /. {t → 1 / 2}] > 0 && c < 1 / 16]
```

Out[∘]=
$$\left(0<c<\frac{1}{16}\ \&\&\ \left(t==\mathrm{Root}\left[c+(-1+4c)\,\#1^2+\#1^3\,\&,2\right]\ ||\ t==\mathrm{Root}\left[c+(-1+4c)\,\#1^2+\#1^3\,\&,3\right]\right)\right)\ ||\ \left(c==\frac{1}{16}\ \&\&\ t==\frac{1}{2}\right)$$

Out[∘]=
$$0<c<\frac{1}{4}\ \&\&\ t==\frac{1}{3}\times(2-8c)$$

Out[∘]=
$$c<\frac{1}{16}$$

Next, we compute the first focal value. We eliminate a using $\mathrm{tr}\,J=0$, so only two parameters are left: b and t.

In[∘]:=

```
fg = Simplify[fgh⟦1 ;; 2⟧ /. {z → ((x + 2 y + 4 z /. equilibrium) - x - 2 y)/4}];
derivatives = Simplify[GetDerivatives[fg, equilibrium, 2], a > 0 && b > 0 && t > 0];
J = D[fg, {{x, y}}] /. equilibrium;
ωsubst = Simplify[{ω → √Det[J]}, a > 0 && b > 0 && t > 0];
L1abt = Simplify[Simplify[L1 /. derivatives] /. ωsubst];
Print["L₁ = ", L1abt];
```

$$L_1 = \frac{1}{8\,t^5\left(1+2\,t^2\right)^2\left((4\,a+b)\times\left(1+t+4\,t^3\right)\right)^{3/2}}\pi\Big(2048\,a^3\,t^3\left(-1+7\,t+6\,t^2-6\,t^3+48\,t^4+8\,t^5+64\,t^7\right)-$$
$$t^5\left(8-34\,t+36\,t^2+t^3-71\,t^4+72\,t^5-12\,t^6-36\,t^7+36\,t^8\right)+$$
$$4\,b\,t^2\left(-4+8\,t+4\,t^2-10\,t^3+21\,t^4+29\,t^5-18\,t^6+38\,t^7+16\,t^8+24\,t^9\right)+$$
$$16\,b^3\left(1+t+20\,t^2+32\,t^3+80\,t^4+228\,t^5+108\,t^6+480\,t^7+496\,t^8+768\,t^{10}\right)+$$
$$8\,b^2\left(2+t+8\,t^2+18\,t^3-32\,t^4+48\,t^5+104\,t^6-220\,t^7+272\,t^8+184\,t^9-320\,t^{10}+448\,t^{11}\right)+$$
$$32\,a^2\,t\Big(-4+11\,t-29\,t^2-80\,t^3+28\,t^4-20\,t^5-372\,t^6+240\,t^7+128\,t^8-512\,t^9+$$
$$448\,t^{10}+8\,b\left(-1+9\,t+66\,t^3+124\,t^4-36\,t^5+480\,t^6+176\,t^7+640\,t^9\right)\Big)+$$
$$8\,a\Big(-t^2\left(4-8\,t+9\,t^2+16\,t^3+49\,t^4+2\,t^5+104\,t^6+116\,t^7-4\,t^8+96\,t^9+48\,t^{10}\right)+$$
$$8\,b^2\left(1+29\,t^2+34\,t^3+132\,t^4+340\,t^5+84\,t^6+864\,t^7+656\,t^8+1280\,t^{10}\right)+$$
$$b\left(4-12\,t+7\,t^2-81\,t^3-376\,t^4-140\,t^5-372\,t^6-1732\,t^7+176\,t^8-544\,t^9-1792\,t^{10}+704\,t^{11}\right)\Big)\Big)$$

Next, we find and plot the regions with the various sign structures of L_1.

In[ ]:=

```
hopf = Solve[trJ == 0 && a > 0 && b > 0 && t > 0, a][[1]];
hopfbt = hopf[[1]][[2]][[2]];
asubst = Normal[hopf];
L1bt = Simplify[L1abt /. asubst];
bsubst = Normal[Solve[L1bt == 0 && hopfbt, b][[1]]];
alim = 1 / 50;
blim = 1 / 16;
amax = 1 / 64;
aspec = Simplify[a /. asubst /. bsubst /. {t → 1 / 2}];
rgn1 = Reduce[Exists[t, trJ == 0 && t < 1/2 && a > 0 && b > 0 && t > 0 && L1abt > 0]];
rgn2a = Reduce[Exists[t, trJ == 0 && t > 1/2 && a > 0 && b > 0 && t > 0 && L1abt > 0]];
rgn2b = Reduce[Exists[t, trJ == 0 && t < 1/2 && a > 0 && b > 0 && t > 0 && L1abt < 0]];
rgn2 = rgn2a && rgn2b;
rgn3 = Reduce[Exists[t, trJ == 0 && t > 1/2 && a > 0 && b > 0 && t > 0 && L1abt < 0]];
rpl1 = RegionPlot[rgn1, {a, 0, alim}, {b, 0, blim},
   PlotStyle → {Darker[Red], Opacity[0.1]}, BoundaryStyle → None,
   GridLines → Automatic, AxesLabel → {Style[a, Bold, 16], Style[b, Bold, 16]},
   Frame → None, Axes → True, PlotLabel → Style["The sign of L1", Bold, 16]];
rpl2 = RegionPlot[rgn2, {a, 0, alim}, {b, 0, blim},
   PlotStyle → {Yellow, Opacity[0.1]}, BoundaryStyle → None];
rpl3 = RegionPlot[rgn3, {a, 0, alim}, {b, 0, blim},
   PlotStyle → {Darker[Green], Opacity[0.1]}, BoundaryStyle → None];
pl1 = Plot[1/16 - 4 a, {a, 0, aspec}, PlotStyle → Darker[Red]];
pl2 = Plot[1/16 - 4 a, {a, aspec, amax}, PlotStyle → Darker[Green]];
ppl1 = ParametricPlot[
   {a /. asubst /. bsubst, b /. bsubst}, {t, 0, 1 / 2}, PlotStyle → Darker[Red]];
ppl2 = ParametricPlot[{a /. asubst /. bsubst, b /. bsubst},
   {t, 1 / 2, 1}, PlotStyle → Darker[Green]];
lpl = ListPlot[{{a /. asubst /. bsubst, b /. bsubst} /. {t → 1 / 2}},
   PlotStyle → Darker[Blue]];
txt = Graphics[{Text[Style["+ +", Bold, 16], {0.0025, 0.036}],
    Text[Style["0 +", Bold, 16], {0.003, 0.028}],
    Text[Style["- +", Bold, 16], {0.004, 0.022}],
    Text[Style["- 0", Bold, 16], {0.005, 0.015}],
    Text[Style["- -", Bold, 16], {0.0075, 0.008}],
    Text[Style["+", Bold, 16], {0.003, 0.054}],
    Text[Style["0", Bold, 16], {0.0075, 0.037}],
    Text[Style["-", Bold, 16], {0.011, 0.022}]}];
Show[rpl1, rpl2, rpl3, pl1, pl2, ppl1, ppl2, lpl, txt]
```

Out[∘]=



Now we find out the sign of L_2 along the $L_1 = 0$ curve in the (a, b)-plane. This is performed by investigating its enumerator and denominator separately. Its denominator is positive, as it becomes apparent below.

In[∘]:=

```
L1abt = Simplify[Simplify[L1 /. derivatives] /. ωsubst];
L2abt = Simplify[Simplify[ω^7 L2 /. derivatives] /. ωsubst];
Reduce[a > 0 && b > 0 && t > 0 && trJ == 0 &&
  Numerator[L1abt] == 0 && Denominator[L2abt] > 0, {b, a}]
Reduce[a > 0 && b > 0 && t > 0 && trJ == 0 && Numerator[L1abt] == 0 && Numerator[L2abt] < 0,
 {b, a}]
Reduce[a > 0 && b > 0 && t > 0 && trJ == 0 &&
  Numerator[L1abt] == 0 && Numerator[L2abt] == 0, {b, a}]
Reduce[a > 0 && b > 0 && t > 0 && trJ == 0 && Numerator[L1abt] == 0 && Numerator[L2abt] > 0,
 {b, a}]
```

Out[∘]=

$$0 < t < 1 \;\&\&\; b == \frac{5t^4 - 11t^5 - 6t^6 - 22t^7 - 64t^8 - 96t^9 - 32t^{10} - 224t^{11}}{4\left(1+4t^2\right)^2\left(1+t+4t^3\right)\times\left(1+t+6t^2+6t^3\right)} + \frac{1}{4}\sqrt{\left(\left(16t^4 + 164t^6 + 244t^7 + 491t^8 + 2766t^9 + 1175t^{10} + 10440t^{11} + 8304t^{12} + 12920t^{13} + 32668t^{14} - 3968t^{15} + 54976t^{16} - 17408t^{17} + 21888t^{18} + 36864t^{19} - 17408t^{20} + 26624t^{21} + 31744t^{22}\right) \Big/ \left(\left(1+4t^2\right)^4\left(1+t+4t^3\right)^2\left(1+t+6t^2+6t^3\right)^2\right)\right)} \;\&\&\; a == \frac{b t^2 - t^4 + 4 b t^4 + t^5}{-4t^2 - 16t^4}$$

Out[•]=

$$0 < t < \boxed{0.958\ldots} \;\&\&\; b == \frac{5 t^4 - 11 t^5 - 6 t^6 - 22 t^7 - 64 t^8 - 96 t^9 - 32 t^{10} - 224 t^{11}}{4 \left(1 + 4 t^2\right)^2 \left(1 + t + 4 t^3\right) \times \left(1 + t + 6 t^2 + 6 t^3\right)} + \frac{1}{4} \sqrt{\left(\left(16 t^4 + 164 t^6 + 244 t^7 + 491 t^8 + 2766 t^9 + 1175 t^{10} + 10440 t^{11} + 8304 t^{12} + 12920 t^{13} + 32668 t^{14} - 3968 t^{15} + 54976 t^{16} - 17408 t^{17} + 21888 t^{18} + 36864 t^{19} - 17408 t^{20} + 26624 t^{21} + 31744 t^{22}\right) \Big/ \left(\left(1 + 4 t^2\right)^4 \left(1 + t + 4 t^3\right)^2 \left(1 + t + 6 t^2 + 6 t^3\right)^2\right)\right)} \;\&\&\; a == \frac{b t^2 - t^4 + 4 b t^4 + t^5}{-4 t^2 - 16 t^4}$$

Out[•]=

$$t == \boxed{0.958\ldots} \;\&\&\; b == \frac{5 t^4 - 11 t^5 - 6 t^6 - 22 t^7 - 64 t^8 - 96 t^9 - 32 t^{10} - 224 t^{11}}{4 \left(1 + 4 t^2\right)^2 \left(1 + t + 4 t^3\right) \times \left(1 + t + 6 t^2 + 6 t^3\right)} + \frac{1}{4} \sqrt{\left(\left(16 t^4 + 164 t^6 + 244 t^7 + 491 t^8 + 2766 t^9 + 1175 t^{10} + 10440 t^{11} + 8304 t^{12} + 12920 t^{13} + 32668 t^{14} - 3968 t^{15} + 54976 t^{16} - 17408 t^{17} + 21888 t^{18} + 36864 t^{19} - 17408 t^{20} + 26624 t^{21} + 31744 t^{22}\right) \Big/ \left(\left(1 + 4 t^2\right)^4 \left(1 + t + 4 t^3\right)^2 \left(1 + t + 6 t^2 + 6 t^3\right)^2\right)\right)} \;\&\&\; a == \frac{-b + t^2 - 4 b t^2 - t^3}{4 + 16 t^2}$$

Out[•]=

$$\boxed{0.958\ldots} < t < 1 \;\&\&\; b == \frac{5 t^4 - 11 t^5 - 6 t^6 - 22 t^7 - 64 t^8 - 96 t^9 - 32 t^{10} - 224 t^{11}}{4 \left(1 + 4 t^2\right)^2 \left(1 + t + 4 t^3\right) \times \left(1 + t + 6 t^2 + 6 t^3\right)} + \frac{1}{4} \sqrt{\left(\left(16 t^4 + 164 t^6 + 244 t^7 + 491 t^8 + 2766 t^9 + 1175 t^{10} + 10440 t^{11} + 8304 t^{12} + 12920 t^{13} + 32668 t^{14} - 3968 t^{15} + 54976 t^{16} - 17408 t^{17} + 21888 t^{18} + 36864 t^{19} - 17408 t^{20} + 26624 t^{21} + 31744 t^{22}\right) \Big/ \left(\left(1 + 4 t^2\right)^4 \left(1 + t + 4 t^3\right)^2 \left(1 + t + 6 t^2 + 6 t^3\right)^2\right)\right)} \;\&\&\; a == \frac{b t^2 - t^4 + 4 b t^4 + t^5}{-4 t^2 - 16 t^4}$$

Now we visualize the sign of L_2.

In[∘]:=
```
tspec = 0.958…;
rpl = RegionPlot[a > 1, {a, 0, alim}, {b, 0, blim},
   PlotStyle → {Darker[Green], Opacity[0.1]}, BoundaryStyle → None,
   GridLines → Automatic, AxesLabel → {Style[a, Bold, 16], Style[b, Bold, 16]},
   Frame → None, Axes → True, PlotLabel → Style["The sign of L₂", Bold, 16]];
ppl1 = ParametricPlot[{a /. asubst /. bsubst, b /. bsubst}, {t, 0, tspec},
   PlotStyle → {Darker[Cyan], Dashed}, PlotRange → {{0, alim}, {0, blim}},
   PlotLegends → Placed[{Style["L₂ < 0", Bold, 16]}, {0.77, 0.81}]];
ppl2 = ParametricPlot[{a /. asubst /. bsubst, b /. bsubst}, {t, tspec, 1},
   PlotStyle → Darker[Magenta], PlotRange → {{0, alim}, {0, blim}},
   PlotLegends → Placed[{Style["L₂ > 0", Bold, 16]}, {0.77, 0.6}]];
lpl = ListPlot[{{a /. asubst /. bsubst, b /. bsubst} /. {t → tspec}}, PlotStyle →
    Darker[Blue], PlotLegends → Placed[{Style["L₂ = 0", Bold, 16]}, {0.78, 0.705}]];
txt = Graphics[{Text[Style["-", Bold, 16], {0.0025, 0.036}],
    Text[Style["0", Bold, 16], {0.003, 0.028}],
    Text[Style["+", Bold, 16], {0.004, 0.022}]}];
Show[rpl, ppl1, ppl2, lpl]
```

Out[∘]=



Finally, we verify that $L_3 < 0$ at the unique point (a, b, t) with $\operatorname{tr} J = L_1 = L_2 = 0$.

In[ ]:=
```
derivatives = Simplify[GetDerivatives[fg, equilibrium, 3], a > 0 && b > 0 && t > 0];
L3bt = Simplify[Simplify[L3 /. derivatives /. asubst] /. ωsubst /. asubst];
Print["L3 = ", N[L3bt /. bsubst /. {t → tspec}]];
```

$L_3 =$ −3.37178